







**P E R S O N A L**  
**COMPUTER**  
COMPUTER **NEWS** LIBRARY

**PAUL STANLEY**

**25 PROGRAMS**  
for the  
**SINCLAIR ZX**  
**MICRODRIVE:**

**MULTI-USER**  
**GAMES**  
for the  
**SPECTRUM**





Pan/Personal Computer News  
Computer Library

**Paul Stanley**

**25 Programs for the**  
**Sinclair ZX Microdrive**

Multi-user games for the Spectrum

Pan Books London and Sydney

First published 1984 by Pan Books Ltd,  
Cavaye Place, London SW10 9PG  
in association with Personal Computer News  
9 8 7 6 5 4 3 2 1

© Paul Stanley 1984  
ISBN 0 330 28274 9

Photoset by Parker Typesetting Service, Leicester  
Printed and bound in Great Britain by  
Richard Clay (The Chaucer Press) Ltd, Bungay, Suffolk

This book is sold subject to the condition that it shall not,  
by way of trade or otherwise, be lent, re-sold,  
hired out or otherwise circulated without the publisher's prior consent  
in any form of binding or cover other than that  
in which it is published and without a similar condition including  
this condition being imposed on the subsequent purchaser



# Contents

<b>1</b>	<b>Damsels In Distress</b>	<b>11</b>
<b>2</b>	<b>Wargame</b>	<b>16</b>
<b>3</b>	<b>Battleships</b>	<b>26</b>
<b>4</b>	<b>Hangman</b>	<b>33</b>
<b>5</b>	<b>Code Breaker</b>	<b>40</b>
<b>6</b>	<b>Consequences</b>	<b>48</b>
<b>7</b>	<b>4-A-Side Soccer</b>	<b>53</b>
<b>8</b>	<b>Split Tennis</b>	<b>61</b>
<b>9</b>	<b>Dominoes</b>	<b>66</b>
<b>10</b>	<b>Boxing</b>	<b>76</b>
<b>11</b>	<b>Hexagrid</b>	<b>84</b>
<b>12</b>	<b>Niagara</b>	<b>92</b>
<b>13</b>	<b>Number Game</b>	<b>98</b>
<b>14</b>	<b>Detective</b>	<b>109</b>
<b>15</b>	<b>The Tortoise and The Hare</b>	<b>119</b>
<b>16</b>	<b>Blackjack</b>	<b>127</b>
<b>17</b>	<b>Pontoon</b>	<b>135</b>
<b>18</b>	<b>Chase</b>	<b>144</b>
<b>19</b>	<b>Bowling Alley</b>	<b>151</b>
<b>20</b>	<b>Minelay</b>	<b>159</b>
<b>21</b>	<b>Nessie</b>	<b>165</b>
<b>22</b>	<b>Morse Trainer</b>	<b>172</b>
<b>23</b>	<b>Bank Robber</b>	<b>179</b>
<b>24</b>	<b>Money Spiders</b>	<b>186</b>
<b>25</b>	<b>Art Dealer</b>	<b>194</b>

# Typing Tips

If you are a new user of the Spectrum, it is important that you should have read through your user manual and be familiar with the keyboard of your machine before you start typing in any programs. Most of all, it is vital that you type the program *exactly* as it appears in the book. To help you avoid the commonest mistakes we have provided some hints below.

- 1 Think before you type. Try to understand as much of what the program is trying to do as you can.
- 2 REMEMBER THAT EVERY CHARACTER IS IMPORTANT TO YOUR COMPUTER AND YOU SHOULD TYPE IN EXACTLY WHAT IS LISTED IN THE BOOK, INCLUDING ALL SPACES.
- 3 Be particularly careful that you distinguish between the following:
  - 1 (one) and I (letter)
  - 0 (zero) and O (letter) (All zeros are 0 in the listings)
  - : (colon) and ; (semi-colon)
  - , (comma) and . (full stop)
  - " (quote marks) and ' (apostrophe)quotation marks with/without a space between them
- 4 When lower case letters are listed they must be entered as such, not as capitals.
- 5 Always SAVE your program on a tape before running it for the first time. If you have any typing errors your program may crash and hours of patient work will be lost.
- 6 Throughout the book, italic characters in the program listings represent graphics characters. To enter the graphics mode, press <CAPS SHIFT> and 9 together, and to exit the graphics mode simply press 9 by itself.  
Capital italics represent the corresponding user-defined graphics characters.

Lower case italic letters represent the pre-defined graphics symbols on the keys 1 to 8:

<b>symbol</b>	<b>how obtained</b>	
<i>a</i>	graphic	8
<i>b</i>	”	1
<i>c</i>	”	2
<i>d</i>	”	3
<i>e</i>	”	4
<i>f</i>	”	5
<i>g</i>	”	6
<i>h</i>	”	7
<i>i</i>	graphic	<CAPS SHIFT> 7
<i>j</i>	”	” 6
<i>k</i>	”	” 5
<i>l</i>	”	” 4
<i>m</i>	”	” 3
<i>n</i>	”	” 2
<i>o</i>	”	” 1
<i>p</i>	”	” 8

See page 92 of the user guide if you require a further explanation.



# Introduction

## **Programming The Spectrum Network**

Welcome to the world of the ZX network! Before we get down to the serious work of the games themselves, a few words of explanation about the programming techniques we've employed in this book. The code required is actually quite straightforward, and with the aid of the following tips you should be able to create your own network games with very little difficulty.

### *Sending the programs across the network and setting up station numbers*

The following line appears in all our programs:

SAVE \*"n";1;"a" LINE 10

This sends the program held in memory to station one, which is then RUN from line 10. When a computer is initially powered up, or when you NEW the memory, the default station setting is number 1. The above statement will not work if station numbers have previously been assigned a value not equal to one, so you must always NEW the memory before LOADING a program. It's important to note that when breaking into a program it's also advisable to NEW the memory and reLOAD the program before attempting to re-commence processing. This is because station numbers quite often change when a program is RUN.

So, when the Spectrum which is to transmit a program is set to station number 1, any other computers on the network will prepare themselves to accept the code, and the first Spectrum will not transmit until this process has been completed. However, if the alternative station number 0 format is used:

SAVE \*"n";0

THE INITIAL COMPUTER DOES NOT WAIT FOR THE REST OF THE NETWORK TO BE READY, BUT TRANSMITS TO ANY COMPUTER WAITING TO RECEIVE A PROGRAM. This process is known as 'Broadcasting', and is demonstrated in our Consequences program (page 49) which can use up to sixty-four Spectrums on the network.

Station numbers can be in the range 1-64 and are best selected by a routine such as this:

```
10 INPUT "Station Number (1 or 2)";st
20 IF st<>1 AND st<>2 THEN GO TO 10
30 FORMAT "n";st
```

Line 30 is the command which actually sets the station number for the network. An alternative to this line could be:

```
30 POKE 23749,st
```

*Ensuring that one computer is transmitting and one is receiving*

This is the most important part of any network program. If two computers are simultaneously waiting for, or sending, an item of information, the program will come to a halt. If a program is badly structured it is quite likely that such a disaster will occur, but with a little care and forward planning the situation can be avoided.

We have found that the simplest and most effective way to make sure that only one computer can transmit or receive at any given moment, is with the inclusion of the following type of routine:

```
100 GOSUB 1000*st
110 GOSUB 1010*st1
120 GO TO 100
```

The variables st and st1 represent the station numbers of your computer and the remote computer respectively. If st=1 and st1=2, the routine at line 1000 will be executed first, followed the subroutine at line 2000. On the other computer st=2 and st1=1, and so processing will go first to 2000 and then to 1000. If the subroutine at 1000 transmits information, and the subroutine at line 2000 receives the same information, an occurrence of the aforementioned disaster is clearly impossible.

An example of this technique can be seen in our Bank Robber program (page 179), where the routine starting at line 1000 moves the robber and transmits his position to the other computer, while lines 2000 onwards receive his position.

The sending and receiving of information is very easy, but you must make sure that variables are received in the same order as they were sent. Before any information is passed a stream must be OPENed. This is achieved by:

```
OPEN #×;"n"st1
```

where stream × is associated to a network channel from your computer to the other. × can be any number in the range 4–15. Information can then be sent by:

```
PRINT #×;k
```

where k is the variable being transmitted and received by:

```
INPUT #×;k
```

If more than one variable is being sent, a series of PRINT #× and INPUT #×

statements can be used, but immediately after these the stream must be CLOSEd by the following command:

CLOSE #×

To simplify matters only stream 4 has been used in the programs in this book.

*Use of INKEY\$ #4*

This command allows the computer to examine a particular stream and access any value which is currently being transmitted. It allows different tasks to be simultaneously performed by both computers, since it does not actually wait for a piece of information to be sent across the network.

An example of this statement can be seen in the Number Game program (p. 98). Line 2060 contains the statement:

LET K\$=INKEY\$#4

This line repeatedly refers to stream 4 to see if an "X" has been sent, which would indicate that the other player had finished their move. Such statements can be used as frequently as desired, but if another computer is sending a variable across the network by means of a PRINT #4 statement, it will still wait until the other computer has received it, just as if INPUT were used instead of INKEY\$ #4.

The INKEY\$ command must be coded with caution if it is being used simultaneously by two computers. For example, our Niagara program (page 92) uses INKEY\$#4 in line 180 to see if the opponent has collected all the fish. There is a possibility that both players could collect the fish at exactly the same time, and would consequently be trying to transmit simultaneously. However, because seven fish have to be collected, such a situation is unlikely to occur. So, when using the INKEY\$ #4 command in this way, make sure that the chance of both tasks being completed simultaneously is as small as possible.

*Removing the border flash*

When any information is transferred from one computer to another the screen border flashes in much the same way as when loading a program from cassette. This can be very irritating, but is easily remedied by a single POKE. If the border colour is black (BORDER 0), then POKE 23750,0 will stop the flashing. Generally, for any colour of border, the same colour number POKED into location 23750 will prevent this flashing.

*Tips on better programming*

The most important feature of a well written program is its structure. A structured program has subroutines for each function, and thus requires a minimal number of GO TO statements. It is very difficult to debug a program which jumps about excessively.

Structured programming techniques are particularly important when writing code for the network. When designing a program, you must work out clearly the task of each routine. You must then decide what information needs to be transferred, and in what order. You can then go on to produce complementary routines to receive the same information in the same order.

If the programs are games in which speed is important, the order of subroutines is a critical factor in the efficiency of the code. The general rule is to put the most commonly used routines at the beginning of the program, and the instructions, graphics and screen displays at the end of the program. The reason for this is that with each call to a subroutine the computer has to search through the program's line numbers from the beginning until it reaches the desired routine. This takes time, and it will obviously find an early subroutine much quicker than one near the end of the program.

Network games written in BASIC cannot be very fast, so you should think carefully about the scenario for any program you write. For example, if you wish to produce a race game, try using snails and not Formula One racing cars as the competing characters!

But enough of the theory let's move on to the listings!



# 1 Damsels in Distress

## **Hardware requirements**

Two 16/48K Spectrums on ZX network

Two televisions

As the title implies, our first excursion into ZX networking gives both players the opportunity to don the shining armour of a gallant knight! The object of the game is simple. Each player must search an invisible maze (created by their opponent), for the obligatory damsel in distress. If you collide with walls they become visible and thus your route is revealed, enabling you to avoid re-tracing your steps.

To succeed in this game you need to master the art of designing impenetrable mazes and learn to apply logic to the task of navigating invisible labyrinths!

When you are designing the maze at the start of the game, you should include plenty of deadends and a winding, awkward route to the damsel. We can't really give tips on the search part of the game, except that you should always think carefully before you make a move!

## **Program Notes:**

- 7 Send program to the other computer
- 30-50 Ensure that one computer is transmitting and the other is receiving.
- 1000-1020 Input player's move and transmit key pressed to the other Spectrum
- 1030 Receive from the other computer the attributes of the new position
- 1035-1060 Move and print your man, unless he hits a wall or encounters the damsel
- 2000-2020 Receive the key pressed by your opponent
- 2030 Transmit the attributes of the new position

- 2035-2050 Move and print your opponent's man, unless he hits a wall or encounters the damsel  
 3000-3010 Routine for your man reaching his damsel  
 3020-3030 Check to see if a key is pressed for a new game  
 4000-4020 Routine for your opponent's man reaching his damsel  
 8000-8157 Set up screen and construct mazes  
 9000-9030 Set screen colours and graphics characters (POKE 23750,7 eliminates border flash)  
 9100-9999 Instructions

```

      1 REM * @ Paul Stanley 1983 *
      7 CLS : PRINT "Type:  LOAD *"
    "n";1 <ENTER>"'"on the other S
pectrum.": SAVE *"n";1;"a" LINE
    10
      10 GO SUB 9000
      20 GO SUB 8000
      30 GO SUB 1000*st
      40 GO SUB 3000-st*1000
      50 GO TO 30
    1000 LET y1=y: LET x1=x: PRINT A
    T 16,0;a$;AT 16,0;"OK PLAYER ";S
    T; ", IT'S YOUR MOVE."; INK 3'"PR
    ESS AN ARROW KEY 5-8"
    1010 LET k$=INKEY$: IF k$<"5" OR
    k$>"8" THEN GO TO 1010
    1020 BEEP .1,0: LET k=VAL k$: OP
    EN #4;"n";st1: PRINT #4;k: CLOSE
    #4
    1030 OPEN #4;"n";st1: INPUT #4;a
    t: CLOSE #4
    1035 LET y=y+(k=6)-(k=7): LET x=
    x+(k=8)-(k=5)
    1040 IF at>64 AND at<128 THEN B
    EEP .5,-20: BEEP .5,-30: PRINT '
    "Bad luck that's a wall!";AT y,x
    ; INK 6; PAPER 1; OVER 1;"B": LE
    T y=y1: LET x=x1: RETURN
    1045 IF at>128 THEN GO TO 3000
    1050 PRINT AT y1,x1; PAPER 3; IN
    K 7;"A"; PAPER 1; INK 6; OVER 1;
    AT y,x;"F"
    1060 BEEP .1,10: GO TO 1000

```

```

2000 PRINT AT 16,0;a$;AT 16,0;"P
LAYER ";ST1;"'S MOVE IN PROGRESS
."
2010 OPEN #4;"n";st1: INPUT #4;k
: CLOSE #4
2020 LET q1=q: LET p1=p: LET q=q
+(k=6)-(k=7): LET p=p+(k=8)-(k=5
)
2030 LET at=ATTR (q,p): OPEN #4;
"n";st1: PRINT #4;at: CLOSE #4
2035 IF at>64 AND at<128 THEN B
EEP .5,-20: BEEP .5,-30: LET q=q
1: LET p=p1: RETURN
2037 IF at>128 THEN GO TO 4000
2040 PRINT AT q1,p1: PAPER 6; IN
K 1;"A";AT q,p;"F"
2050 BEEP .1,10: GO TO 2000
3000 PRINT AT 16,0;a$;AT 16,0; F
LASH 1; INK 2; PAPER 6;"      W
ELL DONE PLAYER ";ST;"!      ";
INVERSE 1;"      YOU SAVED YOUR D
AMSEL!      "
3010 FOR F=1 TO 30: BEEP .03,F:
BEEP .03,30-F: NEXT F
3020 PRINT '"Press any key to pl
ay again."
3030 IF INKEY$="" THEN GO TO 30
30
3040 CLS : GO TO 20
4000 PRINT AT 16,0;A$;AT 16,0; F
LASH 1; INK 2; PAPER 6;" BAD LUC
K! YOUR OPPONENT SAVED      HIS
DAMSEL BEFORE YOU.      "
4010 FOR f=-30 TO 0: BEEP .03,f:
BEEP .03,-f: NEXT f
4020 GO TO 3020
8000 INPUT "Station No? ";st: IF
st<>1 AND st<>2 THEN GO TO 800
0
8002 FORMAT "n";st: LET st1=3-st
8005 LET c$="AAAAAAAAAAAA": LET
b$="CCCCCCCCCCCCCCCC"

```

```

8010 PRINT BRIGHT 1; AT 0,1; INK
    2; PAPER 6; b$; AT 0,17; b$; AT 13,
    1; b$; AT 13,17; b$
8020 FOR f=1 TO 12: PRINT BRIGH
    T 1; PAPER 6; INK 2; AT f,1; "C"; A
    T f,14; "C"; AT f,17; "C"; AT f,30; "
    C"; BRIGHT 0; PAPER 6; INK 0; AT
    f,2; c$; PAPER 1; INK 7; AT f,18; c
    $: NEXT f
8025 LET a$="ENTRANCE": FOR f=1
    TO 9: PRINT AT f,0; INK RND*4; a
    $(f); AT f,16; INK RND*4; a$(f): N
    EXT f
8030 PRINT AT 14,3; INK 3; "YOUR
    MAZE"; TAB 18; "ENEMY'S MAZE"
8040 PRINT "SET UP MAZE:"; INK
    1; " MOVE WITH 5-8. PLACE WIT
    H 9. REMOVE WITH 0" INK 2; "FINA
    LLY POSITION DAMSEL WITH D."; AT
    1,1; " "; AT 1,17; " "
8050 LET b=0: LET x=2: LET y=1:
    PRINT OVER 1; AT y,x; PAPER 6; I
    NK 0; "+"
8060 PRINT BRIGHT b; OVER 1; IN
    K 2*b; PAPER 6; AT y,x; "+"
8070 LET y=y+(INKEY$="6" AND y<1
    2)-(INKEY$="7" AND y>1): LET x=x
    +(INKEY$="8" AND x<13)-(INKEY$="
    5" AND x>2)
8075 LET b=(ATTR (y,x)>64)
8080 PRINT BRIGHT b; AT y,x; OVE
    R 1; INK 2*b; PAPER 6; "+"
8090 IF b=0 AND INKEY$="9" THEN
    PRINT OVER 1; AT y,x; "B": BEEP
    .1,20: LET b=1
8100 IF b=1 AND INKEY$="0" THEN
    PRINT OVER 1; AT y,x; "B": BEEP
    .1,0: LET b=0
8110 IF INKEY$="d" OR INKEY$="D"
    THEN PRINT AT y,x; FLASH 1; IN
    K 3; "D": FOR q=1 TO 5: BEEP .1,q
    : NEXT q: GO TO 8140
8130 BEEP .1,-55: GO TO 8060

```

```

8140 DIM a$(128): PRINT AT 16,0;
a$
8150 INPUT "Press ENTER when you
are both ready to start."; LI
NE b$
8155 LET y=1: LET x=18: PRINT O
VER 1; PAPER 1; INK 6;AT y,x;"F"
8157 LET q=1: LET p=2: PRINT OV
ER 1; PAPER 6; INK 1;AT q,p;"F":
RETURN
9000 CLS : BORDER 7: PAPER 7: IN
K 0: CLS : POKE 23750,7: FOR f=U
SR "a" TO USR "f"+7: READ q: POK
E f,q: NEXT f
9010 DATA 1,1,1,1,1,1,1,1,255,254,
164,254,164,164,254,164,0
9020 DATA 0,127,85,127,85,127,85
,127,56,56,146,124,56,124,254,40
9030 DATA 4,6,31,38,68,64,64,64,
56,56,146,124,56,40,108,0
9050 BORDER 7: PAPER 7: INK 0: P
OKE 23750,7
9100 CLS : PRINT INK 2; PAPER 6
;" DAMSEL IN DISTRESS!
"; PAPER 7;TAB 8; INK 3;"@ PAU
L STANLEY."
9200 PRINT '"This is a 2 player
network game in which the object
is to rescuea damsel who is hid
den in a mazewhich your opponent
constructs."
9300 PRINT '"You also construct
a maze and your opponent tries
to find his damsel as well."
9400 PRINT '"You cannot see your
opponent's maze but if you enc
ounter a wallthis will then be s
hown, but your opponent then
gets a chanceto find his damsel.
"
9500 PRINT '"Press any key to st
art...": PAUSE 0
9999 CLS : RETURN

```

# 2 Wargame

## Hardware requirements

Two 48K Spectrums on ZX network

Two televisions

This is not one of those board games with dice, tin soldiers, steel rules and hedges made out of old sponges! Instead, **Wargame** gives you an opportunity to struggle for domination across the network in an effort to lead your army to victory. The game requires clear, cold tactics and a straight-forward strategy, but be warned – success is never easy!

Each player has at his or her disposal five tank, six gun and seven infantry divisions. The object is to capture the enemy's headquarters or destroy their army. But don't let blood-lust go to your head – you've also got to worry about protecting your own men!

The screen is divided by a river, and until your opponent's army crosses it they will be invisible. Whenever it's time for you to make a move, you must select a division to mobilise and guide it as far as you like, up to the maximum distance allowed. Troops can move the furthest and tanks the least distance.

Should you encounter an enemy division as you move, a battle will take place. If a tank attacks a tank, for example, you have equal chances of winning that battle. Tanks are the most powerful fighting machines at your disposal, but even these could be destroyed by an enemy infantry division.

I make absolutely no apology that this war is totally unrealistic – after all it is a war *game*!

## Program Notes:

- 5 Send program to the other computer
- 100–300 Ensure one player is transmitting and the other is receiving.  
Check to see if an army is destroyed
- 1000–1050 Move and print cursor

1055-1100 Check if the division has gone into the water or a head-quarters  
     1110 Receive from other computer if an enemy division has been encountered  
 1130-1180 Battle action  
     Decide upon the winner and transmit decision to the opponent  
 1500-1525 Calculate which type of division  
 1550-1610 Input a move from the keyboard  
 1700-1720 Decide whether the headquarters invaded is your own or the enemy's  
 2000-2010 Receive from the other Spectrum which division type is being moved  
 2015-2045 Receive the position of the enemy division  
     2050 Send which division type your opponent has met  
     2070 Battle action  
 2080-2150 Receive who won the battle and amend scores accordingly  
 2500-2520 Calculate which division the enemy has met  
 4000-4020 The enemy takes your headquarters  
 4100-4120 You seize the enemy's headquarters  
 5000-5020 Your army has been destroyed  
 5100-5120 You have destroyed the enemy's army  
 8000-8100 Decide which station  
     Set up the screen and variables  
 9000-9060 Set up the graphics characters  
     9080 Data for the machine-code sound  
 9500-9580 Instructions

```

1 REM * @ Paul Stanley 1984 *
2 REM * For 2 48K Spectrums *
5 CLS : PRINT "Type on other
Spectrum: "" "LOAD *""n"";1": SA
VE *""n"";1;"a" LINE 10
10 GO SUB 9000
20 GO SUB 8000
100 GO SUB 1000*st
110 IF s(1)+s(2)+s(3)=0 THEN G
O TO 5000
200 GO SUB 1000*st1
210 IF s(4)+s(5)+s(6)=0 THEN G
O TO 5100
300 GO TO 100

```

```

1000 LET w=0: LET y=11: LET x=a:
  LET i=ATTR (y,x): LET i=4+(2 AND
D st=2): PRINT AT y,x; OVER 1;"+"
"
1005 PRINT #1;"Position + over d
ivision to move"
1010 LET i1=i: LET y1=y: LET x1=
x
1020 LET y=y+(INKEY$="6" AND y<2
1)-(INKEY$="7" AND y>0): LET x=x
+(INKEY$="8" AND x<31)-(INKEY$="
5" AND x>0)
1025 LET i=ATTR (y,x)
1027 IF i>128 THEN LET y=y1: LE
T x=x1: LET i=i1
1030 PRINT INK i1;AT y1,x1; OVE
R 1;"+"; INK i;AT y,x;"+"
1040 IF INKEY$="9" AND i=2+2*st
THEN IF SCREEN$ (y,x)="" THEN
GO SUB 1500: GO TO 1055
1050 GO TO 1010
1055 FOR l=1 TO m: GO SUB 1550
1060 IF attr=175 THEN LET w=2
1070 IF attr=158 THEN GO SUB 17
00
1080 OPEN #4;"n";st1: PRINT #4;y
: PRINT #4;x: PRINT #4;y(f): PRI
NT #4;x(f): PRINT #4;t(f): PRINT
#4;w: CLOSE #4
1085 IF w=2 THEN FOR q=0 TO -40
STEP -5: BEEP .3,q: NEXT q: PRI
NT PAPER 5;AT y(f),x(f); FLASH
1;"F": LET s(t(f))=s(t(f))-1: PR
INT INK 0;AT t(f),29; PAPER 6;s
(t(f)): LET e(f)=0: INPUT "": RE
TURN
1090 IF w=3 THEN GO TO 4100

```



```

1100 IF w=4 THEN FOR q=0 TO -40
STEP -5: BEEP .3,q: NEXT q: PRI
NT INK 6; PAPER 3;AT 0,0; FLASH
1;"AC""BD";AT 20,30;"AC";AT 21
,30;"BD": LET s(t(f))=s(t(f))-1:
PRINT INK 0;AT t(f),29; PAPER
6;s(t(f)): LET e(f)=0: INPUT "":
RETURN
1110 OPEN #4;"n";st1: INPUT #4;k
: CLOSE #4
1120 IF k=0 THEN NEXT 1: INPUT
"": RETURN
1130 FOR q=1 TO 5: PRINT AT y(f)
,x(f); INK i;a$(t(f)): RANDOMIZE
USR USR "1": PRINT AT y(f),x(f)
; INK 4+2*(i=4);a$(k+3): RANDOMI
ZE USR USR "1": NEXT q
1140 RANDOMIZE : LET win=st: IF
k=t(f) THEN LET win=INT (RND*2)
+1
1150 IF k<t(f) THEN LET win=st1
: IF t(f)-k+3<INT (RND*7) THEN
LET win=st
1155 OPEN #4;"n";st1: PRINT #4;w
in: CLOSE #4
1160 IF win=st THEN FOR q=1 TO
20: BEEP .1,q: NEXT q: PRINT AT
y(f),x(f); INK i;a$(t(f)): LET s
(3+k)=s(3+k)-1: PRINT INK 0; PA
PER 6;AT k,31;s(3+k): INPUT "":
RETURN
1170 IF win=st1 THEN FOR q=0 TO
-30 STEP -3: BEEP .2,q: NEXT q:
PRINT AT y(f),x(f); INK 4+2*(st
=1);a$(k+3): LET s(t(f))=s(t(f))
-1: PRINT INK 0; PAPER 6;AT t(f)
,29;s(t(f)): LET e(f)=0: INPUT
"": RETURN
1180 NEXT 1: INPUT "": RETURN
1500 INPUT "": BEEP .1,10: PRINT
INK i; OVER 1;AT y,x;"+"
1510 FOR f=1 TO 15
1520 IF e(f) THEN IF y(f)=y AND
x(f)=x THEN GO TO 1530

```

```

1525 NEXT f
1530 LET m=4+t(f): PRINT #1;"Tha
t division can move ";m;" places
"
1540 OPEN #4;"n";st1: PRINT #4;m
: CLOSE #4: RETURN
1550 LET y=y(f): LET x=x(f): PRI
NT INK 2; BRIGHT 1;AT y,x;a$(t(
f))
1560 LET k$=INKEY$: IF k$<"5" OR
k$>"9" THEN GO TO 1550
1563 IF k$="9" THEN PRINT AT y(
f),x(f); INK i;a$(t(f)): BEEP .2
,0: RETURN
1565 PRINT AT y(f),x(f);" "
1570 IF k$="5" AND x(f)>0 THEN
IF ATTR (y(f),x(f)-1)<>i THEN L
ET x(f)=x(f)-1
1580 IF k$="8" AND x(f)<31 THEN
IF ATTR (y(f),x(f)+1)<>i THEN
LET x(f)=x(f)+1
1590 IF k$="7" AND y(f)>0 THEN
IF ATTR (y(f)-1,x(f))<>i THEN L
ET y(f)=y(f)-1
1600 IF k$="6" AND y(f)<21 THEN
IF ATTR (y(f)+1,x(f))<>i THEN
LET y(f)=y(f)+1
1605 LET attr=ATTR (y(f),x(f))
1610 BEEP .2,20: PRINT INK i;AT
y(f),x(f);a$(t(f)): RETURN
1700 LET w=4: IF st=1 AND y(f)>1
0 THEN LET w=3
1710 IF st=2 AND y(f)<10 THEN L
ET w=3
1720 RETURN
2000 PRINT #1;"Awaiting intelliq
ence.": PAUSE 100: OPEN #4;"n";s
t1: INPUT #4;m: CLOSE #4
2010 INPUT "": PRINT #1;"Opponen
t is moving "; "troops" AND m=7;"
artillery" AND m=6;"tanks" AND m
=5
2015 PAUSE 40: FOR l=1 TO m

```

```

2020 OPEN #4;"n";st1: INPUT #4;y
1: INPUT #4;x1: INPUT #4;yn: INP
UT #4;x: INPUT #4;t: INPUT #4;w:
  CLOSE #4
2030 PRINT AT y1,x1;" "
2031 LET k=0: IF ATTR (yn,x)=2+2
*st THEN GO SUB 2500
2033 IF w=2 THEN PRINT #1;"Enem
y division lost in water.": FOR
f=1 TO 30: BEEP .1,f: NEXT f: LE
T s(t+3)=s(t+3)-1: PRINT INK 0;
  PAPER 6;AT t,31;s(t+3): INPUT "
": RETURN
2035 IF w=3 THEN GO TO 4000
2040 IF w=4 THEN PRINT #1;"Enem
y division destroyed itself": FO
R f=1 TO 30: BEEP .1,f: NEXT f:
LET s(t+3)=s(t+3)-1: PRINT INK
0; PAPER 6;AT t,31;s(t+3): INPUT
  "": RETURN
2045 IF (x>16 AND st1=1) OR (x<1
5 AND st1=2) THEN PRINT INK 2+
2*st1;AT yn,x;a$(t+3)
2050 OPEN #4;"n";st1: PRINT #4;k
: CLOSE #4
2060 IF k=0 THEN NEXT 1: INPUT
"": RETURN
2070 FOR q=1 TO 5: PRINT AT yn,x
; INK 8-2*st;a$(t+3): RANDOMIZE
USR USR "1": PRINT AT yn,x; INK
2+2*st;a$(k): RANDOMIZE USR USR
"1": NEXT q
2080 OPEN #4;"n";st1: INPUT #4;w
in: CLOSE #4
2100 IF win=st1 THEN LET e(f)=0
: FOR q=0 TO -30 STEP -3: BEEP .
2,q: NEXT q: PRINT AT yn,x; INK
2+2*st1;a$(t+3): LET s(k)=s(k)-1
: PRINT INK 0; PAPER 6;AT k,29;
s(k): INPUT "": RETURN

```

```

2110 IF win=st THEN FOR q=1 TO
20: BEEP .1,q: NEXT q: PRINT AT
yn,x; INK 2+2*st;a$(k): LET s(3+
t)=s(3+t)-1: PRINT INK 0; PAPER
6;AT t,31;s(3+t): INPUT "": RET
URN
2150 NEXT 1: INPUT "": RETURN
2500 FOR f=1 TO 15
2510 IF e(f) THEN IF y(f)=yn AN
D x(f)=x THEN LET k=t(f): RETUR
N
2520 NEXT f
4000 PRINT AT yn,x; INK 2+2*st1;
a$(t+3)
4010 FOR f=0 TO -30 STEP -3: BEE
P .1,f: NEXT f
4020 CLS : PRINT AT 10,0;"ENEMY
HAS SEIZED YOUR HQ STOP PRESS
ANY KEY TO PLAY AGAIN STOP": PAU
SE 0: GO TO 20
4100 PRINT AT Y(F),X(F); INK 2+2
*st;a$(t(f));a$(t(f))
4110 FOR f=1 TO 30: BEEP .1,f: N
EXT f
4120 CLS : PRINT AT 10,0;"CONGRA
TULATIONS STOP YOU HAVE SEIZED
THE ENEMY'S HEAD QUARTERSSTOP P
RESS ANY KEY TO PLAY AGAINSTOP":
PAUSE 0: GO TO 20
5000 FOR F=0 TO -30 STEP -3: BEE
P .1,F: NEXT F
5010 CLS : PRINT AT 10,0;"ALL AR
MY HAS BEEN DESTROYED STOPPRESS
ANY KEY TO PLAY AGAIN STOP": PAU
SE 0: GO TO 20
5100 CLS : PRINT AT 10,0;"WELL D
ONE STOP YOUR ARMY HAS WIPED
OUT THE ENEMY STOP YOU AREVICTOR
IOUS STOP PRESS ANY KEY TOPLAY A
GAIN STOP": PAUSE 0: GO TO 20
8000 CLS : INPUT "Station No ? "
;st: IF st<>1 AND st<>2 THEN GO
TO 8000

```

```

8010 FORMAT "n";st: LET st1=1+(s
t=1)
8020 PRINT PAPER 3; INK 6; FLASH
H 1;AT 0,0;"AC""BD";AT 20,30;"A
C";AT 21,30;"BD"
8030 FOR f=0 TO 21: PRINT FLASH
1; PAPER 5;AT f,15;"FF": NEXT f
: PRINT AT 2,15;" ";AT 19,15;"
"
8040 DIM y(15): DIM x(15): DIM t
(15): DIM e(15)
8050 IF st=1 THEN LET a$="IKGJH
6": LET f=1: FOR q=1 TO 3: LET y
=0: FOR h=1 TO q+3: LET e(f)=1:
LET t(f)=q: LET x(f)=q+2-y: PRIN
T INK 4;AT y,x(f);a$(t(f)): LET
y(f)=y: LET f=f+1: LET y=y+1: N
EXT h: NEXT q
8060 IF st=2 THEN LET a$="JHGIK
6": LET f=1: FOR q=1 TO 3: LET y
=21: FOR h=1 TO q+3: LET e(f)=1:
LET t(f)=q: LET x(f)=50-q-y: PR
INT INK 6;AT y,x(f);a$(t(f)): L
ET y(f)=y: LET f=f+1: LET y=y-1:
NEXT h: NEXT q
8070 FOR f=1 TO 8: FOR q=1 TO f:
PRINT AT f+13,q-1; FLASH 1; PAP
ER 5;"F";AT 8-f,32-q;"F": NEXT q
: NEXT f
8080 PRINT AT 0,29;"YEO": DIM s(
6): FOR f=1 TO 3: LET s(f)=f+3:
LET s(f+3)=s(f): PRINT PAPER 6;
INK 0;AT f,29;s(f); INK 1;a$(f)
; INK 0;s(f): NEXT f
8090 PRINT INK 6;AT 18,0;"MOVE"
' INVERSE 1;"6..7""K..6""I..5"
8100 LET a=10+14*(st=2): RETURN
9000 FOR f=USR "a" TO USR "p"+3:
READ q: POKE f,q: NEXT f
9010 DATA 0,102,102,102,102,102,
102,126,126,102,102,102,102,102,
102,0

```

```

9020 DATA 0,48,120,252,204,204,2
04,204,204,236,244,220,124,5
4,2
9030 DATA 0,0,174,168,174,226,76
,0,110,49,204,25,108,35,216,102
9040 DATA 0,32,112,32,84,14,4,10
,0,0,126,12,30,30,12,0
9050 DATA 0,120,127,48,126,255,1
26,0,0,30,254,12,126,255,126,0
9060 DATA 0,0,126,48,120,120,48,
0
9070 REM * Machine Code Data fol
lows.      check CAREFULLY befor
e RUN *
9080 DATA 6,1,197,33,100,0,17,5,
0,229,205,181,3,1,1,0,17,0,1,225
,198,0,237,74,229,198,0,237,82,2
25,56,230,193,16,223,201
9500 CLS : BORDER 1: INK 7: PAPE
R 0: POKE 23750,1
9505 CLS : PRINT FLASH 1; PAPER
2;"WARGAME "; INVERSE 1;" @ Pau
l Stanley "; INVERSE 0;" WARGAME
"
9510 PRINT "A fascinating netwo
rk game of strategy for 2 play
ers."
9520 PRINT "The object is to ca
pture your opponent's HQ, whil
st defending your own.""You can
not see enemy divisions if they
are on the other side of the riv
er."
9530 PRINT "The numbers of tank
s, guns and troops are shown in
the top right (You & Opposi
tion) and the distance these can
move in the bottom left."
9540 PRINT "Moves are alternate
, and the game finishes when
an army has been wiped out or H
Q captured."
9550 INPUT "Press ENTER to read
more."; LINE a$

```

```
9560 CLS : PRINT "It is likely  
that a tank will defeat a gun,  
and a gun will probably defe  
at troops, but you cannot be sur  
e."
```

```
9570 PRINT "All movement is wit  
h cursor keys5-8, and 9 is used  
to select division or as a bl  
ank move."
```

```
9580 INPUT "Press ENTER to comme  
nce battle"; LINE a$: RETURN
```

# 3 Battleships

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

I'm sure everyone has played the pen and paper game of Battleships and Cruisers, well now you can play a zappier update over the ZX network.

Both players set up their flotillas by entering the positions of six ships on a twelve by twelve grid. Having done this, they take turns in dropping 3D bombs on their opponent. Unlike the traditional game, if a ship is hit it will be completely destroyed.

If your bomb misses its mark, the Spectrum may offer some information to help you locate other ships. For example, you will be told if there is a ship in the same vertical column or horizontal row but if there are ships both in that row and column, you will be told only of the one in the horizontal row.

Without this feature the game would involve only luck, but with it, your judgement (and memory) is tested and the game becomes a good deal more interesting than putting crosses on pieces of paper!

## Program Notes:

- 3 Send the program to the other Spectrum
- 60-80 Ensure one computer transmits while the other receives
- 100-195 Move and print the cursor
- 205 Send the position of the cursor to the other computer
- 210-280 Receive from the other Spectrum whether you hit your target  
Receive the position of the enemy ship and print it
- 300-390 Receive the position of the enemy bomb  
Check for a hit  
Transmit whether the bomb hit or not
- 400-450 Transmit the co-ordinates of the ship  
Display ship



500-530 Indicate if the bomb coincides with a co-ordinate of a ship  
 1000-1140 Routine for the falling bomb graphics  
 2000-2400 Decide upon the winner  
           Check to see if a new game is requested  
 7000-7150 Instructions  
 8000-8080 Set up the screen  
 8000-8800 Input the ship positions  
 9000-9999 Define the graphics characters

```

      1 REM * @ PAUL STANLEY 1984 *
      2 CLS : PRINT "Type on other
Spectrum:'''LOAD *""n";1"
      3 SAVE *""n";1;"A" LINE 5
      5 CLS : PAPER 7: BORDER 7: IN
K Ø: CLS : POKE 23658,9: POKE 23
750,7
     10 GO SUB 9000
     15 GO SUB 7000
     20 GO SUB 8000
     50 LET Q=16: LET P=18
     60 GO SUB 100+(200 AND st=2)
     70 GO SUB 100+(200 AND st=1)
     80 GO TO 60
    100 PRINT OVER 1;AT Q,P;"+"
    110 PRINT AT 18,Ø; PAPER 4;"
        YOUR MOVE
    120 PRINT AT 20,Ø; INK 3;"Move
your cross about with 5-8. Fire
by pressing 9."
    130 PRINT OVER 1;AT Q,P;"+"
    140 IF INKEY$="6" THEN LET Q=Q
+1: IF Q=17 THEN LET Q=5
    150 IF INKEY$="7" THEN LET Q=Q
-1: IF Q=4 THEN LET Q=16
    160 IF INKEY$="8" THEN LET P=P
+1: IF P=30 THEN LET P=18
    170 IF INKEY$="5" THEN LET P=P
-1: IF P=17 THEN LET P=29
    180 IF INKEY$="9" THEN GO TO 2
ØØ
    190 PRINT OVER 1; INK 3;AT Q,P
; "+"
    195 BEEP .06,-10: GO TO 130
  
```

```

200 BEEP .05,30: PRINT AT Q,P;
OVER 1;"+"
205 OPEN #4;"n";st1: PRINT #4;q
: PRINT #4;p: CLOSE #4
210 LET Y=Q: LET X=P: GO SUB 10
00: OPEN #4;"n";st1: INPUT #4;hi
t: INPUT #4;po: CLOSE #4
220 IF hit=0 THEN GO SUB 500:
BEEP .3,-10: BEEP .3,-20: BEEP .
3,-30: RETURN
230 OPEN #4;"n";st1: INPUT #4;c
: CLOSE #4
250 FOR G=7 TO 0 STEP -1: PRINT
INK q;AT Q,c+14;"RS"
260 FOR I=0 TO G*8 STEP G+1: BE
EP .06,I: NEXT I
265 NEXT G
270 LET sc1=sc1+1: IF SC1=6 THE
N GO TO 2000
280 RETURN
300 PRINT AT 18,0; PAPER 4;"
OPPONENT'S MOVE ";
PAPER 7;AT 20,0;"

```

"

```

305 OPEN #4;"n";st1: INPUT #4;b
: INPUT #4;a: CLOSE #4
360 LET y=b: LET x=a-14: GO SUB
1000
370 LET PO=0: LET hit=0: FOR i=
1 TO 6
376 IF NOT H(I) AND b=Y(I) THEN
IF a-14=X(I) OR a-14=X(I)+1 TH
EN LET hit=1: GO TO 390
380 IF NOT H(I) THEN IF X(I)=a
-14 OR x(i)=a-15 THEN LET PO=1
382 IF NOT H(I) AND y(I)=b THEN
LET PO=2: GO TO 390
385 NEXT i
390 OPEN #4;"n";st1: PRINT #4;h
it: PRINT #4;PO: CLOSE #4: IF hi
t=0 THEN RETURN

```

```

400 LET H(I)=1: OPEN #4;"n";st1
: PRINT #4;x(i): CLOSE #4: FOR o
=7 TO 0 STEP -1: PRINT AT y(i),x
(i); INK o;"RS"
420 FOR l=0 TO o*8 STEP o+1: BE
EP .06,1: NEXT l
425 NEXT o
440 LET SC2=SC2+1: IF SC2=6 THE
N GO TO 2010
450 RETURN
500 IF PO=1 THEN PRINT AT 20,0
;"THERE IS A SHIP IN THAT COLUMN
"
510 IF PO=2 THEN PRINT AT 20,0
;"THERE IS A SHIP IN THAT ROW.
"
530 RETURN
1000 PRINT OVER 1; INK 5;AT y-1
,x-4;"pppappppo";AT y,x-4;"ppppp
pppp";AT y+1,x-4;"pppapppppl"
1005 BEEP .08,11
1010 PRINT OVER 1;AT y-1,x-4;"p
ppappppo";AT y,x-4;"pppppppppp";A
T y+1,x-4;"pppapppppl"
1020 PRINT OVER 1; INK 5;AT y-1
,x-3;"ppapppo";AT y,x-3;"ppdpppp
c";AT y+1,x-3;"ddaddddc"
1025 BEEP .085,9
1030 PRINT OVER 1;AT y-1,x-3;"p
papppo";AT y,x-3;"ppdppppc";AT y
+1,x-3;"ddaddddc"
1040 PRINT OVER 1; INK 5;AT y-1
,x-3;"em mmA";AT y,x-3;"fpdppB"
1045 BEEP .09,7
1050 PRINT OVER 1;AT y-1,x-3;"e
m mmA";AT y,x-3;"fpdppB"
1060 PRINT OVER 1; INK 5;AT y,x
-2;"pCpDa"
1065 BEEP .095,5
1070 PRINT OVER 1;AT y,x-2;"pCp
Da"

```

```

1080 PRINT OVER 1; INK 5; AT y,x
-1; "EFG"
1085 BEEP .1,4
1090 PRINT OVER 1; AT y,x-1; "EFG"
"
1100 PRINT , OVER 1; INK 5; AT y,x
-1; "HI"
1105 BEEP .105,2
1110 PRINT OVER 1; AT y,x-1; "HI"
1120 PRINT OVER 1; INK 5; AT y,x
; "J"
1125 BEEP .11,0
1130 PRINT OVER 1; AT y,x; "J"
1140 RETURN
2000 CLS : PRINT AT 2,0; FLASH 1
; "WELL DONE! YOU SUCCESSFULLY
DESTROYED YOUR OPPONENT'S SIX
SHIPS!!!!!!": GO TO 2020
2010 CLS : PRINT AT 2,0; FLASH 1
; "BAD LUCK! YOUR OPPONENT HAS
BEATEN YOU THIS TIME. TRY AGAI
N PERHAPS?????"
2020 FOR F=1 TO 20: BEEP .03,F:
BEEP .03,20-F: NEXT F
2200 PRINT AT 8,0; "PRESS ANY KEY
TO PLAY AGAIN..."
2300 IF INKEY$="" THEN GO TO 23
00
2400 CLS : GO TO 20
7000 FOR Y=20 TO 4 STEP -4: LET
X=27-Y: GO SUB 1000: NEXT Y
7005 LET A$="BATTLESHIPS "
7010 PRINT AT 2,0;: FOR f=1 TO 5
: FOR G=1 TO 13: PRINT PAPER 6;
INK RND*4; A$(G);: BEEP .01,G+F*
3: NEXT G: NEXT F
7020 PRINT AT 4,0; " "
7050 PRINT PAPER 4' " @ P A
U L S T A N L E Y "
7100 PRINT "' "A network game for
2 players."
7110 PRINT INK 4' "The winner i
s the first to destroy all
six enemy ships."

```

```

7120 PRINT INK 3;"Full instruc
tions are given as you proceed.
"
7130 PRINT PAPER 4;"PRESS ANY
KEY TO START .... "
7140 IF INKEY$="" THEN GO TO 71
40
7150 CLS : RETURN
8000 INPUT "Enter station No:";s
t: IF st<>1 AND st<>2 THEN GO T
O 8000
8005 FORMAT "n";st: LET st1=1+(s
t=1)
8010 FOR f=5 TO 16: PRINT AT f,4
;"XXXXXXXXXXXXL XXXXXXXXXXXXL": N
EXT f
8025 PRINT AT 0,0; PAPER 2; INK
7;"B A T T L E S H I P
S"
8030 PLOT 32,135: DRAW 95,0
8040 PLOT 144,135: DRAW 95,0
8050 FOR f=1 TO 12: PRINT AT f+4
,2;CHR$(64+f): NEXT f
8060 FOR f=1 TO 9: PRINT INK 2;
AT 3,f+3;f: NEXT f
8070 PRINT INK 2;AT 3,13;"OPQ"
8080 PRINT AT 1,0; PAPER 5;"
YOUR SHIPS ENEMY SHIPS "
8100 PRINT INK 3;AT 18,0;"You c
an place 6 ships by typing ""D4"
" (for example). Then press ENT
ER."
8200 DIM a$(3): DIM Y(6): DIM X(
6): DIM H(6)
8206 FOR f=1 TO 6
8210 INPUT a$
8220 IF a$(1)<"A" OR a$(1)>"L" T
HEN GO TO 8210
8230 LET y=CODE a$(1)-60
8250 LET x=VAL a$(2 TO )+3
8260 IF x<4 OR x>14 THEN GO TO
8210

```

```

8265 FOR I=1 TO 6: IF Y=Y(I) THE
N IF X=X(I)-1 OR X=X(I) OR X=X(
I)+1 THEN GO TO 8210
8270 NEXT I
8280 PRINT OVER 1;AT y,x;"MN"
8290 LET Y(F)=Y: LET X(F)=X
8300 BEEP .04,10
8310 NEXT f
8400 PRINT AT 18,0;"

```

"

```

8740 LET sc1=0: LET sc2=0
8800 RETURN
9000 FOR y=USR "A" TO USR "S"+7:
  READ x: POKE y,x: NEXT y
9010 DATA 0,0,0,0,224,248,252,25
4,255,255,255,255,254,252,248,22
4
9020 DATA 31,31,31,255,255,31,31
,31,224,240,248,248,248,248,240,
224
9030 DATA 0,63,63,63,63,63,63,0,
0,63,63,255,255,63,63,0
9040 DATA 0,224,240,248,248,240,
224,0,0,0,60,63,63,60,0,0
9050 DATA 0,0,252,254,254,252,0,
0,0,0,0,222,255,222,0,0
9060 DATA 128,128,128,128,128,12
8,128,255,129,129,129,129,129,12
9,129,255
9070 DATA 0,2,34,59,63,255,127,0
,0,0,32,48,120,126,124,0
9080 DATA 0,38,41,41,41,41,38,0,
0,20,20,20,20,20,20,0,0,76,82,68
,72,80,94,0
9090 DATA 128,128,208,164,254,23
1,255,255,128,128,136,151,200,17
4,252,255
9999 RETURN

```

# 4 Hangman

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

It's time that this old favourite was livened up a little, and this network version will certainly prove more interesting than traditional implementations, primarily because it allows two players to compete simultaneously.

The computer first asks you to think of a word or phrase and to INPUT the subject area into which it falls. For example, suitable subjects could be television programmes, pop groups or animals. After the subject category has been safely consigned to memory you must enter the word in question.

From this point on it's simply a matter of guessing your opponent's word by entering letters which you think may be in it. When a guess is correct, the appropriate letter(s) will appear in the space at the top of the screen.

You can see how close your opponent is to victory because your word is displayed at the bottom of your screen. As he makes correct guesses the letters in question will 'light up' on the screen.

If you guess the word first, you're the winner, but should you hang . . . well, you are definitely the loser!

## Program Notes:

- 5 Send the program to the other computer
- 100-149 Input the word and subject  
Check the length of these
- 150-160 Transmit and receive words and subjects
- 170-190 Set up the screen
- 200-220 Ensure that one computer is transmitting while the other is receiving

- 1000-1090 Input a letter  
 Transmit this guess to the other Spectrum  
 Check to see if the letter is in the word  
 Print the scores
- 1100-1120 Increase the hangman  
 Check to see if the man is hung
- 2000-2080 Receive from the other computer the guessed letter  
 Check to see if the letter is in the word  
 Print the scores
- 3000-3110 Routines for when a word has been guessed
- 4000-4110 Routines for when a player is hung
- 5000-5100 Draw the various components of the scaffold and the man
- 7000-9700 Title page and instructions

```

1 REM * @ Paul Stanley 1983 *
5 CLS : PRINT "Type on other
Spectrum: ""LOAD *""n"";1": SA
VE *""n"";1;"a" LINE 10
10 CLS : GO SUB 7000: BORDER 7
: PAPER 7: INK 0: CLS : POKE 236
58,8: POKE 23750,7
15 INPUT "Station No? ";st: IF
st<>1 AND st<>2 THEN GO TO 15
20 FORMAT "n";st: LET st1=1+(s
t=1)
100 BEEP .3,20: CLS : PRINT "Th
ink of a word/expression etc.""
INK 2' "Then ENTER the subject of
this word/expression in 32 let
ters or less."
110 INPUT t$: IF LEN t$>32 THEN
BEEP .2,-20: GO TO 110
140 PRINT INK 3' "Now ENTER the
word/expression. (MAX length 3
2 characters)."
145 BEEP .02,20: INPUT w$: IF L
EN w$>32 THEN BEEP .2,-20: GO T
O 145
147 FOR f=1 TO LEN w$: IF w$(f)
=" " THEN GO TO 149
148 IF w$(f)<"A" OR w$(f)>"Z" T
HEN BEEP .2,-20: GO TO 145
149 NEXT F

```



```

150 BEEP .02,20: IF st=1 THEN
OPEN #4;"n";st1: PRINT #4;t$: PR
INT #4;w$: CLOSE #4: OPEN #4;"n"
;st1: INPUT #4;u$: INPUT #4;x$:
CLOSE #4
160 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;u$: INPUT #4;x$: CL
OSE #4: OPEN #4;"n";st1: PRINT #
4;t$: PRINT #4;w$: CLOSE #4
170 CLS : PRINT AT 0,0;"Your op
ponent's subject is:-": PRINT I
NK 2;u$
180 FOR f=1 TO LEN x$: PRINT AT
3,f-1; INK 2;"A" AND x$(f)<>" "
; INK 3;"/" AND x$(f)=" ": NEXT
f
182 PRINT AT 20,0;"Your word is
:-": FOR f=1 TO LEN w$: PRINT AT
21,f-1; INK 4;w$(f) AND w$(f)<>
" "; BRIGHT 1; INK 1;"/" AND w$(
f)=" ": NEXT f
185 INK 4: PLOT 119,47: DRAW 57
,0: DRAW 0,89: DRAW -57,0: DRAW
0,-89: INK 0
190 PRINT AT 5,15; PAPER 2; INK
6;"GUESSES": LET y=6: LET x=15:
LET h=5000: LET h1=h
200 GO SUB 1000*st
210 GO SUB 1000*st1
220 GO TO 200
1000 PRINT AT 6,24;"PRESS";AT 7,
26;"A";AT 8,23;"LETTER:";
1010 LET a$=INKEY$: IF a$>"Z" OR
a$<"A" THEN GO TO 1010
1020 PRINT INK 2;a$: BEEP .4,10
1030 PRINT INK 2;AT y,x;a$: LET
x=x+2: IF x=23 THEN LET x=15:
LET y=y+2
1040 OPEN #4;"n";st1: PRINT #4;a
$: CLOSE #4
1050 LET t=0: LET w=0: FOR f=1 T
O LEN x$

```

```

1060 IF x$(f)=a$ THEN LET w=1:
PRINT AT 3,f-1; INK 3;a$: BEEP .
03,10
1065 IF ATTR (3,f-1)=59 THEN LE
T t=t+1
1070 NEXT f
1080 PRINT AT 6,24;"      ";AT 7,
26;" ";AT 8,23;"      "
1085 IF t=LEN X$ THEN GO TO 310
0
1090 IF w=1 THEN RETURN
1100 BEEP .4,-20: GO SUB h: BEEP
.4,-30
1110 LET h=h+10: IF h=5110 THEN
GO TO 4000
1120 RETURN
2000 PRINT FLASH 1;AT 17,20;"OP
PONENT'S";AT 18,24;"GO"
2010 OPEN #4;"n";st1: INPUT #4;a
$: CLOSE #4
2020 LET q=0: LET a=LEN w$: FOR
f=1 TO a
2030 IF w$(f)=a$ THEN PRINT AT
21,f-1; BRIGHT 1; INK 1;a$: BEEP
.1,10: LET q=q+1
2040 NEXT f
2045 IF q=0 THEN LET h1=h1+10:
IF h1=5110 THEN GO TO 4100
2047 PRINT AT 17,20;"      "
;AT 18,24;"      "
2048 IF q=0 THEN RETURN
2050 LET q=0: FOR f=1 TO a
2060 IF ATTR (21,f-1)>64 THEN L
ET q=q+1
2070 NEXT f: IF q=a THEN GO TO
3000
2080 RETURN
3000 CLS : PRINT AT 10,0;"BAD LU
CK! YOUR OPPONENT BEAT YOU": PRI
NT INK 3;"YOUR WORD WAS:-": PRI
NT INK 2;X$
3010 FOR F=1 TO 10: BEEP .3,F: N
EXT F

```

```

3020 PRINT INK 3;AT 17,0;"PRESS
  ANY KEY TO PLAY AGAIN..": PAUSE
  0
3030 GO TO 15
3100 CLS : PRINT AT 10,0;"WELL D
ONE! YOU CORRECTLY GUESSED"; INK
  2;X$
3110 GO TO 3010
4000 BEEP .5,-20: BEEP .5,-30: C
  LS
4010 PRINT AT 10,0;"BAD LUCK! YO
  U'RE DEAD. THE WORD YOU WERE LOO
  KING FOR WAS:-": PRINT INK 2;X$
4020 GO TO 3010
4100 CLS : PRINT AT 10,0;"WELL D
ONE! YOUR OPPONENT HAS HANGED
  . THE WORD YOU WERE TRYINGTO GET
  WAS:": PRINT INK 2;X$
4110 GO TO 3010
5000 PRINT AT 17,1; INK 5;"ppppp
  pppppppppp": RETURN
5010 FOR F=5 TO 16: PRINT INK 3
  ;AT F,2;"p": NEXT F: RETURN
5020 PRINT AT 4,2; INK 4;"mmmmmm
  mm": RETURN
5030 FOR F=0 TO 5: INK 2: PLOT 2
  4,112+F: DRAW 23-F,23-F: NEXT F:
  INK 0: RETURN
5040 PRINT INK 1;AT 5,9;"k": RE
  TURN
5050 INK 2: CIRCLE 73,118,9: PLO
  T 70,120: DRAW 1,0: DRAW 0,1: DR
  AW -1,0: PLOT 76,120: DRAW 1,0:
  DRAW 0,1: DRAW -1,0
5052 PLOT 73,118: DRAW 3,-3: DRA
  W -2,0: DRAW -1,1
5056 PLOT 69,115: DRAW 10,0,2: I
  NK 0: RETURN
5060 FOR F=72 TO 75: PLOT F,104:
  DRAW 0,4: NEXT F
5062 PRINT INK 3;AT 9,8;"ppk";
  INK 5;AT 10,7;"fppp"; INK 3;AT 1
  1,8;"ppk"; INK 5;AT 12,8;"bd": R
  ETURN

```

```

5070 PLOT 83,104: DRAW 10,10: DR
AW 1,0: DRAW -10,-10: FOR F=1 TO
 3: CIRCLE 97,116,F: NEXT F: RET
URN
5080 PLOT 64,104: DRAW -10,10: D
RAW -1,0: DRAW 10,-10: FOR F=1 T
O 3: CIRCLE 51,116,F: NEXT F: RE
TURN
5090 INK 5: FOR F=1 TO 4: PLOT 6
7+F,79: DRAW -20,-30: NEXT F
5092 INK 0: PRINT AT 16,5;"dc":
RETURN
5100 INK 5: FOR F=1 TO 4: PLOT 7
5+F,79: DRAW 20,-30: NEXT F
5102 INK 0: PRINT AT 16,12;"dc":
RETURN
5110 PRINT AT 20,0;"BAD LUCK! YO
U'VE BEEN HANGED. THE ANSWER W
AS ";B$
5120 GO TO 3010
7000 FOR L=5000 TO 5100 STEP 10:
  GO SUB L: NEXT L
7080 FOR F=1 TO 3: FOR I=1 TO 11
: BEEP .02,I: NEXT I: LET I=USR
3280: NEXT F
8000 PRINT INK 7;"papaepiapiapa
pppapoanpaepiapi papapapapapqapa
paaapfmk papapapqapappppapppapbipa
paaapapapappppapbipapapapapapapqpa
papapaaaapapapapapapapapabpa
pppapaaaapapapapabp"
8010 FOR F=7 TO 0 STEP -1: FOR I
=17 TO 21: PRINT AT I,0; OVER 1;
  INK F;"
  ": BEEP .01,5: BEEP .01,
10: NEXT I: NEXT F
8100 FOR f=0 TO 6: POKE USR "a"+
f,0: NEXT f: POKE USR "a"+7,126
9000 PRINT AT 5,16; INK 3;"@ Pau
l Stanley"
9500 PAUSE 250: CLS

```

```
9600 PRINT "This is a 2 player  
hangman game using the ZX network."  
"It is the standard game, but is a race between the 2 players to find the opponent's word first."  
9700 PAUSE 500: CLS : RETURN
```

# 5 Code Breaker

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

This is a game which involves a lot of brainwork, combined with supple fingers!

The object is to find your opponent's secret code number. To start, each player enters a four digit number; each digit must be in the range 0 to 5. Once the codes are secure in the Spectrums' memories the guessing begins! The start screen displays five option modes: Colour, Position, Move, Delete and Enter. If none of the modes are flashing, you must select an option by pressing C, P, M, D or E. Colour mode allows you to select a colour, code 0-5. Once you have selected a colour, the Spectrum automatically switches to Position mode (1-4) and this will establish the position of the first digit of your code.

At any time, you can select Move mode (where a colour can be moved from one position to another), or Delete mode (which clears your grid so you can begin again). When you are satisfied with your guess, pressing E will transmit it to the other computer. Your opponent then has a limited amount of time to enter his guess. If this time elapses, your opponent's incomplete guess will be scored.

The scoring works by indicating how many numbers are right and in the correct position, and how many are right but in the wrong position.

If your guess is right, then you have won. If neither of you gets the codes in ten guesses, the winner will be the player with the largest number of points. Make sure that you take it in turns to enter a guess since the network won't allow both players to send data simultaneously.

## Program Notes:

- 5 Send the program to the other Spectrum
- 45-999 Select the mode

1000-1040 Select a colour  
 1500-1640 Check the string being transmitted, if any.  
           Decrease time count.  
           Call time up  
 2000-2040 Enter a position  
 3000-3060 Move a colour from one position to another  
 4000-4010 Clear the grid  
 5000-5030 Enter mode – Send a header down the network  
           Transfer your colours to the right hand side of the screen  
 6000-6100 Score the entered codes  
 6500-6900 Print the final score  
           Wait for a key to be pressed to restart  
 7000-7499 Set up the screen and variables  
 7500-7600 Clear a portion of the screen  
           Reset the mode  
 8000-8200 Enter codes  
           Send and receive the codes  
 9000-9010 Set up the graphics  
 9020-9999 Instructions

```

1 REM * @ Paul Stanley *
5 CLS : PRINT "Type on other
Spectrum:''' " LOAD *"n";1": S
AVE *"n";1;"a" LINE 10
10 CLS : BORDER 6: PAPER 7: IN
K 0: CLS : POKE 23750,6: POKE 23
658,8
20 GO SUB 9000
30 GO SUB 7000
40 GO SUB 8000
45 IF mode=0 THEN GO TO 70
50 PRINT OVER 1; FLASH 1;AT m
ode*2+2,1;"
60 GO SUB mode*1000
70 GO SUB 1500
75 LET k$=INKEY$: IF k$="C" TH
EN GO SUB 7600: LET mode=1
80 IF k$="M" AND mode<>2 THEN
GO SUB 7600: LET mode=3
85 IF k$="D" AND mode<>2 THEN
GO SUB 7600: LET mode=4
90 IF k$="E" AND mode<>2 THEN
GO SUB 7600: LET mode=5
999 GO TO 45
  
```

```

1000 BEEP .1,10: PRINT AT 14,0: "
Enter colour: ";
1010 LET k$=INKEY$: IF k$<"0" OR
k$>"5" THEN GO SUB 1500: GO TO
1010
1020 LET hold=VAL k$: PRINT AT 1
4,13: PAPER hold: INK 9:hold:AT
20,2:hold
1030 GO SUB 7600: LET mode=2
1040 BEEP .2,0: GO SUB 7500: RET
URN
1500 IF c<>500 THEN GO TO 1600
1510 OPEN #4:"n":st1: LET i$=INK
EY$#4: CLOSE #4
1520 IF i$="" THEN RETURN
1530 FOR j=1 TO 5: BEEP .2,30: P
AUSE 10: NEXT j
1600 LET c=c-1: 'PRINT AT 0,28: F
LASH 1:c: FLASH 0:" ": IF c>0 TH
EN RETURN
1620 GO SUB 7500: GO SUB 7600: P
RINT AT 14,0: FLASH 1:"TIME UP!"
: FOR f=0 TO -30 STEP -3: BEEP .
2,f: NEXT f
1640 GO TO 5000
2000 BEEP .1,0: PRINT AT 14,0:"E
nter POSITION""of "; PAPER hold
: INK 9:hold: PAPER 7:". (1-4)"
2010 LET k$=INKEY$: IF k$<"1" OR
k$>"4" THEN GO SUB 1500: GO TO
2010
2020 PRINT AT 20,2:" "; PAPER ho
ld: INK 9:AT 20,VAL k$*2+3:hold
2030 LET q$(VAL k$)=STR$ hold: B
EEP .2,0: GO SUB 7500
2040 GO SUB 7600: LET mode=0: RE
TURN
3000 BEEP .1,10: IF q$=" " TH
EN GO SUB 7600: LET mode=0: RET
URN
3005 PRINT AT 14,0:"Type positio
ns:" INK 2:"FROM:"

```



```

3010 LET k$=INKEY$: IF k$<"1" OR
k$>"4" THEN GO SUB 1500: GO TO
3010
3015 IF q$(VAL k$)="" THEN GO
TO 3010
3020 BEEP .2,0: PRINT INK 3;AT
15,6;k$; INK 2;" TO : "
3030 LET l$=INKEY$: IF l$<"1" OR
l$>"4" THEN GO SUB 1500: GO TO
3030
3040 BEEP .2,0: PRINT AT 16,6; I
NK 3;l$
3050 LET l=VAL q$(VAL k$): PRINT
AT 20,VAL k$*2+3;" ";AT 20,VAL
l$*2+3; PAPER 1; INK 9;l
3060 LET q$(VAL k$)="" : LET q$(
VAL l$)=STR$ l: GO SUB 7600: GO
SUB 7500: LET mode=0: RETURN
4000 BEEP .1,10: FOR f=1 TO 4: P
RINT AT 20,f*2+3;" ": LET q$(f)=
" ": NEXT f
4010 GO SUB 7600: LET mode=0: RE
TURN
4500 IF l$="" THEN LET p=7: RE
TURN
4510 LET p=VAL l$: RETURN
5000 BEEP .1,10: IF c<500 THEN
FOR f=1 TO 4: LET l$=SCREEN$ (20
,f*2+3): GO SUB 4500: PRINT AT 2
0,f*2+3;" ";AT y,f*2+19; PAPER p
; INK 9;l$: NEXT f: OPEN #4;"n";
st1: PRINT #4;"a": CLOSE #4: GO
TO 6000
5002 FOR f=1 TO 4: IF q$(f)=""
THEN GO SUB 7600: LET mode=0: R
ETURN
5005 NEXT f: FOR f=1 TO 4: LET l
=VAL (q$(f)): PRINT AT 20,f*2+3;
" ";AT y,f*2+19; PAPER 1; INK 9;
l: NEXT f
5010 PRINT AT 14,0;"OK so you've
""finished. Your""opponent has
""limited time to""finish. ";
FLASH 1;" WAIT "

```

```

5020 OPEN #4;"n";st1: PRINT #4;"
A": CLOSE #4
5030 OPEN #4;"n";st1: INPUT #4;q
$: CLOSE #4
6000 LET q$=s$: LET r=0: LET er=
r: FOR f=1 TO 4
6010 IF q$(f)=q$(f) THEN LET q$(
f)="x": LET q$(f)="o": LET er=e
r+1
6015 NEXT f
6020 FOR f=1 TO 4: FOR q=1 TO 4
6030 IF q$(f)=q$(q) THEN LET q$(
q)="x": LET r=r+1: NEXT f: GO T
O 6050
6040 NEXT q: NEXT f
6050 LET q$="      ": PRINT AT y,1
6;r;AT y,18;er
6060 LET sc1=sc1+5*er+r
6070 LET b$=s$: LET c$=STR$ er+S
TR$ r: GO SUB 8000+100*st: GO SU
B 8000+100*st1
6080 LET l=VAL s$(1): LET sc2=sc
2+5*l+VAL s$(2)
6090 LET c=500: LET s$=b$: LET y
=y-2: PRINT AT 0,28;"      ";AT 1,8
; INK 2;sc1;"-";sc2
6100 IF er<4 AND l<4 AND y>0 THE
N GO SUB 7500: GO SUB 7600: LET
mode=1: RETURN
6500 FOR f=1 TO 30: BEEP .1,f: N
EXT f: CLS
6510 PRINT AT 10,0;: IF er=4 AND
l=4 THEN PRINT "It is a draw!
You both correctly quessed the hi
dden codes.": GO TO 6900
6520 IF er=4 THEN PRINT "Well d
one! You correctly quessed the hi
dden code ";s$: GO TO 6900
6530 IF l=4 THEN PRINT "Bad luc
k! Your opponent quessed your co
de. You should have tried";s$: G
O TO 6900

```

```

6540 IF sc1=sc2 THEN PRINT "It
is a draw! You both scored the
same.": GO TO 6900
6550 IF sc1>sc2 THEN PRINT "Wel
l done! You won because you hav
e a higher score than your opp
onent.": GO TO 6900
6560 PRINT "Bad luck! You have l
ost because your opponent has a
higher score"
6900 PRINT "'Press any key to p
lay again..": PAUSE 0: CLS : GO
TO 20
7000 FOR y=6 TO 156 STEP 16: FOR
x=166 TO 228 STEP 16
7010 INK 2: PLOT x,y: DRAW 11,0:
DRAW 0,11: DRAW -11,0: DRAW 0,-
11
7020 NEXT x: NEXT y: FOR x=38 TO
91 STEP 16: PLOT x,6: DRAW 11,0
: DRAW 0,11: DRAW -11,0: DRAW 0,
-11: NEXT x
7030 INK 4: FOR x=124 TO 156 STE
P 16: PLOT x,3: DRAW 0,162: NEXT
x: DRAW -31,0: PLOT 124,3: DRAW
31,0
7040 INK 1: PRINT AT 0,2;"CORREC
T COLOURS B SPOT ON!"
7050 INK 3: FOR f=1 TO 10: PRINT
AT 22-2*f,29;f: NEXT f
7060 INK 0: PRINT AT 4,1;"Colour
.....C"" Position....P"" Mo
ve.....M"" Delete.....D""
" Enter Move..E"; PAPER 6; BRIGH
T 1; AT 3,5;"MODES"
7070 INK 5: PLOT 5,70: DRAW 109,
0: DRAW 0,83: DRAW -109,0: DRAW
0,-83
7080 INK 0
7090 DIM q$(4): LET a$="
"
7100 LET sc1=0: LET sc2=0: LET y
=20: LET mode=1: LET c=500

```

```

7200 PRINT INK 2;AT 1,0;"SCORES
: ";sc1;"-";sc2
7499 RETURN
7500 FOR f=14 TO 18: PRINT AT f,
0;a$: BEEP .01,f: NEXT f: RETURN

7600 PRINT OVER 1;AT mode*2+2,1
;" " AND mode>0: RET
URN
8000 PRINT AT 14,0;"Enter your c
ode""using 4 numbers""0-5: ";
8010 DIM c$(4): FOR f=1 TO 4
8020 LET k$=INKEY$: IF k$<"0" OR
k$>"5" THEN GO TO 8020
8030 LET c$(f)=k$: PRINT " "; PA
PER VAL k$: INK 9;k$;: BEEP .2,0
: NEXT f
8040 GO SUB 8000+100*st
8050 GO SUB 8000+100*st1
8060 GO SUB 7500: RETURN
8100 OPEN #4;"n";st1: PRINT #4;c
$: CLOSE #4: RETURN
8200 OPEN #4;"n";st1: INPUT #4;s
$: CLOSE #4: RETURN
9000 RESTORE : FOR f=USR "a" TO
USR "b"+7: READ q: POKE f,q: NEX
T f
9010 DATA 0,112,8,4,4,21,14,4,0,
14,16,32,32,168,112,32
9020 PRINT PAPER 5;" CODE BREAK
ER - PAUL STANLEY '84"
9030 PRINT '"This is a network q
ame for two players""The objec
t is to find out your opponent'
s 4 figure code."
9040 PRINT '"After each guess, y
ou will be told how many numbe
rs are right and how many are in
the correct place as well."
9050 PRINT '"When you enter your
guess, your opponent then has l
imited time to enter his guess.
"

```

```
9060 PRINT "You are allowed 10
guesses. If the codes are not g
uessed the winner will be the
one with the highest score."
9500 INPUT "Enter station No (1
or 2) ";st: IF st<>1 AND st<>2 T
HEN GO TO 9500
9510 FORMAT "n";st: LET st1=1+(s
t=1)
9999 CLS : RETURN
```

# 6 Consequences

## Hardware requirements

Two/sixty-four 16/48K Spectrums on ZX network

One television for each Spectrum

This program is fairly simple, but illustrates just how effectively data can be transferred over a local area network. As you can see from the hardware specification, Consequences allows you to stretch the network to the limit – you can use it with up to sixty-four Spectrums!

Each player fills in the missing blanks to a short story. After each input, the story so far is passed on to a neighbouring Spectrum. This means that after several inputs your original entry is several times removed from the initial version (and computer). If less than seven Spectrums are on the network, then more than one of your lines will be in the completed story.

The story is this:

(boy's name)                      MET (girl's name)                      AT (a place)  
HE SAID.....                      SHE SAID.....  
THEY (what they did) AND THE CONSEQUENCE WAS... (consequence)

When transferring the program across the network, ENTER must be pressed following LOAD\*“n”;0 before the program is broadcast. This is because the computer initially holding the program will not wait for all the others to be ready.

The finished stories are compilations of several others, and the results can be very amusing.

## Program Notes:

20–25 Set up border, paper and ink colours

30–50 Call the various routines

Count the number of lines entered

100-130 Transmit story lines  
     110 Send if even station number, receive if odd  
     120 Receive if even station number, send if odd  
 500-580 Print the finished story  
     Wait for a key press  
 600-615 Transmit and receive a storyline  
 1000-1080 Type in a line on to the screen  
 1500-1520 Delete a character  
 2000-2020 Print a character of the storyline  
 8000-8040 Set up the stations  
 8500-8580 Instructions  
 9000-9010 Data for the screen colours and prompts  
 9500-9800 Broadcast the program to all computers on the  
     network which are ready to receive it

```

1 REM * @ Paul Stanley 1984 *
5 GO SUB 9500
10 CLS : BORDER 7: PAPER 7: IN
K 0: CLS : GO SUB 8000: RESTORE
9000: LET 1=1: DIM a$(7,105): DI
M 1(7)
20 CLS : READ b,p,i,q$: BORDER
b: POKE 23750,b: PAPER p: INK i
: CLS : PRINT INK 9;AT 4,0;q$
25 PRINT AT 0,9; PAPER 2; INK
6; FLASH 1;"CONSEQUENCES!"
30 GO SUB 1000
35 PRINT AT 15,0; FLASH 1;"PLE
ASE WAIT": GO SUB 100
40 LET 1=1+1: IF 1=8 THEN GO
TO 500
50 GO TO 20
100 FOR f=1 TO 1
105 LET b$a$(f): LET b=1(f)
110 GO SUB 600+10*(st/2-INT (st
/2)=0)
120 GO SUB 600+10*(st/2-INT (st
/2)<>0)
130 NEXT f: RETURN
500 CLS : BORDER 5: PAPER 7: IN
K 0: CLS

```

```

510 PRINT "Are you ready for the completed story?": FOR f=1 TO 20: BEEP .1,f: NEXT f: PRINT "Here it comes...": FOR f=20 TO 1 STEP -1: BEEP .1,f: NEXT f
520 CLS: LET l=1: GO SUB 2000: PRINT INK 2;"met ";: LET l=2: GO SUB 2000
530 PRINT INK 2;"at": LET l=3: GO SUB 2000:
540 PRINT INK 2;"They...": LET l=4: GO SUB 2000
550 PRINT INK 2;"He said:-": LET l=5: GO SUB 2000
560 PRINT INK 2;"She said:-": LET l=6: GO SUB 2000
570 PRINT INK 2;"And the consequence was...": LET l=7: GO SUB 2000
580 INPUT "Press ENTER to play again";f$: CLS: GO TO 10
600 OPEN #4;"n";st1: PRINT #4;b$: CLOSE #4
605 OPEN #4;"n";st1: PRINT #4;b$: CLOSE #4: RETURN
610 OPEN #4;"n";st2: INPUT #4;a$(f): CLOSE #4
615 OPEN #4;"n";st2: INPUT #4;l(f): CLOSE #4: RETURN
1000 LET a=1: LET y=6: LET x=0
1010 LET k$=INKEY$: IF k$="" THEN PRINT AT y,x;" ": GO TO 1010
1020 IF CODE k$=12 AND a>1 THEN GO SUB 1500: GO TO 1010
1030 IF CODE k$=13 THEN BEEP .4,30: LET l(1)=a: RETURN
1040 PRINT AT y,x;k$: LET a$(1,a)=k$
1050 LET x=x+1: IF x=32 THEN LET x=0: LET y=y+2
1060 LET a=a+1: IF a=101 THEN BEEP .4,30: LET l(1)=a: RETURN
1070 IF a=90 THEN BEEP .4,20
1080 BEEP .05,20: GO TO 1010

```



```

1500 PRINT AT y,x;" ": LET x=x-1
: IF x<0 THEN LET y=y-2: LET x=
31
1510 PRINT AT y,x;" "
1520 BEEP .06,20: LET a=a-1: LET
a$(1,a)=" ": RETURN
2000 FOR f=1 TO 1(1)
2010 PRINT a$(1,f);
2020 NEXT f: RETURN
8000 INPUT "How many Spectrums a
re on the network? ";no
8010 IF no<2 OR no>64 THEN GO T
O 8000
8020 INPUT "What station is this
(make sure no 2 computers have
the same station number) ? "
;st
8025 FORMAT "n";st
8030 LET st1=st+1: IF st=no THEN
LET st1=1
8040 LET st2=st-1: IF st=1 THEN
LET st2=no
8500 PRINT FLASH 1; INK 3; PAPE
R 6;"CONSEQUENCES! BY P.STANLEY
1984."
8510 PRINT "This is a network g
ame, which aswell as being fun t
o play, is good typing practic
e."
8520 PRINT "It demonstrates jus
t how effect-ively information c
an be passed on the network."
8530 PRINT "The idea of the gam
e is that several pieces of i
nformation are asked for, and
after each, the information is
passed to the adjacent computer."
8540 PRINT "This means that if
quite a few computers are used,
then the information you fir
st typed in is several computer
s away!"
8550 PRINT "Press any key to se
e more.": PAUSE 0: CLS

```

```

8560 PRINT "The object is that
you tell the computer a story. A
fter each bit of information, the
story up to that point is passe
d on to the next computer."
8570 PRINT "This means that the
resulting story is a mixture
of up to 7 different ones. The
results can be very amusing!!"
8580 PRINT "Press any key to st
art..": PAUSE 0: CLS : RETURN
9000 DATA 5,4,7,"<boy's name>",1
,3,6,"MET... <girl's name>",6
,1,4,"AT... <place>"
9010 DATA 2,5,0,"THEY.... <ac
tion>",4,0,5,"HE SAID....",3,6,2
,"SHE SAID....",7,2,6,"AND THE C
ONSEQUENCE WAS...."
9500 CLS : PRINT "Type on all c
omputers on the network:-'"'"
LOAD *"n";0 <ENTER>": PRINT
'"When you are sure this has b
een done, press any key."
9600 IF INKEY$="" THEN GO TO 96
00
9700 SAVE *"n";0;"a" LINE 10
9800 RETURN

```

# 7 4-A-Side Soccer

## **Hardware requirements**

Two 16/48K Spectrums on ZX network

One television

This program should be loaded into the computer that is connected to the only television set required by this game. The other computer only uses one line of the program and merely acts as a joystick.

This is football which must be played at the highest level! There is no dribbling allowed, no tackling and no crowd trouble! The ball can only be passed, and the game should be approached with steely tactics, boundless enthusiasm and nimble fingers!

You can only move one player at a time, and this rugged hero is selected by entering a value between 1–4. If one of your players has the ball, he can't dribble it, but can retain it for a time, during which you can move another player to a position to which the ball can be passed. However, when the time limit runs out the ball is automatically booted away in the direction it was last kicked.

The winning team is the first to score five goals.

## **Program Notes:**

- 9–10 These two lines are only used by the remote computer and transmit which key is being pressed
- 100–140 Call the various routines  
Increase the count which prevents players from retaining the ball indefinitely
- 400–495 Read the keyboard  
Move and print the players  
Check to see if the player reaches the ball
- 500–620 Receive the key pressed on the remote computer  
Move and print the players  
Check to see if the player reaches the ball

1000-1090 Move the ball  
 1500-1530 Check to see if the ball hits a player  
 4000-4030 Set the ball colour to the colour of the team  
 5000-5080 Decide whether to award a goal or a goal kick  
 5500-5540 The winner is named  
         Wait for a key to be pressed  
 8000-8999 Set up the screen and the players  
 9000-9080 Define the graphics characters  
         9090 Data for the machine code routine  
 9100-9110 Data for the player positions  
 9500-9590 Instructions  
         9900 Send the program to the other computer

```

1 REM * @ Paul Stanley *
5 GO SUB 9000: GO TO 20
9 POKE 23658,8: FORMAT "N";2
10 OPEN #4;"N";1: PRINT #4;INK
EY$: CLOSE #4: GO TO 10
20 CLS : POKE 23658,8: POKE 23
750,1: BORDER 1: PAPER 7: INK 3:
CLS
30 LET sc1=0: LET sc2=0
40 GO SUB 8000
50 FORMAT "N";1
100 LET q=1: LET k$=INKEY$
110 GO SUB 500
120 LET q=2: OPEN #4;"n";2: LET
k$=INKEY$#4: CLOSE #4
130 GO SUB 400
135 IF c>0 THEN LET c=c+1: IF
c=40 THEN GO SUB 1000
140 GO TO 100
200 POKE 23692,200: OPEN #4;"n"
;1: PRINT INKEY$#4: CLOSE #4: GO
TO 200
400 LET e=p2+4: IF CODE k$=13 T
HEN RETURN
410 IF k$>"0" AND k$<"5" THEN
LET p2=VAL k$: BEEP .03,30: PRIN
T OVER 0; INK 4: FLASH 1;AT 1,3
1;p2: RETURN
415 IF k$="S" AND B=P2+4 THEN
GO TO 1000

```

```

420 LET ey=(k$="C")+(k$="X")+(k$="Z")-(k$="Q")-(k$="W")-(k$="E")
)
430 LET ex=(k$="E")+(k$="D")+(k$="C")-(k$="Q")-(k$="A")-(k$="Z")
)
450 IF b=e THEN LET by=ey: LET bx=ex: RETURN
460 PRINT AT y(e),x(e);a$(e);AT y(e)+1,x(e);b$(e)
470 LET o=y(e)+ey: LET p=x(e)+ex
X
480 IF ATTR(o,p)=57 OR ATTR(o,p+1)=57 OR ATTR(o+1,p)=57 OR ATTR(o+1,p+1)=57 THEN GO TO 400
0
490 IF o>=0 AND o<22 AND p<29 AND p>1 AND ATTR(o,p)=59 AND ATTR(o,p+1)=59 AND ATTR(o+1,p)=59 AND ATTR(o+1,p+1)=59 THEN LET y(e)=o: LET x(e)=p
495 PRINT AT y(e),x(e); INK 4;a$(e);AT y(e)+1,x(e);b$(e): RETURN
N
500 IF k$="" THEN RETURN
510 IF k$>"0" AND k$<"5" THEN LET p1=VAL k$: BEEP .03,30: PRINT OVER 0; INK 2; FLASH 1;AT 1,0;p1: RETURN
515 IF K$="S" AND B=P1 THEN GO TO 1000
520 LET dy=(k$="C")+(k$="X")+(k$="Z")-(k$="Q")-(k$="W")-(k$="E")
)
530 LET dx=(k$="E")+(k$="D")+(k$="C")-(k$="Q")-(k$="A")-(k$="Z")
)
550 IF b=p1 THEN LET by=dy: LET bx=dx: RETURN
560 PRINT AT y(p1),x(p1);a$(p1);AT y(p1)+1,x(p1);b$(p1)
570 LET o=y(p1)+dy: LET p=x(p1)+dx

```

```

580 IF ATTR (o,p)=57 OR ATTR (o
,p+1)=57 OR ATTR (o+1,p)=57 OR A
TTR (o+1,p+1)=57 THEN GO TO 400
0
590 IF o>=0 AND o<22 AND p<29 A
ND p>1 AND ATTR (o,p)=59 AND ATT
R (o,p+1)=59 AND ATTR (o+1,p)=59
AND ATTR (o+1,p+1)=59 THEN LET
y(p1)=o: LET x(p1)=p
620 PRINT AT y(p1),x(p1); INK 2
;a$(p1);AT y(p1)+1,x(p1);b$(p1):
RETURN
1000 LET c=0: BEEP .1,0: LET B=0
: FOR f=1 TO 7
1010 LET y1=y: LET x1=x
1030 LET y=y+by: LET x=x+bx
1040 IF y=21 OR y=0 THEN LET by
=-by
1070 IF x=0 OR x=31 THEN GO TO
5000
1080 IF ATTR (y,x)<>59 THEN GO
TO 1500
1090 PRINT AT y1,x1;"0"; INK 1;A
T y,x;"0": NEXT f: RETURN
1500 LET c=1: PRINT AT y1,x1;"0"
1510 FOR f=1 TO 8
1520 IF (y=y(f) OR y=y(f)+1) AND
(x=x(f) OR x=x(f)+1) THEN LET
b=f: LET y=y(f)+1: LET x=x(f)-1+
3*(f<5): PRINT INK 2+2*(f>4);AT
y,x;"0": BEEP .03,5: RETURN
1530 NEXT f
4000 LET c=1: LET b=(p1 AND q=1)
+(p2+4 AND q=2): PRINT AT y,x;"0
"
4010 LET i=2+2*(b>4): PRINT INK
i;AT y(b),x(b);a$(b);AT y(b)+1,
x(b);b$(b): BEEP .03,20
4020 LET y=y(b)+1: LET x=x(b)-1+
3*(b<5): PRINT AT y,x; INK i;"0"
4030 RETURN
5000 PRINT AT y1,x1;"0";AT Y,X;
INK 1;"0"

```

```

5010 IF y>8 AND y<13 THEN FOR f
=1 TO 5: RANDOMIZE USR USR "p":
NEXT f: GO TO 5500
5030 FOR f=1 TO 15: BEEP .1,15+R
ND*5: NEXT f
5040 LET bx=1: IF x=31 THEN LET
bx=-1
5050 LET by=INT (RND*3)-1
5060 IF y=21 THEN LET by=-INT (
RND*2)
5070 IF y=0 THEN LET by=INT (RN
D*2)
5080 GO TO 1000
5500 IF x=31 THEN LET sc1=sc1+1
: IF sc1=5 THEN CLS : PRINT IN
K 0; FLASH 1;AT 10,3;"THE RED TE
AM IS VICTORIOUS": GO TO 5540
5510 IF x=0 THEN LET sc2=sc2+1:
IF sc2=5 THEN CLS : PRINT INK
0; FLASH 1;AT 10,2;"THE GREEN T
EAM IS VICTORIOUS": GO TO 5540
5520 FOR f=1 TO 15: BEEP .1,15+R
ND*5: NEXT f
5530 GO TO 40
5540 PRINT AT 20,2;"Press any ke
y to play again.": PAUSE 0: CLS
: GO TO 20
8000 OVER 0: CLS : PLOT 8,3: DRA
W 239,0: DRAW 0,169: DRAW -239,0
: DRAW 0,-169
8005 PRINT AT 0,13; INVERSE 1;sc
1;" - ";sc2
8010 PLOT 127,3: DRAW 0,169: CIR
CLE 127,87,25
8020 PLOT 8,40: DRAW 40,0: DRAW
0,95: DRAW -40,0
8030 PLOT 247,40: DRAW -40,0: DR
AW 0,95: DRAW 40,0
8040 PLOT 8,64: DRAW 12,0: DRAW
0,47: DRAW -12,0
8050 PLOT 247,64: DRAW -12,0: DR
AW 0,47: DRAW 12,0

```

```

8060 FOR f=9 TO 12: PRINT AT f,0
; PAPER 4;" ";AT f,31;" ": NEXT
f
8100 OVER 1: DIM b$(8,2): DIM a$(
8,2): DIM y(8): DIM x(8)
8110 RESTORE 9100: FOR f=1 TO 8
8120 LET a$(f)=("AB" AND f>4)+("
HI" AND f<5)
8130 READ y(f),x(f),b$(f)
8140 PRINT INK 2+2*(f>4);AT y(f
),x(f);a$(f);AT y(f)+1,x(f);b$(f
)
8150 NEXT f
8160 LET p1=4: LET p2=p1
8170 RANDOMIZE : LET y=11: LET x
=15+INT (RND*2): PRINT AT y,x; I
NK 1;"O"
8180 LET by=1: LET c=0: LET bx=c
:: LET b=c
8190 PRINT FLASH 1;AT 1,0; INK
2;p1; INK 4;AT 1,31;p2
8999 RETURN
9000 RESTORE : FOR f=USR "a" TO
USR "q"+5: READ q: POKE f,q: NEX
T f
9010 DATA 3,6,12,7,3,3,15,119,12
8,192,192,128,0,0,128,240
9020 DATA 7,7,7,15,25,16,8,24,13
6,128,128,2,6,130,66,130
9030 DATA 136,128,128,7,1,135,68
,135,136,128,128,7,1,135,65,135
9040 DATA 136,128,128,1,3,133,79
,129,1,3,3,1,0,0,1,15
9050 DATA 192,96,48,224,192,192,
240,238,17,1,1,32,96,33,34,33
9060 DATA 17,1,1,224,32,225,130,
225,17,1,1,224,32,225,34,225
9070 DATA 17,1,1,16,48,81,242,17
,224,224,224,240,152,8,16,24
9080 DATA 60,126,255,253,249,243
,126,60
9085 REM # Care when entering th
e next line, it contains machine
code data #

```



```

9090 DATA 6,255,197,6,255,120,21
1,254,16,251,193,16,245,201
9100 DATA 10,3,"JN",1,12,"KN",19
,12,"LN",10,9,"MN"
9110 DATA 10,27,"CD",1,18,"CE",1
9,18,"CF",10,21,"CG"
9500 PRINT INK 0; PAPER 5;" 4-A
-SIDE SOCCER @ P.STANLEY "
9510 PRINT INK 1;"This is a gam
e which uses the ZXnetwork, but
in an unusual way. This computer
holds the program,and the other
is just used as a glorified joy
stick."
9520 PRINT INK 2;"The player yo
u want to move is shown by the
number in the top corners of th
e pitch. This can be changed by
pressing the corr-responding num
ber."
9530 PRINT INK 0;"If your playe
r has the ball, he is not allowe
d to dribble with it but can ki
ck it, by selecting that player,
pressing a dirn button, and f
inally S."
9540 INPUT "Press ENTER to read
more."; LINE q$
9550 CLS : PRINT INK 2;"The win
ning team is the first to score 5
goals."
9560 PRINT INK 1;"The direction
keys are logically laid out and
should therefore present no pr
oblems:"
9570 PRINT ' INK 2;" Q W E"
'" A D"" Z X C"
9580 PRINT INK 0;"These move yo
ur man (or aim his kick if he ha
s the ball) in the direction in
which the key aims."

```

```
9585 PRINT INK 1;"If your playe  
r has the ball, he could hold on  
to it forever. Therefore to  
stop this, after a certain time,  
the ball will be kicked away."  
9590 INPUT "Press ENTER to start  
."; LINE q$  
9900 CLS : PRINT "Type on other  
spectrum: ""LOAD *""n";1 <ENT  
ER>": SAVE *""n";1;"a" LINE 9  
9910 RETURN
```

# 8 Split Tennis

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

Do you remember, years ago, when the tennis game appeared in pubs and cafes? Arcade games have come a long way since then, but for some reason tennis still has its appeal, and this novel variation on the old favourite is fiendishly addictive!

The new feature of our implementation is that each player only sees half of the court on their screen, whilst their opponent sees the other half. When the ball is hit out of sight on one screen, it will then appear on the other player's screen.

You can move your bat in eight directions and all you need to do is prevent the ball from hitting the back wall (the wall with the score printed on it). The sooner you return the ball after it appears on the screen, the more points you will get. Should you miss with your bat, you lose 100 points.

This is a simple, but compelling game, which should give hours of amusement and only take a short time to type in.

## Program Notes:

- 10 Send the program to the other Spectrum
- 150–210 Check for a keypress  
Move the bar and print it
- 220 If the ball is not on your screen, check to see if a header has been sent by the other computer
- 240–299 Move and print the ball  
Check to see if the ball hits a wall or a bat
- 1000–1040 Receive position and directions of the ball from the other Spectrum
- 1100–1120 Increase score and print it when opponent misses ball

2000-2050 Transmit position and direction of the ball  
 Increase and print score  
 3000-3030 Send header to tell your opponent that you missed  
 Increase and print opponent's score  
 5000-5050 Check the scores to see if they are 1000  
 Wait for a keypress  
 8000-8999 Set up the screen and graphics  
 9000-9070 Instructions

```

1 REM * @ Paul Stanley *
10 CLS : PRINT "Type on other
Spectrum:''' LOAD *"n";1
<ENTER>": SAVE *"n";1;"a" LINE 5
0
50 CLS
60 POKE 23750,4: PAPER 2: INK
7: BORDER 4: CLS
70 GO SUB 9000: CLS
100 INPUT "Station No. (1 or 2)
? ";st: IF st<>1 AND st<>2 THEN
GO TO 100
110 FORMAT "n";st: LET st1=1+(s
t=1)
120 GO SUB 8000
150 LET co=co-1
200 LET x1=x: LET y1=y: LET x=x
+(2 AND (IN 65278=253 OR IN 6527
8=189) AND x<28)-(2 AND (IN 6527
8=254 OR IN 65278=190) AND x>1)
205 LET y=y+((IN 32766=251 OR I
N 32766=187) AND y<19)-((IN 4915
0=251 OR IN 49150=187) AND y>2)
210 PRINT AT y1,x1;" "; INK 6;
AT y,x;"pp"
220 IF scr<>st THEN OPEN #4;"n
";st1: LET k$=INKEY$#4: CLOSE #4
: IF k$<>" " THEN GO SUB 1000
230 IF scr<>st THEN GO TO 200
240 LET y1=by: LET x1=bx
250 LET by=by+dy: LET bx=bx+dx
260 IF bx=1 OR bx=30 THEN LET
dx=-dx
265 IF (by=0 AND st=1) OR (by=2
1 AND st=2) THEN GO SUB 3000

```

```

270 IF ATTR (by,bx)=22 THEN LE
T dy=-dy: BEEP .03,10: LET by=by
+2*dy
280 PRINT AT y1,x1;" ";AT by,bx
;"A"
290 IF by=0 OR by=21 THEN GO T
O 2000
299 GO TO 150
1000 IF k$="M" THEN GO TO 1100
1005 LET co=60: LET scr=st
1010 OPEN #4;"n";st1: INPUT #4;b
x: INPUT #4;dx: INPUT #4;s(st1):
CLOSE #4
1015 PRINT AT 21*(st=2),8;"SCORE
: ";s(st);" - ";s(st1)
1017 GO SUB 5000
1020 LET dy=1-2*(st=1)
1030 PRINT AT by,bx;"A"
1040 RETURN
1100 FOR f=0 TO 30 STEP 5: BEEP
.2,f: NEXT f
1110 LET s(st)=s(st)+100: PRINT
AT 21*(st=2),8;"SCORE: ";s(st);"
- ";s(st1)
1115 GO SUB 5000
1120 RETURN
2000 OPEN #4;"n";st1: PRINT #4;"
X": CLOSE #4
2005 LET s(st)=s(st)+co*(co>0)
2010 OPEN #4;"n";st1: PRINT #4;b
x: PRINT #4;dx: PRINT #4;s(st):
CLOSE #4
2020 PRINT AT by,bx;" "
2025 LET scr=st1
2040 PRINT AT 21*(st=2),8;"SCORE
: ";s(st);" - ";s(st1)
2045 GO SUB 5000
2050 GO TO 150
3000 OPEN #4;"n";st1: PRINT #4;"
M": CLOSE #4
3010 FOR f=0 TO -30 STEP -5: BEE
P .2,f: NEXT f

```

```

3020 LET s(st1)=s(st1)+100: PRIN
T AT 21*(st=2),8;"SCORE: ";s(st)
;" - ";s(st1)
3025 GO SUB 5000
3030 LET dy=-dy: LET by=by+3*dy:
RETURN
5000 IF s(1)<1000 AND s(2)<1000
THEN RETURN
5010 FOR f=1 TO 30: BEEP .03,f:
BEEP .03,30-f: NEXT f
5020 IF s(st)>1000 THEN PRINT A
T 10,1; FLASH 1; INK 4; PAPER 2;
BRIGHT 1;"WELL DONE ! YOU ARE T
HE WINNER"
5030 IF s(st1)>1000 THEN PRINT
AT 10,1; FLASH 1; INK 4; PAPER 2
; BRIGHT 1;"BAD LUCK! YOU LOST T
HIS TIME !"
5040 INPUT "PRESS ENTER TO PLAY
AGAIN"; LINE a$
5050 GO TO 120
8000 CLS : FOR f=0 TO 21: PRINT
AT f,0; PAPER 1; INK 5;"f";AT f,
31;"k": NEXT f
8030 PRINT AT (21 AND st=2),11;"
"
8050 LET x=13: LET y=2+(17 AND s
t=2)
8060 LET by=3: LET bx=15: LET dy
=1: LET dx=1-2*(RND>.5)
8070 RESTORE : FOR f=0 TO 7: REA
D q: POKE USR "a"+f,q: NEXT f
8080 DATA 60,126,255,255,255,255
,126,60
8090 LET scr=1: LET co=scr
8100 DIM s(2)
8110 PRINT AT 21*(st=2),8;"SCORE
: ";s(st);" - ";s(st1)
8120 BEEP .3,20
8999 RETURN
9000 PRINT PAPER 3; INK 6; FLAS
H 1;" SPLIT TENNIS @ Paul Stan
ley "

```

```
9020 PRINT "Remember that tenni
s game which was around long bef
ore SPACE INVADERS? Well in t
hat version you could see the w
hole court. Not any more! You c
an now only see your half of th
e court."
9030 PRINT "You must prevent th
e ball from hitting the back wa
ll, otherwise you lose 100 points
."
9040 PRINT "If you hit the ball
soon after it crosses the net
you will get bonus points. There
fore, as the winner is the first
to reach 1000 points, you wo
uld do well to be close to the
net."
9050 INPUT "PRESS ENTER TO SEE C
ONTROLS."; LINE A$
9060 CLS : PRINT "The controls
are: "" Caps-LEFT","Z-RIGHT""
      K-UP","M-DOWN"
9070 PRINT ""Press any key to s
tart..": PAUSE 0: CLS : RETURN
```

# 9 Dominoes

## Hardware requirements

Two 48K Spectrums on ZX network

Two televisions

This is an excellent re-vamp of that old pub favourite, Fives and Threes. The computer holds the full set of twenty-eight dominoes. It deals seven to each player, and then goes on to place the first domino.

When it comes to your turn to place a domino, you can choose any which has the same number of spots as either end of the dominoes already placed. To calculate your score you should add up the two ends of the dominoes placed, and divide the result by both three and five. If it is exactly divisible by three, then you score the number of times three divides into it. You do this for five as well. If a double is placed, then the domino will Flash and be displayed in purple. Both sides of the double add to the total (for example, a double three would count as six). The highest score you can get is eight, if the total number of spots is fifteen.

If you place all seven dominoes, you get five bonus points and start first in the next round. If neither player can place a domino, then the computer adds up the total number of spots remaining for each player, and the player with the lowest receives five points and starts first.

At the beginning of the game you can either choose a long or short game, and between each round the computer 'shuffles' the dominoes. The computer only displays the end dominoes to prevent cluttering the screen.

This is a very flashily presented program, which could be persuaded to fit into 16K if some of its refinements were removed.

## Program Notes:

- 10 Send the program to the other Spectrum
- 40-50 Set up the stations
- 60-80 Input short or long game
- 95-140 Ensure one computer transmits while the other waits



1000-1130 Enter the domino number and check validity  
           Check to see if the player passed  
       1200 Transmit the ends of the domino  
 1210-1280 Increase the score  
 1900-1920 Work out which end to place the domino  
 2000-2190 Receive the ends of the domino and opponent's score  
 2700-2740 When all dominoes have been place, calculate the winner  
 3000-3130 Print the dominoes  
 4000-4080 Count the number of spots  
           Transmit and receive these numbers, then compare them  
 5500-5520 Transmit and receive option short or long game  
           Check that these are both the same  
 7000-7030 Calculate the score  
 7500-7550 Check that a player has finished the match  
           Wait for a keypress to restart  
 8000-8120 Generate, shuffle and deal dominoes  
 8130-8160 Station one sends station two's dominoes across the network  
 8170-8200 Print the seven dominoes  
 8250-8600 Set up the screen and variables  
 8700-8800 Print the score increment  
 9000-9050 Define the graphics characters  
 9500-9999 Instructions

1 REM Program Software and Design

2 REM @ Paul Stanley 1984

10 CLS : PRINT "Type on other  
Spectrum:-"" LOAD \*""n"";1  
<enter>": SAVE \*""n"";1;"a" LINE 2  
0

20 CLS : BORDER 7: PAPER 7: IN  
K 0: POKE 23750,7: CLS

30 GO SUB 9000

40 INPUT "Station No? (1 or 2)  
";st: IF st<>1 AND st<>2 THEN  
GO TO 40

50 FORMAT "n";st: LET st1=1+(s  
t=1)

60 LET step=2: PRINT "Press L  
for long game (to 120) or S fo  
r short game (to 80):";

70 LET k\$=INKEY\$: IF k\$<>"S" A  
ND k\$<>"s" AND k\$<>"L" AND k\$<>"  
I" THEN GO TO 70

```

75 PRINT k$: IF k$="S" OR k$="
s" THEN LET step=3
77 BEEP .1,10: GO SUB 5500
80 IF step<>q THEN PRINT "'Y
ou and your opponent want to p
lay different games. When you h
ave agreed on what length, try a
gain!!": PAUSE 100: GO TO 60
85 GO SUB 8000
90 GO SUB 8050
95 IF start=2 THEN GO TO 120
100 GO SUB 1000*st
105 GO SUB 2500
110 GO SUB 2700
120 GO SUB 1000*st1
125 GO SUB 2500
130 GO SUB 2700
140 GO TO 100
999 REM Input a move
1000 PRINT AT 7,0;"Which domino
do you want to place (1-7 or
0 if you cannot place one) :
";
1005 LET o1=e1: LET o2=e2: LET t
=o: LET d1=o: LET e=o
1010 LET k$=INKEY$: IF k$<"0" OR
k$>"7" THEN GO TO 1010
1020 IF k$="0" THEN BEEP .1,10:
LET t=1: LET kn=kn+1: GO TO 120
0
1030 LET k=VAL k$
1040 IF d(k,1)=9 THEN BEEP .3,-
10: GO TO 1010
1050 IF d(k,1)<>e1 AND d(k,1)<>e
2 AND d(k,2)<>e1 AND d(k,2)<>e2
THEN BEEP .3,-10: GO TO 1010
1060 LET kn=o: LET dom1=dom1-1:
BEEP .1,10: LET c=o: IF d(k,1)=e
1 OR d(k,2)=e1 THEN LET e=1: LE
T c=c+1
1070 IF d(k,1)=e2 OR d(k,2)=e2 T
HEN LET e=2: LET c=c+1

```

```

1080 IF c=2 THEN PRINT AT 7,0;"
You can place domino ";k;" at ei
therend. Which end do you want t
o place it (L or R) : ": GO SU
B 1900
1090 LET c=e2: IF e=1 THEN LET
c=e1
1100 LET d1=d(k,1): IF d1=c THEN
LET d1=d(k,2)
1110 GO SUB 3000
1120 FOR f=2 TO 5: PRINT AT f,k*
3+2;" ": NEXT f
1130 LET d(k,1)=9
1200 OPEN #4;"n";st1: PRINT #4;t
: PRINT #4;e: PRINT #4;d1: CLOSE
#4
1210 IF kn=2 THEN GO TO 4000
1240 FOR f=7 TO 9: PRINT AT f,o;
c$: NEXT f
1250 IF t=o THEN GO SUB 7000
1260 IF t=o AND add>o THEN GO S
UB 8700: LET sc1=sc1+add: GO SUB
7500
1280 RETURN
1900 IF INKEY$="l" OR INKEY$="L"
THEN LET e=1: BEEP .1,10: RETU
RN
1910 IF INKEY$="r" OR INKEY$="R"
THEN LET e=2: BEEP .1,10: RETU
RN
1920 GO TO 1900
1999 REM Receive move from netw
ork
2000 PRINT AT 7,o;"Please wait f
or your opponent's move."
2010 OPEN #4;"n";st1: INPUT #4;t
: INPUT #4;e: INPUT #4;d1: CLOSE
#4
2020 LET o1=e1: LET o2=e2
2025 IF t=o THEN LET kn=o
2030 IF t=1 THEN PRINT AT 7,o;
FLASH 1;" Your opponent is kno
cking. "; FLASH o;" ": PAU
SE 60: LET kn=kn+1: GO TO 2110

```

```

2040 LET dom2=dom2-1
2100 GO SUB 3000
2110 PRINT AT 7,0;c$;AT 8,0;c$
2120 IF kn=2 THEN GO TO 4000
2130 IF t=o THEN GO SUB 7000: I
F add>o THEN GO SUB 8000: LET s
c2=sc2+add: GO SUB 7500
2190 RETURN
2500 BEEP .2,0: BEEP .2,10: BEEP
.2,0: RETURN
2699 REM Placed all dominoes ro
utine
2700 IF NOT dom1 THEN LET start
=st: CLS : PRINT AT 7,0;"Well do
ne! You get 5 points.": LET sc1=
sc1+5: GO SUB 7500: GO TO 90
2710 IF NOT dom2 THEN LET start
=st1: CLS : PRINT AT 7,0;"Your o
pponent has placed all hisdomino
es and gets 5 bonus points": LET
sc2=sc2+5: GO SUB 7500: GO TO 9
0
2720 PRINT AT 21,0;"DOMINOES: ";
INK 2;"YOU: "; INK 1;dom1; INK
2;" OPPONENT: "; INK 1;dom2
2740 RETURN
2999 REM Print board
3000 IF e=2 THEN GO TO 3100
3005 LET a1=(d1=o1)
3010 LET e1=d1: PRINT INVERSE a
1; FLASH a1; PAPER 1+2*a1;AT 11,
5; INK 6;a$(e1*2+1 TO e1*2+2);AT
12,5;b$(e1*2+1 TO e1*2+2)
3020 PRINT FLASH a1; PAPER 2+a1
;AT 11,7; INK 6;a$(o1*2+1 TO o1*
2+2);AT 12,7;b$(o1*2+1 TO o1*2+2
)
3030 PRINT PAPER 1;AT 11,9; INK
6;a$(o1*2+1 TO o1*2+2);AT 12,9;
b$(o1*2+1 TO o1*2+2)
3040 RETURN
3100 LET a2=(d1=o2)

```

```

3105 LET e2=d1: PRINT INVERSE a
2; FLASH a2; PAPER 2+a2; AT 11,23
; INK 6;a$(e2*2+1 TO e2*2+2); AT
12,23;b$(e2*2+1 TO e2*2+2)
3110 PRINT PAPER 2; AT 11,19; IN
K 6;a$(o2*2+1 TO o2*2+2); AT 12,1
9;b$(o2*2+1 TO o2*2+2)
3120 PRINT FLASH a2; PAPER 1+2*
a2; AT 11,21; INK 6;a$(o2*2+1 TO
o2*2+2); AT 12,21;b$(o2*2+1 TO o2
*2+2)
3130 RETURN
3999 REM Assess score at end of
round
4000 LET to=o: FOR f=1 TO 7
4010 IF d(f,1)<9 THEN LET to=to
+d(f,1)+d(f,2)
4015 NEXT f
4020 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;to1: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;to: CLOSE
#4
4030 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;to: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;to1: CLOSE
#4
4040 IF to<to1 THEN LET start=s
t: LET sc1=sc1+5: PRINT AT 7,o;"
Your total number of spots is
less than your opponent's. You
get 5 bonus points."
4050 IF to>to1 THEN LET start=s
t1: LET sc2=sc2+5: PRINT AT 7,o;
"Your total number of spots is
more than your opponent, who
therefore gets 5 points."
4060 IF to=to1 THEN PRINT AT 7,
o;"You and your opponent have eq
ual numbers of spots unplayed. Yo
u both get 2 bonus points.": LE
T sc1=sc1+2: LET sc2=sc2+2
4070 GO SUB 7500
4080 GO TO 90

```

```

5499 REM   Check for length contr
adiction
5500 GO SUB st*10+5500
5505 GO SUB st1*10+5500
5507 RETURN
5510 OPEN #4;"n";st1: PRINT #4;s
tep: CLOSE #4: RETURN
5520 OPEN #4;"n";st1: INPUT #4;q
: CLOSE #4: RETURN
6999 REM   Calculate score for mo
ve
7000 LET add=o: LET s=e1+e2+e1*a
1+e2*a2
7010 IF s/5=INT (s/5) THEN LET
add=add+s/5
7020 IF s/3=INT (s/3) THEN LET
add=add+s/3
7030 RETURN
7499 REM   Check for end of game
7500 IF sc1<240/step AND sc2<240
/step THEN RETURN
7510 FOR f=1 TO 20: BEEP .1,RND*
10: NEXT f
7520 CLS : IF sc1>=240/step THEN
PRINT AT 10,o;"Well done! You
won by ";sc1;" to ";sc2
7530 IF sc2>=240/step THEN PRIN
T AT 10,o;"Bad luck! You lost by
";sc1;" to ";sc2
7540 INPUT "Press ENTER to play
again."; LINE a$
7550 CLS : GO TO 40
7999 REM   Initialise
8000 CLS : LET a$="  ABE GBEIGKM
N"
8010 LET b$="  CD FCHJFLHOP"
8015 PRINT AT 10,0; FLASH 1;"Ple
ase Wait. Shuffling dominoes!"
8020 DIM p(28,2): LET c=1
8025 FOR f=0 TO 12: FOR q=(f-6 A
ND f>6) TO f/2
8030 LET p(c,1)=q: LET p(c,2)=f-
q: LET c=c+1
8040 NEXT q: NEXT f

```

```

8045 LET sc1=o: LET sc2=o: LET s
tart=1
8049 RANDOMIZE : RETURN
8050 FOR f=1 TO 30
8060 LET a1=INT (RND*28)+1: LET
a2=INT (RND*28)+1
8070 LET p1=p(a1,1): LET p2=p(a1
,2)
8080 LET p(a1,1)=p(a2,1): LET p(
a1,2)=p(a2,2)
8090 LET p(a2,1)=p1: LET p(a2,2)
=p2
8100 NEXT f
8110 LET c$="
"
8120 DIM d(16,2)
8130 FOR f=1 TO 7
8140 IF st=1 THEN LET d(f,1)=p(
f*4-3,1): LET d(f,2)=p(f*4-3,2):
OPEN #4;"n";st1: PRINT #4;p(f*4
-2,1): PRINT #4;p(f*4-2,2): CLOS
E #4
8150 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;d(f,1): INPUT #4;d(
f,2): CLOSE #4
8160 NEXT f
8170 CLS : FOR f=1 TO 7
8180 FOR q=1 TO 2
8190 LET d=d(f,q)*2+1: PRINT AT
0,f*3+2;f; PAPER q; INK 6;AT q*2
,f*3+2;a$(d TO d+1);AT q*2+1,f*3
+2;b$(d TO d+1)
8200 NEXT q: NEXT f
8210 PRINT "The computer will p
lace the starting domino."
8215 LET e1=p(28,1): LET e2=p(28
,2)
8220 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;e1: INPUT #4;e2: CL
OSE #4
8225 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;e1: PRINT #4;e2: CL
OSE #4

```

```

8250 PRINT PAPER 4; AT 11,11;"
      "; AT 12,11;"
8270 LET p$="DOMINOES"
8280 FOR f=1 TO 8
8290 LET s=18538+f: LET po=s
8300 FOR q=0 TO 7: BEEP .01,q*3
8310 LET pe=PEEK (15360+CODE p$(
f)*8+q)
8320 POKE po,pe: LET po=po+256:
POKE po,pe
8330 LET po=po+256: IF q=3 THEN
LET po=s+32
8340 NEXT q: NEXT f
8350 PRINT AT 11,9; PAPER 1; INK
6;a$(e1*2+1 TO e1*2+2); AT 12,9;
b$(e1*2+1 TO e1*2+2)
8360 PRINT AT 11,19; PAPER 2; IN
K 6;a$(e2*2+1 TO e2*2+2); AT 12,1
9;b$(e2*2+1 TO e2*2+2)
8370 PRINT AT 7,o;c$; AT 8,o;c$
8380 LET dom1=7: LET dom2=7
8390 LET kn=o
8400 PRINT AT 14,o; PAPER 6;" YO
UR SCORE: "; AT 17,o;" OPPONENT'S
SCORE: "
8410 LET a1=o: LET a2=o
8500 REM Print scores
8510 INK 2: FOR f=o TO sc1*step
STEP step: IF f<256 THEN PLOT f
,42: DRAW 0,11: NEXT f
8520 INK 4: FOR f=o TO sc2*step
STEP step: IF f<256 THEN PLOT f
,18: DRAW 0,11: NEXT f
8530 INK o: PLOT 240,16: DRAW 0,
47
8600 RETURN
8700 INK 2: FOR f=sc1*step TO (s
c1+add)*step STEP step: IF f<256
THEN PLOT f,42: DRAW 0,11: NEX
T f: INK o: RETURN
8800 INK 4: FOR f=sc2*step TO (s
c2+add)*step STEP step: IF f<256
THEN PLOT f,18: DRAW 0,11: NEX
T f: INK o: RETURN

```



```

8999 REM  Graphics and instructi
ons
9000 LET o=0: FOR f=USR "a" TO U
SR "p"+7: READ q: POKE f,q: NEXT
f
9020 DATA o,o,o,o,o,o,1,3,o,o,o,
o,o,o,128,192,3,1,o,o,o,o,o,19
2,128,o,o,o,o,o,o
9030 DATA o,48,120,120,48,o,o,o,
o,o,o,12,30,30,12,o,o,48,120,120
,48,o,1,3,192,128,o,12,30,30,12,
o
9040 DATA o,12,30,30,12,o,o,o,o,
o,o,48,120,120,48,o,o,12,30,30,1
2,o,128,192,3,1,o,48,120,120,48,
o
9050 DATA o,48,120,120,48,o,48,1
20,o,12,30,30,12,o,12,30,120,48,
o,48,120,120,48,o,30,12,o,12,30,
30,12,o
9500 PRINT PAPER 2; INK 6;" DOM
INOES BY PAUL STANLEY 1984. "
9510 PRINT '"This is a 2 player
network game based on the 5s and
3s version of dominoes."
9520 PRINT '"Each player is deal
t 7 dominoes and places one by m
atching a no.on your domino with
one of the dominoes placed."
9530 PRINT '"You score by adding
the 2 end values together, an
d dividing the total by 5 and
3. If it is divisible by 3 (for
example) youscore how many 3s t
here are in the total."
9540 PRINT '"If a double is play
ed, it will flash. When you add
the scores up, both ends of th
e double are added to the total.
"
9550 INPUT "Press ENTER to start
. "; LINE a$
9999 CLS : RETURN

```

# 10 Boxing

## Hardware requirements

Two 16/48K Spectrums on ZX network

One television

This utilises the network in the same way as the Soccer game. The remote Spectrum merely transmits a keypress and serves as a joystick for the second player.

Both players control a boxer, who can punch with left and right gloves. If you manage to hit your opponent's head, you score a point and his boxer is forced to take a step backwards. If a boxer is trapped against the ropes he can only be punched a limited number of times, after which the referee will push the other fighter away. Once you begin to play the game you'll see the need for this feature, since without it every fight would end in a one-sided flurry of punches which couldn't be avoided by their target!

If you hit your opponent a certain number of times, you will knock him out. If there are no KOs, the winner will be player with the largest number of points after three rounds of the bout have been fought.

Remember that you can only press one key at a time. This is to prevent either player from holding down the punch buttons and moving at the same time. Consequently punches have to be fast, accurate and well timed.

## Program Notes:

- 1-90 Set up the variables
- 94-95 These two lines are used only by the remote computer to transmit any keypress to the other computer
- 100-120 Print the boxers
- 125 Receive which key is pressed by your opponent
- 130-150 Reduce the time by two seconds
- 400-700 Move the boxers and punch
- 1000-2140 Punch with both hands for both boxers

3000-3600 Referee separates the two boxers  
 4000-4560 Boxers take a step backwards when hit  
 6000-6800 End of game messages  
     Wait for a key to restart the game  
 7500-7999 Instructions  
 8500-8545 Return the boxers to the corners  
 9000-9610 Set up the screen and graphics  
     Move the score-grid around the ring  
 9950 Send the program to the other Spectrum

```

      1 BORDER 5: PAPER 6: INK 0: C
LS : GO SUB 9950
      2 REM @ P.STANLEY
      5 LET s1=0: LET s2=s1
      8 GO SUB 7000
     10 GO SUB 9000
     15 LET round=1
     20 GO SUB 9500
     30 GO SUB 9550
     62 PRINT AT 17,2;round
     63 BEEP .25,0: BEEP .25,0
     65 LET y=2: LET x=7: LET y1=y:
    LET x1=x
     70 LET q=23: LET f=17: LET f1=
    f: LET q1=q
     90 LET e1=0: LET e2=e1: LET tm
    =3: LET ts=2: GO TO 100
     94 POKE 23658,8
     95 OPEN #4;"n":1: PRINT #4;INK
    EY$: CLOSE #4: GO TO 95
    100 PRINT AT y1,x1;" ";AT y1+1
    ,x1;" ";AT y1+2,x1;" "; INK 2;A
    T y,x;"AC"; PAPER 0;AT y+1,x;"D"
    ; PAPER 6;AT y+2,x;"EG": LET y1=
    y: LET x1=x
    120 PRINT AT f1,q1;" ";AT f1+1
    ,q1+1;" ";AT f1+2,q1;" ";AT f,q
    ;"HI"; PAPER 7;AT f+1,q+1;"D"; P
    APER 6;AT f+2,q;"JK": LET f1=f:
    LET q1=q
    125 OPEN #4;"n":1: INPUT #4;k$:
    CLOSE #4
  
```

```

130 LET ts=ts-1: IF ts=0 THEN
LET tm=tm-1: LET ts=59: IF tm=-1
THEN GO TO 8500
150 PRINT AT 17,27;tm;": ";("0"
AND ts<10);ts
400 IF INKEY$="W" THEN GO SUB
1000
410 IF k$="W" THEN GO SUB 2000
420 IF INKEY$="P" THEN GO SUB
1100
430 IF k$="P" THEN GO SUB 2100
500 LET y=y+2*(INKEY$="6" AND y
<16)-2*(INKEY$="7" AND y>2)
510 LET x=x+(INKEY$="8" AND x<q
-1 AND x<22)-(INKEY$="5" AND x>7
)
520 LET f=f+2*(k$="6" AND f<17)
-2*(k$="7" AND f>3)
530 LET q=q+(k$="8" AND q<23)-(
k$="5" AND q>7 AND q>x+1)
700 GO TO 100
1000 PRINT INK 2;AT y,x+1;"BC":
BEEP .1,20
1010 PRINT INK 2;AT y,x+1;"C "
1020 IF y=f+1 AND x=q-1 THEN LE
T s1=s1+1: BEEP .04,0: PRINT AT
8,2;s1: GO SUB 4000
1030 IF s1=k2 THEN GO TO 6000
1090 RETURN
1100 PRINT INK 2;AT y+2,x+1;"FG
": BEEP .1,20
1110 PRINT INK 2;AT y+2,x+1;"G
"
1120 IF y=f-1 AND x=q-1 THEN LE
T s1=s1+1: BEEP .04,0: PRINT AT
8,2;s1: GO SUB 4000
1130 IF s1=k2 THEN GO TO 6000
1190 RETURN
2000 PRINT AT f,q-1;"HB": BEEP .
1,30
2020 PRINT AT f,q-1;"H"
2030 IF f=y+1 AND q=x+1 THEN LE
T s2=s2+1: BEEP .04,0: PRINT AT
8,28;s2: GO SUB 4500

```

```

2035 IF s2=k1 THEN GO TO 6500
2040 RETURN
2100 PRINT AT f+2,q-1;"JF": BEEP
    .1,30
2110 PRINT AT f+2,q-1;" J"
2130 IF f=y-1 AND q=x+1 THEN LET
    s2=s2+1: BEEP .04,0: PRINT AT
    8,28;s2: GO SUB 4500
2135 IF s2=k1 THEN GO TO 6500
2140 RETURN
3000 LET e1=0: PRINT AT y1,x1;"
    ";AT y1+1,x1;" ";AT y1+2,x1;"
    "
3010 FOR x=19 TO 12 STEP -1: LET
    x1=x
3020 PRINT AT y1,x1; INK 2;"AC";
    AT y1+2,x1;"EG";AT y1+1,x1; PAPER
    0;"D"; PAPER 6;" "; PAPER 2; I
    NK 6;"D"; PAPER 6;"D";AT y1,x1+2
    ; INK 5;"I ";AT y1+2,x1+2;"K "
3022 BEEP .02,0
3025 NEXT x
3030 PRINT AT y1,x1+2;" ";AT y1+
    1,x1+2;" ";AT y1+2,x1+2;" "
3040 RETURN
3500 LET e2=0: PRINT AT f1,q1;"
    ";AT f1+1,q1;" ";AT f1+2,q1;"
    "
3510 FOR q=9 TO 15: LET q1=q
3520 PRINT AT f1,q1+2;"HI";AT f1
    +2,q1+2;"JK";AT f1+1,q1; INK 6;"
    "; PAPER 2;"D"; PAPER 6;" "; IN
    K 0; PAPER 7;"D";AT f1,q1; INK 5
    ; PAPER 6;" A";AT f1+2,q1;" E"
3522 BEEP .02,0: NEXT q
3600 PRINT AT f1,q1;" ";AT f1+1
    ,q1;" ";AT f1+2,q1;" ": LET q1
    =17: RETURN
4000 LET q=q+2: IF q>23 THEN LET
    q=23
4010 IF q>21 THEN LET e1=e1+1:
    IF e1=3 THEN GO TO 3000

```

```

4050 PRINT AT f1,q1;"  ";AT f1+1
,q1+1;"  ";AT f1+2,q1;"  ";AT f,q
;"HI"; PAPER 7;AT f+1,q+1;"D"; P
APER 6;AT f+2,q;"JK": LET f1=f:
LET q1=q
4060 RETURN
4500 LET x=x-2: IF x<7 THEN LET
x=7
4510 IF x<9 THEN LET e2=e2+1: I
F e2=3 THEN GO TO 3500
4550 PRINT AT y1,x1;"  ";AT y1+1
,x1;"  ";AT y1+2,x1;"  "; INK 2;A
T y,x;"AC"; PAPER 0;AT y+1,x;"D"
; PAPER 6;AT y+2,x;"EG": LET y1=
y: LET x1=x
4560 RETURN
6000 PRINT AT f,q;"RHI";AT f+2,q
;"JK"; INK 2;AT f+1,q;"S"; INK
0;"pp"; INK 6; PAPER 3;"D"
6020 GO TO 6600
6500 PRINT AT y,x-1; INK 2;"AC0"
;AT y+2,x-1;"EG0"; INK 6; PAPER
2;AT y+1,x-2;"D"; PAPER 6; INK 2
;"pp"; INK 4;"P"
6600 FOR f=1 TO 20: BEEP .03,f:
BEEP .03,20-f: NEXT f
6700 INPUT "Press ENTER to play
again. "; LINE a$
6800 CLS : GO TO 5
7510 PRINT INVERSE 1;"      C H A
M P I O N S H I P      "' ' "
      B O X I N G      "
7520 PRINT INK 2;"DESIGNED & WR
ITTEN BY P.STANLEY"
7530 PRINT "This is a 2 player
network game where the remote co
mputer is      used as a joystick.
"
7540 PRINT "Each player can mov
e his boxer about with 5-8 and
punch with W and P (left and rig
ht hands.)"

```

```

7550 PRINT "Knock Outs are possible if you reach a certain score, & points are gained if a punch is made to the head."
7560 PRINT "Each bout is three 3 minute rounds."
7580 INPUT "Press ENTER to start ."; LINE b$: CLS
7590 INPUT "What is the name of the player using the remote Spectrum (up to 6 characters)? "; c$:
  IF LEN c$ > 6 THEN GO TO 7590
7600 INPUT "What is the name of the player using this Spectrum (up to 6 characters)? "; b$:
  IF LEN b$ > 6 THEN GO TO 7600
7700 LET k1 = 15 + INT (RND * 10)
7710 LET k2 = 20 + INT (RND * 10)
7999 RETURN
8500 PRINT AT 17, 27; "0:00"
8505 BEEP .25, 0; BEEP .25, 0
8510 LET round = round + 1; IF round = 4 THEN LET round = 3; GO TO 6600
8520 PRINT AT 21, 0; "Prepare to fight the next round."; INK 6; PAPER 0; AT 2, 7; "D"; AT 19, 24; "D": FOR i = 1 TO 20
8525 LET y = y - (y > 2); LET x = x - (x > 7)
8530 PRINT AT y1, x1; " "; AT y1 + 1, x1; " "; AT y1 + 2, x1; " "; INK 2; AT y, x; "AC"; PAPER 0; AT y + 1, x; "D"; PAPER 6; AT y + 2, x; "EG": LET y1 = y: LET x1 = x
8535 LET f = f + (f < 17); LET q = q + (q < 23)
8540 PRINT AT f1, q1; " "; AT f1 + 1, q1 + 1; " "; AT f1 + 2, q1; " "; AT f, q; "HI"; PAPER 7; AT f + 1, q + 1; "D"; PAPER 6; AT f + 2, q; "JK": LET f1 = f: LET q1 = q

```

```

8542 IF ATTR (2,7)=50 THEN IF A
TTR (19,24)=48 THEN GO SUB 9550
: FOR f=-1 TO 2: PRINT AT y1+f,x
1;" ": NEXT f: PRINT AT 21,0;"
":

```

```

GO TO 62

```

```

8545 BEEP .1,-18: NEXT i
9000 RESTORE 9010: LET f=0: LET
q=255: FOR y=USR "a" TO USR "t"+
7: READ x: POKE y,x: NEXT y
9010 DATA f,63,127,q,q,252,254,2
54,f,q,q,q,q,f,f,f,28,254,254,25
4,254,28,f,f
9020 DATA 195,129,f,f,f,f,129,19
5,254,254,252,q,q,127,63,f,f,f,f
,q,q,q,q,f
9030 DATA f,f,28,254,254,254,254
,28,56,127,127,127,127,56,f,f
9035 DATA f,252,254,q,q,63,127,1
27,f,f,56,127,127,127,127,56,127
,127,63,q,q,254,252,f
9045 DATA 66,129,153,165,165,90,
66,60,60,60,24,24,60,126,q,q,108
,108,108,40,40,40,40,108
9050 DATA 28,30,15,15,31,62,126,
252,248,240,240,224,224,240,240,
248,252,126,62,31,15,15,30,28
9060 DATA 56,120,240,240,248,124
,126,63,31,15,15,7,7,15,15,31,63
,126,124,248,240,240,120,56
9100 RETURN
9500 INK 4: FOR f=0 TO 7 STEP 2:
PLOT 55-f,15-f: DRAW 0,145+2*f:
DRAW 145+2*f,0: DRAW 0,-2*f-145
: DRAW -2*f-145,0: NEXT f
9510 PRINT INK 3; BRIGHT 1; AT 1
,6;"p"; AT 1,25;"p"; AT 20,6;"p"; A
T 20,25;"p"
9520 INK 0
9525 PRINT AT 4,0;b$; AT 4,26;c$
9530 PRINT AT 6,0; INK 2;"POINTS
"; INK 0; AT 6,26;"POINTS"; INK 3
; AT 15,0;"ROUND"; AT 15,27;"TIME"
9540 PRINT AT 8,2; s1; AT 8,28; s2

```



```

9545 RETURN
9550 LET y1=16: LET x1=9: RESTOR
E 9560: FOR f=1 TO 4: READ y,x
9560 DATA 0,1,-1,0,0,-1,1,0
9570 FOR e=1 TO 12: PRINT AT y1,
x1;"Z";AT y1+1,x1;"M";AT y1+2,x1
;"N"; PAPER 2; INK 7;AT y1-1,x1;
round
9580 BEEP .06,-45
9585 IF e=12 THEN FOR o=25 TO 3
5: BEEP .005,o: NEXT o: FOR o=1
TO 20: NEXT o: FOR o=35 TO 25 ST
EP -1: BEEP .005,o: NEXT o
9590 PRINT AT y1,x1;" ";AT y1+1,
x1;" ";AT y1+2,x1;" ";AT y1-1,x1
;" "
9600 LET y1=y1+y: LET x1=x1+x: N
EXT e: NEXT f
9610 RETURN
9950 POKE 23658,8: POKE 23750,5:
PRINT "Type on other Spectrum:
"""" LOAD *""n"";1 <enter>": S
AVE *""n"";1;"a" LINE 94: CLS : RE
TURN

```

# 11 Hexagrid

## **Hardware requirements**

Two 16/48K Spectrums on ZX network

Two televisions

Hexagrid is easier to play than it is to describe, but let's have a go. Each player has ten Foodle-flips (ugly frog-like creatures), which cannot be seen by their opponent.

The Foodle-flips start the game by making a slimey excursion, down one edge of a colourful hexagon grid. If they are white they move towards the right of the screen and, if purple, they move from right to left.

You can select a Foodle-flip by guiding the cursor on to the creature using the cursor keys (5–8), and then pressing 0. Once this has been accomplished, the computer will ask you whether you want the Foodle-flip to move up or down. You should respond by pressing 7 for up or 6 for down.

If your creature lands in the same hexagon as an enemy then your opponent's Foodle-flip will be destroyed. Victory goes to the first player to destroy three of the enemy.

So far so good, Now for the refinements! When your opponent moves, an indication of his position will be given by a sequence of bars down the side and at the bottom of the screen. These divide the screen into nine sections, so you can assess whether you're about to be attacked or the direction in which you should mount your assault.

This is a compulsive game for arcade obsessionals and features a very attractive screen display. A final word of tactical advice: try to spread your creatures about the screen and don't put all your Foodle-flips in one basket by just advancing one or two of them across the grid.

## **Program Notes:**

5–130 Send the program to the other computer  
Initialise by calling various routines

1000-1100 Read keyboard  
           Move and print the cursor  
 1110-1230 Decide which creature is being moved and in which direction  
 1250-1310 Read keyboard for up or down movement  
 1320-1390 Erase old creature and insert the old colours  
           Print new Foodle-flip and calculate new attributes  
           1400 Transmit the new position to the awaiting Spectrum  
 1410-1499 Receive whether there was a hit or not  
           Adjust score accordingly  
 2000-2050 Receive the co-ordinate of your opponent's creature  
           Check to see if there has been a collision  
 2060-2170 If not a hit, send a "0" over the network  
           Decide which segments to flash  
           Restore the segments to their former colours  
 2500-2560 If a hit, send a "1" over the network  
           Adjust score accordingly  
           Erase your Foodle-flip  
 4000-4050 Compare the scores with three  
           Print the winner  
           Wait for a key to be pressed to restart  
 8000-8499 Set up the screen  
           DEFine FNa to calculate the address in memory of the  
           attributes of a particular character square  
 8500-8590 Set up the creatures and variables  
 9000-9010 Define the graphics characters  
 9100-9999 Instructions

```

1 REM @ Paul Stanley 1984
5 CLS : PRINT "Type on other
Spectrum: "" " LOAD *""n";1 <
enter>": SAVE *""n";1;"a" LINE 10
10 CLS : BORDER 0: PAPER 0: IN
K 7: CLS
50 INPUT "Enter Station No (1
or 2) - ";st: IF st<>1 AND st<>2
THEN GO TO 50
60 FORMAT "n";st: LET st1=1+(s
t=1)
100 GO SUB 9000
110 GO SUB 8000
120 GO SUB 8500
130 GO SUB 1000*st
140 GO SUB 4000
150 GO SUB 1000*st1
  
```

```

160 GO SUB 4000
170 GO TO 130
1000 LET a$=" Y O U R M O V E "
1010 FOR f=1 TO 18 STEP .5: PRINT
INK INT (RND*6)+2; BRIGHT 1; A
T f,1;a$(f): BEEP .004,f+20: NEX
T f
1020 PRINT #1;"Move cursor with
5-8 and then 0"
1030 LET y=10: LET x=16: LET a=A
TTR (y,x)
1040 PRINT OVER 1;AT y,x;"+"
1045 LET y1=y: LET x1=x: LET a1=
a
1050 LET y=y+(2 AND INKEY$="6" A
ND y<18)-(2 AND INKEY$="7" AND y
>0)
1060 LET x=x+(2 AND INKEY$="8" A
ND x<30)-(2 AND INKEY$="5" AND x
>6)
1070 PRINT OVER 1;AT y1,x1;"+" :
POKE FN a(y1,x1),a1
1075 LET a=ATTR (y,x)
1080 PRINT AT y,x; OVER 1; INK 6
;"+"
1090 IF INKEY$="0" THEN GO TO 1
110
1100 GO TO 1045
1110 BEEP .1,10: PRINT OVER 1;A
T y,x;"+" : POKE FN a(y,x),a
1200 INPUT "": FOR f=1 TO 10
1210 IF m(f)=1 AND x(f)=x AND (y
(f)=y OR y(f)+1=y) THEN GO TO 1
230
1220 NEXT f: BEEP .2,-20: GO TO
1020
1230 LET m=0: LET dx=2-4*(i(f)=3
)
1250 PRINT #1;" Press 6 or 7 for
movement dirn."
1255 LET m=m+1: IF m=3 THEN INP
UT "": GO TO 1220

```

```

1260 LET k$=INKEY$: BEEP .002,20
: IF k$<>"6" AND k$<>"7" THEN G
O TO 1260
1270 BEEP .1,10: FOR q=1 TO 10
1280 IF k$="6" THEN IF y(f)+1=y
(q) AND m(q)=1 AND x(q)=x(f)+dx
THEN BEEP .2,-20: GO TO 1255
1290 IF k$="7" THEN IF y(f)-1=y
(q) AND m(q)=1 AND x(q)=x(f)+dx
THEN BEEP .2,-20: GO TO 1255
1300 NEXT q
1305 IF k$="6" AND y(f)=18 THEN
BEEP .2,-20: GO TO 1255
1310 IF k$="7" AND y(f)=0 THEN
BEEP .2,-20: GO TO 1255
1320 PRINT AT y(f),x(f); " ";AT y
(f)+1,x(f); " "
1325 POKE FN a(y(f),x(f)),a(f)
1330 POKE FN a(y(f)+1,x(f)),b(f)
1340 LET x(f)=x(f)+dx: LET y(f)=
y(f)+(k$="6")-(k$="7")
1350 LET a(f)=ATTR (y(f),x(f)):
LET b(f)=ATTR (y(f)+1,x(f))
1360 IF x(f)=6 THEN LET i(f)=7
1370 IF x(f)=30 THEN LET i(f)=3
1380 PRINT AT y(f),x(f); INK i(f
);"C";AT y(f)+1,x(f);"D"
1390 INPUT ""
1400 OPEN #4;"n";st1: PRINT #4;y
(f): PRINT #4;x(f): CLOSE #4
1410 OPEN #4;"n";st1: INPUT #4;h
it: CLOSE #4
1420 IF hit=0 THEN PRINT #1;"
No enemy foodle-flips here!": FO
R f=0 TO -20 STEP -.5: BEEP .03,
f: NEXT f
1430 IF hit=1 THEN LET sc1=sc1+
1: PRINT AT 20,0;sc1;"-";sc2: PR
INT #1;"Well done! Foodle-flip d
estroyed": FOR f=0 TO 20 STEP .5
: BEEP .03,f: NEXT f
1450 INPUT ""
1999 RETURN
2000 LET a$=" OPPONENT'S MOVE "

```

```

2010 FOR f=1 TO 18 STEP .5: PRIN
T INK INT (RND*6)+2; BRIGHT 1;A
T f,1;a$(f): BEEP .004,f+20: NEX
T f
2020 OPEN #4;"n";st1: INPUT #4;y
: INPUT #4;x: CLOSE #4
2030 FOR f=1 TO 10
2040 IF m(f) THEN IF y=y(f) AND
x=x(f) THEN GO TO 2500
2050 NEXT f
2060 OPEN #4;"n";st1: PRINT #4;0
: CLOSE #4
2070 IF y<6 THEN FOR q=0 TO 5:
PRINT AT q,3; PAPER 2; FLASH 1;"
": NEXT q
2080 IF y>5 AND y<13 THEN FOR q
=7 TO 12: PRINT AT q,3; PAPER 2;
FLASH 1;" ": NEXT q
2090 IF y>12 THEN FOR q=14 TO 1
9: PRINT AT q,3; PAPER 2; FLASH
1;" ": NEXT q
2100 IF x<14 THEN PRINT PAPER
2; FLASH 1;AT 21,5;" "
2110 IF x>12 AND x<24 THEN PRIN
T PAPER 2; FLASH 1;AT 21,14;"
"
2120 IF x>22 THEN PRINT PAPER
2; FLASH 1;AT 21,24;" "
2130 FOR q=20 TO 40: BEEP .01,q:
BEEP .01,40-q: NEXT q
2140 FOR q=40 TO 20 STEP -1: BEE
P .01,q: BEEP .01,40-q: NEXT q
2150 PRINT INK 4;AT 21,5;"ppppp
ppp pppppppppp ppppppppp"
2160 PRINT INK 4;AT 0,3;"p": FO
R f=1 TO 19: PRINT INK 4;AT f,3
;"p" AND (f+1)/7<>INT ((f+1)/7):
NEXT f
2170 RETURN
2500 OPEN #4;"n";st1: PRINT #4;1
: CLOSE #4
2510 FOR q=1 TO 5: FOR i=7 TO 0
STEP -1

```

```

2520 PRINT AT y,x; INK i;"C";AT
y+1,x;"D"
2530 BEEP .01,20-i: NEXT i: NEXT
q
2540 PRINT AT y,x;" ";AT y+1,x;"
": POKE FN a(y,x),a(f): POKE FN
a(y+1,x),b(f)
2550 LET m(f)=0: LET sc2=sc2+1:
PRINT AT 20,0;sc1;"-";sc2
2560 RETURN
4000 IF sc1<3 AND sc2<3 THEN RE
TURN
4010 IF sc1=5 THEN PRINT #1;"
You are the victor!"
4020 IF sc2=5 THEN PRINT #1;"
You have lost the game!"
4030 FOR f=1 TO 40: BEEP .03,f:
BEEP .03,41-f: NEXT f
4040 CLS : PRINT AT 10,0;"Press
any key to play again..": PAUSE
0
4050 CLS : GO TO 110
8000 LET a$=" SETTING UP SCREEN"
8010 FOR f=1 TO 18 STEP .5: PRIN
T INK INT (RND*6)+2; BRIGHT 1;A
T f,1;a$(f): BEEP .01,f: NEXT f
8050 LET a$=" B A B A B A B A B
A B A B A"
8060 LET b$=" A B A B A B A B A
B A B A B"
8100 FOR f=1 TO 17 STEP 2: PRINT
AT f,4;: FOR q=1 TO 25 STEP 4
8105 LET i=((f-1)/4=INT ((f-1)/4
))
8110 PRINT INK 1+i; PAPER 5+i;a
$(q TO q+1); PAPER 1+i; INK 5+i;
a$(q+2 TO q+3);
8120 NEXT q: NEXT f
8130 FOR f=2 TO 18 STEP 2: PRINT
AT f,4;: FOR q=1 TO 25 STEP 4
8135 LET i=(f/4=INT (f/4))
8140 PRINT INK 1+i; PAPER 5+(i=
0);b$(q TO q+1); PAPER 1+i; INK
5+(i=0);b$(q+2 TO q+3);

```

```

8150 NEXT q: NEXT f
8160 PRINT AT 0,4;: FOR q=1 TO 2
5 STEP 4: PRINT INK 2; PAPER 0;
b$(q TO q+1); PAPER 2; INK 0;b$(
q+2 TO q+3);: NEXT q
8170 PRINT AT 19,4;: FOR q=1 TO
25 STEP 4: PRINT INK 1; PAPER 0
;a$(q TO q+1); PAPER 1; INK 0;a$(
q+2 TO q+3);: NEXT q
8180 LET i=1: FOR f=0 TO 18 STEP
2: LET i=i+1-(2 AND i=2): PRINT
AT f,4; INK i;" A";AT f+1,4;" B
"; PAPER i; INK 0;AT f,31;"B";AT
f+1,31;"A": NEXT f
8190 PRINT INK 4;AT 21,5;"ppppp
ppp pppppppppp ppppppppp"
8200 PRINT INK 4;AT 0,3;"p": FO
R f=1 TO 19: PRINT INK 4;AT f,3
;"p" AND (f+1)/7<>INT ((f+1)/7):
NEXT f
8210 PRINT PAPER 3; INK 0;AT 0,
0;"<"; PAPER 0;" "; PAPER 7;">"
8300 DEF FN a(1,m)=22528+1*32+m
8400 LET sc1=0: LET sc2=0
8410 PRINT AT 20,0;"0-0"
8499 RETURN
8500 LET a$="STARTING POSITIONS"
8510 FOR f=1 TO 18 STEP .5: PRIN
T INK INT (RND*6)+2; BRIGHT 1;A
T f,1;a$(f): BEEP .01,f: NEXT f
8530 DIM i(10): DIM m(10): DIM y
(10): DIM x(10): DIM a(10): DIM
b(10)
8540 FOR f=1 TO 10
8550 LET i(f)=3+4*(st=1): LET m(
f)=1: LET x(f)=6+24*(st=2)
8560 LET y(f)=f*2-2: LET a(f)=AT
TR (y(f),x(f)): LET b(f)=ATTR (y
(f)+1,x(f))
8570 PRINT INK i(f);AT y(f),x(f
);"C";AT y(f)+1,x(f);"D"
8580 FOR q=0 TO 30 STEP 5: BEEP
.01,q: NEXT q: NEXT f
8590 RETURN

```



```

9000 FOR f=USR "a" TO USR "d"+7:
  READ q: POKE f,q: NEXT f
9010 DATA 1,3,7,15,31,63,127,255
,255,127,63,31,15,7,3,1,0,102,15
3,189,255,195,70,60,153,189,255,
126,60,36,102,0
9100 FOR q=0 TO 21 STEP 21: PRIN
T AT q,0;: FOR f=1 TO 4: PRINT
FLASH 1; PAPER f; INK 9; INVERSE
(f/2=INT (f/2));"HEXAGRID";: BE
EP .1,f+q: NEXT f: NEXT q
9110 PRINT AT 2,2; INK 6;"Writte
n by Paul Stanley 1984"
9120 PRINT "Each player has 10
foodle-flips and can move them a
cross the grid to attack enem
ies. If they are white they move
L to R, and if magenta R to L."
9130 PRINT "You cannot see your
opponents foodle-flips, but w
hen one is moved, flashing ban
ds indicate in which segment th
e foodle-flip is to be found."
9140 PRINT "The winner is the f
irst player to destroy 3 enemy
foodle-flips."
9150 PRINT "Press any key to st
art...": PAUSE 0: CLS
9999 RETURN

```

# 12 Niagara

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

This is an idiosyncratic aquatic game in which success is primarily determined by speed, although it does help if your luck co-efficient is high at the start of a session!

The screen is split into two halves. The bottom half shows the Niagara Falls and your man in a tiny boat, plaintively yelling for assistance. The top half has several crabs scattered about the screen, along with seven flashing fish.

You control a starfish in the upper section, and must move this to one of the flashing fish. The fish are numbered one to seven but you do not know which number relates to which fish. This being the case, your starfish must make contact with each fish in turn until you find number one, then two, and so on until you have caught all seven.

If your opponent catches the seven fish before you, then your man and his puny craft will move closer to the edge of the waterfall. The first boat over the edge loses the game.

In the unlikely event that both players catch the final fish at exactly the same time, one player should press **BREAK** and type in the following:

```
OPEN #4; "n"; st1: GO TO 180
```

This will resume the game and neither boat will move.

## Program Notes:

- 5-60 Send the program from one computer to the other  
Set up the stations
- 120 Open stream 4 so that the Spectrum is ready to receive a byte
- 130-170 Read the keyboard  
Move and print your starfish

- 180 Check stream 4 to see if a header has been sent
- 500-510 Check the fish number
- Increase the score
- 1000-1030 Send a header to the other Spectrum to say that all the fish have been collected
- Receive the number of fish caught by your opponent
- 2000-2015 Transmit the number of fish caught
- Move your boat
- 2020-2100 Routine for when your boat falls off the cliff
- 5000-2020 Wait for a keypress to restart the game
- 7000 Erase the fish
- 8000-8110 Set up the screen and variables
- 8300-8499 Station one sends seven fish positions
- Station two receives seven fish positions
- 8500-8510 Erase old boat and print the new one
- 9000-9080 Define the graphics characters
- 9100-9999 Instructions

```

1 REM # @ Paul Stanley 1984 #
5 CLS : PRINT "Type on other
Spectrum: "" " LOAD *""n"";1 <
enter>": SAVE *""n"";1;"a" LINE 10
10 CLS : BORDER 1: POKE 23750,
1: PAPER 0: INK 7: CLS
30 GO SUB 9000
50 INPUT "Station No (1 or 2)?
";st: IF st<>1 AND st<>2 THEN
GO TO 50
60 FORMAT "n";st: LET st1=3-st
100 GO SUB 8000
110 GO SUB 8300
120 OPEN #4;"n";st1: FOR f=20 T
O 30: BEEP .003,f: NEXT f
130 LET y1=y: LET x1=x
140 LET y=y+(y<8 AND (IN 32766=
251 OR IN 32766=187))-(y>0 AND (
IN 49150=251 OR IN 49150=187))
150 LET x=x+(x<31 AND (IN 65278
=253 OR IN 65278=189))-(x>0 AND
(IN 65278=254 OR IN 65278=190))
160 LET a=ATTR (y,x)
165 IF a>64 AND a<128 THEN LET
y=y1: LET x=x1
167 IF a>128 THEN GO SUB 500

```

```

170 PRINT AT y1,x1;" ";AT y,x;"
6"; OVER 1;AT 11,bx+1;"BC"
175 IF fi=8 THEN GO TO 1000
180 LET k$=INKEY$#4: IF k$="X"
THEN CLOSE #4: GO TO 2000
190 GO TO 130
500 IF y<>y(fi) OR x<>x(fi) THE
N LET y=y1: LET x=x1: BEEP .1,-
10: RETURN
510 BEEP .02,20: LET fi=fi+1: R
ETURN
1000 BEEP .1,20: CLOSE #4
1010 GO SUB 7000
1020 OPEN #4;"n";st1: PRINT #4;"
X": CLOSE #4
1025 OPEN #4;"n";st1: INPUT #4;1
: CLOSE #4
1040 FOR f=1 TO 10: BEEP .02,f:
NEXT f
1045 LET bx1=bx1-8+1: IF bx1<2 T
HEN PRINT FLASH 1;AT 5,0;" YO
UR OPPONENT HAS JUST MET A ";AT
6,9;" WATERY END! ": GO TO 5000
1050 GO TO 110
2000 OPEN #4;"n";st1: PRINT #4;f
i: CLOSE #4
2005 GO SUB 7000: FOR f=1 TO (8-
fi): LET ox=bx: LET bx=bx-1
2010 GO SUB 8500
2012 IF bx=1 THEN GO TO 2040
2015 NEXT f
2020 FOR f=0 TO -20 STEP -2: BEE
P .02,f: NEXT f
2030 GO TO 110
2040 PRINT AT 11,bx;" ";AT 12,
bx;" "
2050 FOR f=10 TO 18
2060 PRINT AT f,2;" "; INK 4;AT
f+1,2;"J";AT f+2,1;" K";AT f+3,2
;"L"; INK 6;AT f+3,1;"M"
2070 BEEP .03,f: NEXT f
2080 PRINT AT 19,2;" "; INK 4;AT
20,2;"J";AT 21,1;" K": BEEP .04
,19

```

```

2090 PRINT AT 20,2;" "; INK 4;AT
  21,2;"J": BEEP .04,20
2100 PRINT AT 21,2;" ": BEEP .04
,21
2110 PRINT FLASH 1;AT 5,0;"W H
A T A W A Y T O G O!"
5000 FOR f=1 TO 30: BEEP .03,f:
BEEP .03,31-f: NEXT f
5010 INPUT "Press ENTER to play
again.."; LINE i$
5020 CLS : GO TO 100
7000 FOR f=1 TO 7: PRINT AT y(f)
,x(f);" ": NEXT f: RETURN
8000 LET a$="
"
8010 FOR f=14 TO 21: PRINT PAPER
R 3;AT f,8;a$: NEXT f
8020 RANDOMIZE 100: LET l=1: FOR
  f=32 TO 63
8030 PLOT f,63: DRAW INK 3;0,-1
8040 LET l=1+RND*4
8050 NEXT f
8060 PRINT PAPER 1; INK 5; FLAS
H 1;AT 13,4;"HHHHHHHHHHHHHHHHHH
HHHHHHHHH"
8070 FOR f=13 TO 21: PRINT INK
5; PAPER 1; FLASH 1;AT f,3;"H":
NEXT f
8080 LET ox=28: LET bx=ox: GO SU
B 8500
8085 LET bx1=bx
8090 PRINT AT 9,0; PAPER 1; INK
2;"mdmdmdmdmdmdmdmdmdmdmdmdmdm
dmd"
8095 DIM y(7): DIM x(7)
8100 RANDOMIZE 5: FOR f=1 TO 25:
  PRINT AT INT (RND*9),INT (RND*3
2); BRIGHT 1; INK INT (RND*4)+3;
"N": NEXT f
8105 LET y=5: LET x=15
8110 RANDOMIZE : RETURN
8300 LET fi=1: LET q=INT (RND*7)
+1

```

```

8302 FOR f=1 TO 7
8305 LET x(q)=INT (RND*32): LET
y(q)=INT (RND*9)
8310 IF ATTR (y(q),x(q))>64 THEN
GO TO 8305
8320 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;y(q): INPUT #4;x(q)
: CLOSE #4
8330 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;y(q): PRINT #4;x(q)
: CLOSE #4
8340 PRINT FLASH 1;AT y(q),x(q)
;"I": BEEP .1,20
8350 LET q=q+1: IF q=8 THEN LET
q=1
8360 NEXT f
8499 RETURN
8500 PRINT AT 12,ox;" ";AT 11,
ox;" "
8510 PRINT AT 11,bx; INK 6;"A";
INK 7;"BC";AT 12,bx; INK 4;"DEF"
: RETURN
9000 FOR f=USR "a" TO USR "n"+7:
READ q: POKE f,q: NEXT f
9010 DATA 0,30,59,57,63,63,30,12
,0,3,42,187,42,3,0,0
9020 DATA 7,69,69,71,68,68,100,0
,63,255,255,127,127,63,31,7
9030 DATA 0,255,255,255,255,255,
255,255,0,255,255,254,254,248,24
0,192
9040 DATA 131,198,108,56,28,54,9
9,193,123,45,78,234,3,45,23,90
9050 DATA 0,140,222,251,255,222,
140,0
9060 DATA 96,120,124,126,126,127
,127,127,127,127,127,127,127,127
,127,127
9070 DATA 255,255,255,254,254,25
2,120,96,60,110,95,127,126,60,0,
0
9080 DATA 195,60,255,126,235,66,
165,165

```

```

9100 PRINT FLASH 1; PAPER 2;" N
    I A G A R A @ PAUL STANLEY "
9110 PRINT '"This is an unusual
game with a decidedly aquatic t
heme. Perhaps crabs, starfish and
the Niagara falls are not alway
s associated with each other, bu
t here they are."
9120 PRINT '"The object is to ca
tch 7 fish as quickly as you can,
because if your opponent catch
es them first, then you will inc
h closer to the edge of the fal
ls and a watery grave!"
9130 PRINT '"Move your starfish
(sights) with the following keys:
"
9140 PRINT INK 6;" Caps..LE
FT Z..RIGHT" TAB 8;"M..DOWN
K..UP"
9145 PRINT INK 5;"To catch a fi
sh, just move over it."
9150 INPUT "Press ENTER to read
more.."; LINE i$
9160 CLS : PRINT '"The crabs are
obstructions to make your pat
h to the fish more hazardous."
9170 PRINT '"If you catch 7 fish
before your opponent, he then m
oves the difference between
the number he caught and your 7.
(ie if he got 4 he would move 3 c
haracters.)"
9175 PRINT '"Oh, by the way, the
fish must be caught in a certain
sequence. Therefore you need
to remember which you have been
to and try another."
9180 PRINT '"Press any key to be
gin...": PAUSE 0
9999 CLS : RETURN

```

# 13 Number Game

## Hardware requirements

Two 48K Spectrums on ZX network

Two televisions

This is quite a complex game which should provide many hours of enjoyment. This said, Number Game has been designed to tax both your concentration and your mental resources, so if you don't like straining your grey matter, steer clear!

The start screen is a sixteen by sixteen grid, beneath which are ten numbers. You can place numbers anywhere in the grid, but they must form a series such as 0-1-2-3-4-5, 6-6-6 or 9-8-7-6-5. It is permissible, but not obligatory, for sequences to cross other numbers on the grid. For example, if you had the numbers 0, 2, 6 and 8 and there was a 4 on the screen in an appropriate position, you could arrange the numbers around the 4 to give the series 0-2-4-6-8.

Your score is determined by the sum of the numerical series, with each digit taking its literal value except for 0 which scores ten points. Thus, high scores are not necessarily dependent upon the length of a given series since, for example, 8-9 scores more points than 1-2-3-4-5. Note that a series can be any two numbers such as 70, and the 0 does not count as 10 in a series. (This means that 7-8-9-0 is not a valid series.)

If your series goes off one edge of the screen, it will continue on the other side. If your series meets another which does not follow correctly, then the computer will enter your series automatically.

When you are waiting for your opponent to make a move, I hope you enjoy one of the six tunes stored in the computer!

## Program Notes:

130-190 Ensure that one computer is transmitting and one receiving

1000-1085 Move the cursor and print it

1090-1120 Input across or down



1130-1205 Enter the sequence  
           Check validity of keypresses  
 1210-1240 Make sure that the series has a common increment  
 1245-1280 Place the number in the grid  
 1295-1315 Calculate the score  
 1325-1330 Send a header across the network  
           Send the starting position, the numbers and the score  
 1400-1999 Print the score and replace any tiles used  
 2000-2090 Select a tune and play it note by note  
 2100-2157 Receive opponent's score, numbers and starting position  
           Print the numbers into the grid  
 2160-2199 Increase and print your opponent's score  
 2800-2850 Bubble sort to place the tiles in numerical order  
 3500-3540 Erase the cursor  
 5000-5150 Compare the scores with 600  
           Wait for a key to play again  
       6000 Clear the right-hand side of the screen  
 7500-7540 Check for an end to the sequence  
 7700-7720 Move print position of the sequence and wraparound  
 8000-8499 Set up the screen  
 8500-8540 Print double-height characters  
 9000-9030 Define graphics characters  
 9100-9620 Data for the six tunes  
 9700-9989 Instructions  
       9990 Send the program to the other computer

```

      1 REM @ PAUL STANLEY
      5 GO SUB 9990
     10 CLS : BORDER 7: PAPER 7: IN
    K 0: CLS : POKE 23750,7
     15 RANDOMIZE
     20 POKE 23658,8
     30 GO SUB 9000
     50 INPUT "Station (1 or 2) ?";
    st: IF st<>1 AND st<>2 THEN GO
    TO 50
     60 FORMAT "n";st: LET st1=3-st
    100 GO SUB 8000
    130 GO SUB 1000*st
    140 GO SUB 5000
    160 GO SUB 1000*st1
    170 GO SUB 5000
    190 GO TO 130
  
```

```

1000 PRINT AT 0,20; FLASH 1;"YOU
R MOVE": PRINT AT 2,19;"MOVE CUR
SOR";AT 3,20;"WITH 5-8";AT 4,19;
"TO THE "; FLASH 1;"START"; FLAS
H 0;AT 5,20;"OF SERIES.";AT 7,19
;"PRESS 0 WHEN";AT 8,19;"IN POSI
TION."
1005 LET x=1: LET y=x: PRINT OV
ER 1;AT y,x; INVERSE 1;"A": LET
a$=SCREEN$(y,x)
1010 LET y1=y: LET x1=x: LET b$=
a$
1020 LET y=y+(INKEY$="6" AND y<1
6)-(INKEY$="7" AND y>1): LET x=x
+(INKEY$="8" AND x<16)-(INKEY$="
5" AND x>1)
1030 PRINT OVER 1;AT y1,x1; PAP
ER 1+(6 AND b$<>""); INK 7-(5 AN
D b$<>""); INVERSE 1;"A"
1040 LET a$=SCREEN$(y,x)
1050 PRINT OVER 1;AT y,x; INVER
SE 1; INK 3;"A"
1060 IF INKEY$="0" THEN BEEP .1
,20: LET x1=x: LET y1=y: GO TO 1
085
1080 GO TO 1010
1085 PRINT FLASH 1; INK 3; OVER
1; INVERSE 1;AT y,x;"A"
1090 LET put=0: PRINT FLASH 1;A
T 10,19;"A"; FLASH 0;"cross/"; F
LASH 1;"D"; FLASH 0;"own?";
1100 LET k$=INKEY$: IF k$<>"A" A
ND k$<>"D" THEN GO TO 1100
1110 IF k$="D" THEN LET dx=0: L
ET dy=1
1115 IF k$="A" THEN LET dx=1: L
ET dy=0
1120 PRINT INK 2; FLASH 1;k$: B
EEP .1,10
1130 PRINT AT 12,18;"ENTER SEQUE
NCE"
1133 LET sy=y: LET sx=x
1135 LET s=1: DIM s$(16): DIM u(
10)

```

```

1140 PRINT AT 14,19;: GO SUB 750
0
1150 LET k$=INKEY$: IF CODE k$=1
3 THEN GO TO 1210
1155 IF k$<"0" OR k$>"9" THEN G
O TO 1150
1160 BEEP .1,10: FOR q=1 TO 10
1170 LET k=VAL k$: IF t(q)=k AND
u(q)=0 THEN GO TO 1180
1175 NEXT q: GO TO 1150
1180 BEEP .1,15: LET s$(s)=k$: P
RINT INK 1;k$;
1190 LET put=put+1: LET u(q)=1:
LET x=x+dx: LET y=y+dy: LET s=s+
1
1195 GO SUB 7700
1200 GO SUB 7500: IF inc=77 THEN
GO TO 1210
1205 GO TO 1150
1210 BEEP .1,15: IF s<3 THEN LE
T add=0: GO SUB 3500: BEEP .4,-2
0: GO TO 1325
1218 LET inc=VAL s$(1)-VAL s$(2)
1220 FOR q=1 TO s-2
1230 IF VAL s$(q)-VAL s$(q+1)<>i
nc THEN BEEP .3,-20: GO SUB 600
0: GO SUB 3500: GO TO 1000
1240 NEXT q
1245 LET y=sy: LET x=sx
1250 FOR f=1 TO s-1
1260 PRINT AT y,x; INK 2;s$(f)
1270 LET y=y+dy: LET x=x+dx
1275 GO SUB 7700
1280 NEXT f
1290 LET add=0
1295 FOR f=1 TO s-1
1300 LET add=add+VAL s$(f)+10*(s
$(f)="0")
1310 NEXT f
1315 IF put=0 THEN LET add=0
1325 OPEN #4;"n";st1: PRINT #4;"
X": CLOSE #4: BEEP .1,10:

```

```

1330 OPEN #4;"n";st1: PRINT #4;s
y: PRINT #4;sx: PRINT #4;dy: PRI
NT #4;dx: PRINT #4;s-1: PRINT #4
;s$: PRINT #4;add: CLOSE #4
1340 GO SUB 6000
1400 LET sc1=sc1+add: PRINT AT 1
8,19; PAPER 3; INK 6;"SCORE:";SC
1;"-";SC2
1730 FOR f=1 TO 10
1740 IF u(f)=1 THEN LET t(f)=IN
T (RND*10)
1750 NEXT f
1999 RETURN
2000 PRINT AT 0,19; FLASH 1;"OPP
ONENT'S GO"
2010 PRINT AT 2,20; INVERSE 1; F
LASH 1;"PLEASE WAIT"
2015 GO SUB 2800
2020 PRINT AT 6,21;"Enjoy the";A
T 7,22;"music!!"
2030 LET t=t+1: IF t=7 THEN LET
t=1
2035 RESTORE 9000+100*t: LET h=q
(t)
2040 LET n=1
2050 OPEN #4;"n";st1
2060 LET k$=INKEY$#4: IF k$="X"
THEN CLOSE #4: BEEP .1,10: GO T
O 2100
2070 READ dur,pit: BEEP dur,pit
2080 LET n=n+1: IF n=h THEN RES
TORE 9000+100*t: LET n=1
2090 GO TO 2060
2100 OPEN #4;"n";st1: INPUT #4;y
: INPUT #4;x: INPUT #4;dy: INPUT
#4;dx: INPUT #4;l: INPUT #4;s$:
INPUT #4;add: CLOSE #4
2105 IF add=0 THEN GO TO 2160
2110 LET sy=y: LET sx=x: FOR q=1
TO 0 STEP -1: FOR f=1 TO 1
2120 PRINT AT y,x; FLASH q; INK
2+q; PAPER 7-q;s$(f)
2130 LET y=y+dy: LET x=x+dx
2140 GO SUB 7700

```

```

2150 NEXT f: LET y=sy: LET x=sx
2155 IF q=1 THEN FOR h=10 TO 15
  STEP .3: BEEP .01,15-h: BEEP .0
1,h: NEXT h
2157 NEXT q
2160 LET SC2=SC2+ADD: PRINT PAP
ER 3; INK 6; AT 18,19; "SCORE:"; SC
1; "-"; SC2
2170 GO SUB 6000
2199 RETURN
2800 LET C=0: FOR f=1 TO 9
2810 IF t(f+1)<t(f) THEN LET a=
t(f+1): LET t(f+1)=t(f): LET t(f
)=a: LET c=1
2820 NEXT f
2830 IF c<>0 THEN GO TO 2800
2840 FOR f=1 TO 10: LET c=t(f)+4
8: LET x=3*f-2: GO SUB 8500: NEX
T f
2850 NEXT f: RETURN
3500 IF SCREEN$ (y1,x1)=" " THEN
  PRINT OVER 1; AT y1,x1; PAPER 1
; INK 7; " "
3520 IF SCREEN$ (y1,x1)<>" " THEN
  PRINT OVER 1; AT y1,x1; INK 2;
" "
3540 RETURN
5000 IF sc1<600 AND sc2<600 THEN
  RETURN
5010 FOR f=20 TO 50 STEP .3: BEE
P .02,f: NEXT f
5100 IF sc1>=600 THEN PRINT AT
10,0; FLASH 1;"W E L L D O N E
! Y O U W O N"
5110 IF sc2>=600 THEN PRINT AT
10,0; FLASH 1;"B A D L U C K !
Y O U L O S T"
5120 INPUT "PRESS ENTER TO PLAY
AGAIN.."; LINE 1$
5130 CLS
5150 GO TO 100
6000 FOR f=0 TO 14: PRINT AT f,1
8;"
": NEXT f: RETU
RN

```

```

7500 LET code=CODE SCREEN$ (y,x)
7505 LET inc=99: IF s$(1)<>" " A
ND s$(2)<>" " THEN LET inc=VAL
s$(2)-VAL s$(1)
7510 IF code<CODE "0" OR code>CO
DE "9" THEN RETURN
7520 LET s$(s)=SCREEN$ (y,x)
7522 IF inc<>99 THEN IF VAL s$(
s)-inc<>VAL s$(s-1) THEN LET s$
(s)=" ": LET INC=77: RETURN
7528 PRINT INK 1;s$(s);: LET s=
s+1: LET y=y+dy: LET x=x+dx
7530 GO SUB 7700
7535 IF s=17 THEN RETURN
7540 GO TO 7500
7700 IF y=17 THEN LET y=1
7710 IF x=17 THEN LET x=1
7720 RETURN
8000 CLS : PRINT PAPER 4; INK 2
; "emmmmmmmmmmmmmmmmmmmmmi"; AT 17,0; "b
dddddcccccccccccccccc"
8010 FOR f=1 TO 16: PRINT AT f,0
; PAPER 4; INK 2;"f"; PAPER 1; I
NK 7;"AAAAAAAAAAAAAAAA"; PAPER 4
; INK 2;"k": NEXT f
8030 DIM t(10): FOR f=1 TO 10: L
ET t(f)=INT (RND*10)
8040 LET x=3*f-2
8050 PRINT AT 20,x-1; INK 1;"B C
"; AT 21,x-1;"D E"; BRIGHT 1; INK
6; PAPER 1; AT 20,x;" "; AT 21,x;
" "
8060 NEXT f
8070 GO SUB 2800
8100 LET t=0: LET sc1=0: LET sc2
=0
8200 LET m=.5: LET d=.25: LET q=
.13: LET i=.08
8300 DIM q(6): LET q(1)=38: LET
q(2)=57: LET q(3)=46: LET q(4)=3
5: LET q(5)=28: LET q(6)=59
8499 RETURN
8500 LET s=20608+x: LET po=s

```

```

8510 FOR k=0 TO 7: LET pe=PEEK (
15360+c*8+k)
8520 POKE po,pe: LET po=po+256:
POKE po,pe
8530 LET po=po+256: IF k=3 THEN
LET po=s+32
8540 NEXT k: RETURN
8600 LET a$="AEIOU": LET i=INT (
RND*5)+1
8610 LET l(f)=CODE a$(i): RETURN

9000 LET q=240: LET m=1: LET d=1
5: FOR f=USR "a" TO USR "e"+7: R
EAD q: POKE f,q: NEXT f
9010 DATA m,m,m,m,m,m,m,255,3,7,
d,d,d,d,d,d
9020 DATA 192,224,q,q,q,q,q,q,d,
d,d,d,d,d,7,3
9030 DATA q,q,q,q,q,q,224,192
9100 DATA d,12,d,14,q,12,q,10,m,
9,q,7,q,5,d,4,d,2,m,4,m,4,d,5,d,
7,q,5,q,4,m,2,q,4,q,5,d,7,d,9,m,
10,d,12,d,14,q,12,q,10,m,9
9110 DATA q,7,q,5,d,4,d,2,m,4,m,
4,d,5,d,10,m,4,m,4,m,5
9200 DATA q,7,q,5,m,5,q,9,q,11,q
,13,q,14,q,16,q,17,d,16,q,14,m,1
4
9210 DATA q,14,q,14,q,12,q,10,q,
9,q,7,d,10,q,9,d,9,d,7,d,5,d,9,d
,7,d,2,d,5,q,9,m,9
9220 DATA m,9,m,9,d,14,d,16,d,17
,q,16,q,14,m,16,q,14,d,12,d,14,m
,9
9230 DATA m,9,m,9,d,14,d,16,d,17
,q,16,q,14,m,16,q,14,d,12,d,16,d
,17,d,12,d,10,d,9
9300 DATA q,7,q,6,q,7,d,12,i,4,d
,3,i,4,d,7,i,0,d,-1,m,2
9310 DATA q,7,q,6,q,7,d,12,i,4,d
,3,i,4,d,7,i,0,m,2

```

```

9320 DATA q,4,q,3,q,4,d,7,i,0,d,
-1,i,0,d,4,i,7,d,9,m,12,i,12,q,1
4,q,14,q,14,q,14,q,14,q,12,
q,12,q,12,d,12,i,12,m,11
9400 DATA d,12,m,9,q,9,q,12,q,14
,m,7,q,7,q,9,d,10,m,17,q,17,q,16
,q,12,q,14,m,9
9410 DATA q,12,q,14,d,14,q,14,q,
19,q,17,q,16,q,17,q,14,m,12,q,5,
q,7,d,9,d,14,q,12,d,9,q,4,d,4,m,
5
9500 DATA m,4,d,7,m,14,m,12,d,7,
m,5,m,4,d,7,d,7,d,9,d,11,m,12,m,
12,d,14,q,7,d,7,d,11,d,9
9510 DATA d,7,m,4,d,7,m,12,m,9,d
,12,m,14,d,12,m,11,m,7
9600 DATA m,5,d,9,q,12,q,19,d,17
,q,10,q,14,m,12,d,9,q,10,q,12,d,
14,q,5,q,7,d,9,q,9,q,9,m,9,d,9,q
,10
9610 DATA q,9,d,9,d,7,d,5,q,7,q,
9,d,16,d,14,m,7,q,7,q,8,q,10,q,1
2,q,15,m,12,q,10,q,8,q,10,q,12,q
,13
9620 DATA q,10,q,7,q,7,q,8,q,10,
q,12,q,8,m,5,q,10,q,12,q,13,q,10
,q,7,q,7,q,8,q,10,q,12,q,9
9700 PRINT PAPER 3; INK 6; FLAS
H 1;" THE MUSICAL NUMBER G
AME "
9710 PRINT INK 2;"DESIGNED & WR
ITTEN BY P.STANLEY"
9720 PRINT '"This is a complex q
ame which involves entering s
eries of numbers in a grid.
These series must be such that t
he difference between successive
numbers is equal."'
9730 PRINT '"Valid series are:-"
'"0123"' "8642"' "000"' "or even ju
st 38"
9740 PRINT '"Invalid series are:
-"' "2323"' "9765"' "67890 (NB: 0 d
oes not equal 10!)"

```



```

9745 INPUT  INK 3;"Press ENTER..
"; LINE p$: CLS
9750 PRINT "The series you posit
ion may use numbers already in t
he grid (eq if there was a 5 on
the screen and you had 4,6&7, t
hen you can place the 4, the 5 w
ill be notedby the computer, the
n you place 6&7). However, your
series does NOT have to meet ano
ther."
9760 PRINT '"Should your series
go off one edge of the grid, i
t will appear on the other side.
When you type in your series, if
it meets another which does
not follow in the sequence, your
series will be entered automati
cally without you pressing ENTER.
"
9770 INPUT  INK 3;"Press ENTER..
"; LINE p$: CLS
9780 PRINT "You must position th
e cursor at the FIRST number of
the series even if it is not on
e you have placed yourself."
9790 PRINT '"Note that the serie
s which are adjacent to the one
you place do not have to fit in
with yours."
9800 PRINT '"The score system wo
rks by giving you the sum of the
numbers in the series. 0 score
s 10 points (but remember that
it does not represent 10 in a s
eries)."
9810 PRINT '"The winner is the f
irst person to reach 600 points
."
9820 PRINT  INK 3'"Do you want t
o read these instructions
again? (Y or N)"

```

```
9830 LET k$=INKEY$: IF k$<>"Y" A
ND k$<>"N" THEN GO TO 9830
9840 BEEP .1,20: IF k$="Y" THEN
  CLS : GO TO 9700
9989 CLS : RETURN
9990 CLS : PRINT "Type on other
  Spectrum:''" LOAD *"n";1
<enter>": SAVE *"n";1;"a" LINE 1
0
9995 RETURN
```

# 14 Detective

## **Hardware requirements**

Two 48K Spectrums on ZX network

Two televisions

This is a race against time for two players. The object of the game is very straightforward: you must attempt to solve a crime before your opponent. Four witnesses hold crucial pieces of information which will enable you to discover the identity of the murderer, as well as where and at what time the crime took place.

Whenever it's your turn you have the chance to interview any witness and propose a murderer, a time and a location for the dastardly deed. If the witness knows that any of your suggestions are wrong he will tell you. If he knows that more than one of your guesses are wrong, then you'll be offered one correction (the priority being murderer, time and finally place).

If you discover anything new from an interview, you should switch to the computerised notepad and make appropriate comments. If you want a space in your notes you should press 8 (cursor right) and not SPACE which will BREAK into the program. Pressing 0 allows the deletion of the character under the cursor.

When you know who the murderer is and at what time and what part of town the murder took place, you should switch to Prosecute Mode. If your allegations are correct you will win the game. But be warned – an incorrect prosecution will lose you the game!

To succeed as a detective you need to logically check each possibility in turn and note every item of information provided by your witness.

## **Program Notes:**

- 1-130 Call initial routines
- 400-470 Check stream 4 to see if a header has been sent
- 500-550 Set up the notebook screen

590-700 Move and print notebook cursor  
     Enter letters  
 1000-1080 Enter a function  
 2000-2050 Input the name of the interviewee  
 2060-2240 Fill in the questions  
 2250-2290 Compare the questions with information held by the witness  
 2300-2510 Print the results of the interview  
 4000-4040 Wait for the results of your opponent's prosecution  
 4050-4070 Print the final result  
 6000-6100 Input a prosecution  
     6110 Send a header to prepare the other computer for receiving the  
         result of the prosecution  
 6120-6140 Calculate if the prosecution was correct  
     Transmit the result of the prosecution  
 6150-6600 Print the result of the prosecution  
     Wait for a key to restart  
 7000-7060 Flash the murderer, time and place onto the screen  
 8000-8040 Set up the variables  
 8070-8999 If station one, then send the variables  
     If station two, then receive the variables  
     9000 Define the graphics character  
 9100-9200 Instructions  
 9500-9540 Data for places, times, murderers and witnesses  
     9999 Send the program to the other computer

```

1 REM @@ Paul Stanley 1984 @@
5 GO SUB 9999
10 CLS : RANDOMIZE : BORDER 7:
PAPER 7: INK 0: CLS : POKE 2375
0,7
30 INPUT "Station No (1 or 2)?
";st
35 IF st<>1 AND st<>2 THEN GO
TO 35
40 FORMAT "n";st
45 LET st1=3-st
50 POKE 23658,8
100 GO SUB 9000
130 GO SUB 8000
400 GO TO 1000
450 OPEN #4;"n";st1: LET a$=INK
EY$#4: CLOSE #4
460 IF a$="X" THEN GO TO 4000
470 RETURN

```

```

500 CLS : PRINT PAPER 2; INK 6
;" SUSPECTS "; PAPER 7;" "; P
APER 1;" 9=QUIT 0=DELETE "
510 FOR f=1 TO 10: PRINT m$(f);
INK 3;TAB 21; BRIGHT 1;n$(f): N
EXT f
520 PRINT PAPER 2; INK 6;" POS
SIBLE TIMES "; PAPER 7;TAB 21; B
RIGHT 1;" "
530 FOR f=1 TO 5: PRINT INK 1;
t$(f); INK 3;TAB 21; BRIGHT 1;n$(
(10+f)): NEXT f
540 PRINT PAPER 2; INK 6;" POS
SIBLE PLACES "; PAPER 7;TAB 21;
BRIGHT 1;" "
550 FOR f=1 TO 4: PRINT INK 2;
p$(f); INK 3;TAB 21; BRIGHT 1;n$(
(f+15)): NEXT f
570 LET y=1: LET x=21: GO SUB 4
50
580 LET y1=y: LET x1=x
590 LET y=y+(INKEY$="6" AND y<2
1)-(INKEY$="7" AND y>1)
600 LET x=x+(INKEY$="8" AND x<3
1)-(INKEY$="5" AND x>21)
610 PRINT OVER 1; INK 3; BRIGH
T 1;AT y1,x1;" "
620 PRINT OVER 1; PAPER 4; BRI
GHT 1;AT y,x;" "
625 LET w=y-(y>11)-(y>17)
630 LET k$=INKEY$
640 IF k$="0" THEN PRINT AT y,
x; BRIGHT 1;" ": LET n$(w,x-20)=
" ": LET x=x-(x>21): PRINT BRIG
HT 1; OVER 1; PAPER 4;AT y,x;" "
: BEEP .1,5: GO TO 580
650 IF k$>"4" AND k$<"9" THEN
GO TO 580
655 IF k$="9" THEN RETURN
660 IF CODE k$<32 OR CODE k$>90
THEN GO TO 580
665 IF y=11 OR y=17 THEN GO TO
580
667 GO SUB 450

```

```

670 PRINT BRIGHT 1;AT y,x; INK
3;k$
675 LET n$(w,x-20)=k$
680 LET x=x+(x<31): PRINT BRIG
HT 1; PAPER 4;AT y,x; OVER 1;" "
700 BEEP .1,5: GO TO 580
1000 BEEP .1,10: CLS : PRINT IN
K 1; PAPER 6; BRIGHT 1;" WHICH F
UNCTION? "
1010 PRINT "1...Interview Someb
ody"
1020 PRINT INK 1'"2...Note book
"
1030 PRINT "3...Prosecute someb
ody"
1040 LET k$=INKEY$: GO SUB 450
1050 IF k$="1" THEN BEEP .2,10:
GO SUB 2000: GO TO 1000
1060 IF k$="2" THEN BEEP .2,10:
GO SUB 500: GO TO 1000
1070 IF k$="3" THEN BEEP .2,10:
GO SUB 6000: GO TO 1000
1080 GO TO 1040
2000 CLS : PRINT PAPER 1; INK 6
;"Who do you want to interview?"
2010 FOR f=1 TO 4
2020 PRINT 'f; INK 2;"..."; INK
1;w$(f)
2030 NEXT f
2040 LET k$=INKEY$: GO SUB 450:
IF k$<"1" OR k$>"4" THEN GO TO
2040
2050 LET k=VAL k$: BEEP .2,10: C
LS
2060 PRINT INK 3'"Fill in the b
lanks to question" 'w$(k)
2065 PRINT INK 4'"Select with 1
then press 0."
2070 PRINT INK 2'"On the night
of 30th February,"'"could the m
urder have been done by"
2080 PRINT INK 2'"between the h
ours""in";TAB 13;"part of town
?"

```

```

2090 LET m=1
2100 PRINT INK INT (RND*5);AT 9
,3;m$(m)
2105 GO SUB 450
2110 LET k$=INKEY$: IF k$<>"0" A
ND k$<>"1" THEN GO TO 2100
2120 BEEP .05,10: IF k$="1" THEN
LET m=m+1: IF m=11 THEN LET m
=1
2130 IF k$="1" THEN GO TO 2100
2140 PRINT AT 9,3;m$(m): LET mur
d1=m: LET m=1
2150 PRINT INK INT (RND*5);AT 1
1,18;t$(m)
2155 GO SUB 450
2160 LET k$=INKEY$: IF k$<>"0" A
ND k$<>"1" THEN GO TO 2150
2170 BEEP .05,10: IF k$="1" THEN
LET m=m+1: IF m=6 THEN LET m=
1
2180 IF k$="1" THEN GO TO 2150
2190 PRINT AT 11,18;t$(m): LET t
ime1=m: LET m=1
2200 PRINT INK INT (RND*5);AT 1
3,3;p$(m)
2205 GO SUB 450
2210 LET k$=INKEY$: IF k$<>"0" A
ND k$<>"1" THEN GO TO 2200
2220 BEEP .05,10: IF k$="1" THEN
LET m=m+1: IF m=5 THEN LET m=
1
2230 IF k$="1" THEN GO TO 2200
2240 PRINT AT 13,3;p$(m): LET pl
ace1=m
2250 LET i=0: FOR n=1 TO 3: FOR
q=1 TO 4
2260 IF n=1 AND i(k,q,1)=n AND i
(k,q,2)=murd1 THEN LET i=n: GO
TO 2300
2270 IF n=2 AND i(k,q,1)=n AND i
(k,q,2)=time1 THEN LET i=n: GO
TO 2300

```

```

2280 IF n=3 AND i(k,q,1)=n AND i
(k,q,2)=place1 THEN LET i=n: GO
TO 2300
2290 NEXT q: NEXT n
2300 FOR f=1 TO 10: BEEP .05,INT
(RND*7): NEXT f
2310 CLS : IF i=0 THEN PRINT "
AI'm sorry, I cannot help you.A"
2320 IF i=1 THEN PRINT "AI can
tell you the murderer is not "
;m$(murdl);".A"
2330 IF i=2 THEN PRINT "AI can
tell you that the murder did no
t take place between the times
";t$(time1);".A"
2340 IF i=3 THEN PRINT "AI can
tell you that the murder did no
t take place in ";p$(place1)"pa
rt of town.A"
2350 PRINT AT 16,0; INK 3;"Press
any key..."
2400 GO SUB 450
2450 IF INKEY$="" THEN GO TO 24
00
2500 CLS
2510 RETURN
4000 CLS : FOR f=1 TO 10 STEP .3
: BEEP .03,f: NEXT f
4010 PRINT "Your opponent has p
rosecuted somebody for the mu
rder."
4020 FOR f=10 TO 1 STEP -.3: BEE
P .03,f: NEXT f
4030 OPEN #4;"n";st1: INPUT #4;r
ight: CLOSE #4
4040 GO SUB 7000
4050 IF right=0 THEN PRINT AT 2
0,0;"Your opponent made an incor
rect prosecution. Therefore you
win!"
4060 IF right=1 THEN PRINT AT 2
0,0;"Your opponent has correctly
prosecuted and has beaten y
ou."

```



```

4070 GO TO 6500
6000 CLS : PRINT "This decision
      is vital. If you enter the wrong
      prosecution, you are the loser
      ."
6010 PRINT "It is a good idea to
      know what your prosecution is
      for the next stage. Therefore I
      suggest that you check your notes
      before you proceed."
6020 PRINT INK 2; FLASH 1; "Press
      1 to see notes or 0."
6030 LET k$=INKEY$: GO SUB 450:
      IF k$="" THEN GO TO 6030
6040 BEEP .1,10: IF k$="1" THEN
      GO SUB 500
6050 CLS : PRINT "Who is the murderer?": PRINT : FOR f=1 TO 10:
      PRINT f; ".."; m$(f): NEXT f
6060 INPUT "Murderer Number? "; q1
6070 CLS : PRINT "What time was
      the murder?": PRINT : FOR f=1 TO
      5: PRINT f; ".."; t$(f): NEXT f
6080 INPUT "Time Number? "; q2
6090 CLS : PRINT "In which part
      of the town was the murder?":
      PRINT : FOR f=1 TO 4: PRINT f; ".
      ."; P$(f): NEXT f
6100 INPUT "Place Number? "; q3
6110 OPEN #4;"n";st1: PRINT #4;"
      X": CLOSE #4
6120 BEEP .1,10
6125 LET right=0
6130 IF q1=murd AND q2=time AND
      q3=place THEN LET right=1
6140 OPEN #4;"n";st1: PRINT #4;right:
      CLOSE #4
6150 GO SUB 7000
6160 IF right=0 THEN PRINT AT 2
      1,0;"You were wrong, hence you lose!"

```

```

6170 IF right=1 THEN PRINT AT 2
1,0;"You were correct, hence you
win!"
6500 FOR f=1 TO 20: BEEP .03,INT
(RND*10): NEXT f
6530 INPUT "Press ENTER to play
again.. "; LINE v$
6600 GO TO 130
7000 CLS : FOR f=1 TO 10: PRINT
m$(f);TAB 21; BRIGHT 1; INK 3;n$
(f): NEXT f
7010 PRINT FLASH 1; OVER 1;AT m
urd-1,0;"
7020 FOR f=1 TO 5: PRINT AT 9+f,
0; INK 1;t$(f);TAB 21; BRIGHT 1;
INK 3;n$(f+9): NEXT f
7030 PRINT FLASH 1; OVER 1; INK
1;AT 9+time,0;"
7040 FOR f=1 TO 4: PRINT AT 14+f
,0; INK 2;p$(f);TAB 21; BRIGHT 1
; INK 3;n$(f+14): NEXT f
7050 PRINT FLASH 1; OVER 1; INK
2;AT 14+place,0;"
7060 RETURN
8000 DIM t(19,2): DIM m$(10,19):
DIM t$(5,10): DIM p$(4,9): DIM
n$(19,11): DIM w$(6,13): DIM i(4
,4,2)
8005 CLS : PRINT FLASH 1; INK 3
;AT 9,9;" ";AT 10,9;
" PLEASE WAIT ";AT 11,9;"
"
8010 RESTORE 9500: FOR f=1 TO 10
: READ m$(f): LET t(f,1)=1: LET
t(f,2)=f: NEXT f
8020 FOR f=1 TO 5: READ t$(f): L
ET t(f+10,1)=2: LET t(f+10,2)=f:
NEXT f
8030 FOR f=1 TO 4: READ p$(f): L
ET t(15+f,1)=3: LET t(15+f,2)=f:
NEXT f
8040 FOR f=1 TO 4: READ w$(f): N
EXT f

```

```

8060 IF st=2 THEN GO TO 8500
8070 OPEN #4;"n";st1
8100 LET murd=INT (RND*10)+1: LET
  t(murd,1)=0
8105 PRINT #4;murd
8110 LET time=INT (RND*5)+1: LET
  t(time+10,1)=0
8115 PRINT #4;time
8120 LET place=INT (RND*4)+1: LET
  t(place+15,1)=0
8125 PRINT #4;place
8130 FOR f=1 TO 4: FOR q=1 TO 4
8140 LET a=INT (RND*19)+1: IF t(
  a,1)=0 THEN GO TO 8140
8150 LET i(f,q,1)=t(a,1)
8160 LET i(f,q,2)=t(a,2)
8170 LET t(a,1)=0
8175 PRINT #4;i(f,q,1): PRINT #4
  ;i(f,q,2)
8180 NEXT q: NEXT f
8190 CLOSE #4
8200 RETURN
8500 OPEN #4;"n";st1
8510 INPUT #4;murd: INPUT #4;tim
  e: INPUT #4;place
8520 FOR f=1 TO 4: FOR q=1 TO 4
8530 INPUT #4;i(f,q,1): INPUT #4
  ;i(f,q,2)
8540 NEXT q: NEXT f
8550 CLOSE #4
8999 RETURN
9000 FOR f=0 TO 7: READ q: POKE
  USR "a"+f,q: NEXT f
9010 DATA 0,40,40,0,0,0,0,0
9100 PRINT INK 2; BRIGHT 1;" D
  E T E C T I V E @ P.STANLEY "
9110 PRINT "A murder was commit
  ted and the body dumped. You pl
  ay the part of a detective, and
  you must find who, where and
  when by asking witnesses fo
  r information"

```

```

9120 PRINT "You suggest a murderer, place & time to the witness who may be able to tell you if any of your suggestions are wrong."
9130 PRINT "Each witness has 4 facts and if he knows more than one of your suggestions are wrong he will only tell you one (in the order murderer, time, place)"
9140 INPUT "Press ENTER.."; LINE 1$: CLS
9150 PRINT "When you have found out anything you should enter it in your computerised notebook. When you have decided who, what & where, you should enter it as quickly as possible, because the winner is the first to solve the crime."
9160 INPUT "Press ENTER to start .."; LINE 1$: CLS
9200 RETURN
9500 DATA "Bill Basher", "Jack AT he KnifeA", "Freddie Fingers", "Mad AAXe ManA Ollie", "Elusive Eddie"
9510 DATA "Conan Con-Man", "The Professor", "Keith AThe ButcherA", "Mick Meat-Cleaver", "Mr AXA"
9520 DATA "10-11 pm", "11-12 pm", "12 pm-1 am", "1-2 am", "2-3 am"
9530 DATA "the north", "the south", "the east", "the west"
9540 DATA "Honest Henry", "Generous Jim", "Kind Conrad", "Peaceful Pete"
9999 CLS : PRINT "Type on the other Spectrum: "" LOAD *""n"";1
<enter>": SAVE *""n"";1;"a" LINE 10: RETURN

```

# 15 The Tortoise and the Hare

## **Hardware requirements**

Two 16/48K Spectrums on ZX network

Two televisions

This is an unusual game in that, unlike most of the others in this collection, you need to be able to see both televisions at the same time. This is because you have control over two characters – the hare on your screen, and the tortoise on your opponent's screen.

You can move the hare in four directions, and as he bounces about the screen you must guide him to carrots to collect points. However, don't let your quest for high scores distract you from your second objective – trying to beat your opponent to the finishing line! In addition to advancing your hare and consuming carrots, you can also get to move your tortoise. Each time the tortoise advances you will be rewarded with points, and should the tortoise beat your opponent's hare, then you will be given a large bonus.

Since you can't simultaneously move both your hare and the tortoise you must decide which would produce the greater point advantage. If your hare is almost at the finish, but your opponent's is nowhere near, then you could advance the tortoise for bonus points and revert back to the hare at the last moment. As the winner is the first player to reach 300 points, you could move the tortoise for easy points if you only needed a few to win.

This game features up-market graphics and tests both your agility and tactics. Tortoise and Hare also features an additional handicap to test your reflexes, but to make things a little trickier we'll let you work it out for yourself! The programmers amongst you will be at a distinct advantage since our listing should provide sufficient information to allow you to work out this particular hurdle before you start.

**Program Notes:**

10-30 Call the starting routines  
 100-150 Read the keyboard  
     Move and print hare  
 155 Check to see if hare has finished  
 160-170 Receive across the network the key pressed by your opponent  
     Send a key pressed by you to the other Spectrum  
 210 Decrease the delay before keys change their function  
 220 Move and print the opposing tortoise  
 500-510 Increase the score if a carrot is eaten  
 520-570 Work out who is the winner  
     Wait for a key to be pressed to play again  
 1000-1040 Change the keys' function  
 2000-3040 Send and receive information when somebody wins the race  
 4000-4030 End of race statements  
 8000-8190 Set up the screen display  
 8200-8999 Send your score and receive your opponent's  
     Compare with 300  
 9000-9060 Define the graphics characters  
 9100-9210 Tortoise and Hare title page  
 9220-9900 Instructions  
     9990 Send the program to the other Spectrum

```

1 REM @ PAUL STANLEY 1983
5 GO SUB 9990
9 CLS : BORDER 0: PAPER 0: IN
K 7: CLS : RANDOMIZE 1
10 GO SUB 9000
11 INPUT "Station Number (1/2)
? ";st: IF st<>1 AND st<>2 THEN
GO TO 11
12 FORMAT "n";st: LET st1=3-st
15 LET sc1=0: LET sc2=0
20 GO SUB 8000
30 POKE 23458,0
100 LET y1=y: LET x1=x: IF INKE
Y$="x" THEN LET y=y+dyh: LET x=
x+dxh
105 IF INKEY$="z" THEN LET y=y
-dyh: LET x=x-dxh
110 IF INKEY$="m" THEN LET y=y
+dyv: LET x=x+dxv
115 IF INKEY$="k" THEN LET y=y
-dyv: LET x=x-dxv

```

```

117 IF INKEY$="1" THEN LET sc1
=sc1+1: PRINT AT 0,3;sc1;"-";sc2
120 IF y>19 THEN LET y=19
125 IF y<3 THEN LET y=3
130 IF x>29 THEN LET x=29
140 IF ATTR (y,x)>64 OR ATTR (y
+1,x)>64 OR ATTR (y,x+1)>64 OR A
TTR (y+1,x+1)>64 THEN GO SUB 50
0
150 PRINT AT y1,x1;" ";AT y1+1
,x1;" ";AT y,x;"EF";AT y+1,x;"G
H"
155 IF x=1 THEN GO TO 2000
160 IF st=1 THEN OPEN #4;"n";s
t1: INPUT #4;k$: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;INKEY$: CLO
SE #4
170 IF st=2 THEN OPEN #4;"n";s
t1: PRINT #4;INKEY$: CLOSE #4: O
PEN #4;"n";st1: INPUT #4;k$: CLO
SE #4
173 IF CODE k$>122 THEN GO TO
4000
200 PRINT AT y,x;"AB";AT y+1,x;
"CD": BEEP .003,15
210 LET c=c+1: IF c=25 THEN GO
SUB 1000
220 IF k$="1" THEN LET tx=tx-.
5: PRINT AT 2,INT tx; INK 4;"I7
": BEEP .01,10: IF INT tx=1 THEN
GO TO 3000
499 GO TO 100
500 IF ATTR (y,x)=70 OR ATTR (y
+1,x)=70 OR ATTR (y,x+1)=70 OR A
TTR (y+1,x+1)=70 THEN BEEP .04,
30: LET sc1=sc1+10: PRINT AT 0,3
;sc1;"-";sc2: RETURN
510 LET x=x1: LET y=y1: RETURN
520 IF sc1>sc2 THEN PRINT AT 7
,10; INK 6; PAPER 2; FLASH 1;" Y
OU WON! "

```

```

530 IF sc2>sc1 THEN PRINT AT 7
,10; FLASH 1; INK 6; PAPER 2;" YOU LOST "
532 IF sc2=sc1 THEN PRINT AT 7
,9; FLASH 1; INK 6; PAPER 2;" IT 'S A DRAW "
535 FOR f=1 TO 30: BEEP .02,INT
(RND*8)+10: NEXT f
537 LET A$=" PRESS ANY KEY T
O HAVE ANOTHER ACTION PACKED GAM
E OF THE TORTOISE AND THE HARE"
540 PRINT AT 10,2;A$( TO 29)
550 LET A$=A$(2 TO )+a$(1): PAU
SE 4
560 IF INKEY$<>" THEN BEEP .1
,10: GO TO 15
570 GO TO 540
1000 FOR f=1 TO 14: BEEP .03,f:
PRINT OVER 1;AT 0,10;"ALL CHANG
E!": NEXT f
1010 LET c=0: LET s=INT (RND*3)-
1: LET t=1: IF RND>.5 THEN LET
t=-t
1020 LET dyh=s: LET dxh=t-s: IF
ABS dxh=2 THEN LET dxh=dxh+s+s
1025 LET t=1: IF RND>.5 THEN LE
T t=-t
1030 LET dxv=s: LET dyv=t-s: IF
ABS dyv=2 THEN LET dyv=dyv+s+s
1040 RETURN
2000 IF st=1 THEN OPEN #4;"n";s
t1: INPUT #4;k$: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;"{": CLOSE
#4
2010 IF st=2 THEN OPEN #4;"n";s
t1: PRINT #4;"{": CLOSE #4: OPEN
#4;"n";st1: INPUT #4;k$: CLOSE
#4
2020 FOR f=1 TO 20: BEEP .04,f:
PRINT OVER 1;AT 10,10;"WELL DON
E!": NEXT f
2030 LET sc1=sc1+30: GO TO 20

```



```

3000 IF st=1 THEN OPEN #4;"n";s
t1: INPUT #4;k$: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;"}": CLOSE
#4
3010 IF st=2 THEN OPEN #4;"n";s
t1: PRINT #4;"}": CLOSE #4: OPEN
#4;"n";st1: INPUT #4;k$: CLOSE
#4
3015 PRINT AT 10,0;"Bad luck! Yo
ur hare was defeated"
3020 FOR f=1 TO 20: BEEP .05,f:
NEXT f
3040 GO TO 20
4000 IF k$="{ " THEN PRINT AT 10
,3;"Your Tortoise Was Beaten."
4010 IF k$="}" THEN PRINT AT 10
,0;"Your Tortoise Beat Beat Oppo
nent": LET sc1=sc1+30
4020 FOR f=1 TO 20: BEEP .05,f:
NEXT f
4030 GO TO 20
8000 CLS : FOR f=0 TO 31: PRINT
AT 1,f; INK RND*5+2;"M";AT 21,f;
"M": NEXT f
8010 FOR f=2 TO 20: PRINT AT f,3
1; INK RND*5+2;"M";AT f,0;"M"; P
APER 5;" ": NEXT f
8020 LET a$="FINISH": FOR f=1 TO
6: PRINT PAPER 5; INK 1;AT f*3
,1;a$(f): NEXT f
8030 FOR f=1 TO 15
8040 LET x=INT (RND*27)+2: LET y
=INT (RND*17)+3
8060 PRINT INK 4; BRIGHT 1;AT y
,x;"K";AT y+1,x;"L": NEXT f
8070 LET tx=29: PRINT AT 2,tx; I
NK 4;"IJ"
8082 FOR f=1 TO 5
8084 LET y=INT (RND*17)+3: LET x
=INT (RND*28)+2
8086 IF ATTR (y,x)>64 THEN GO T
O 8084
8088 PRINT BRIGHT 1; INK 6;AT y
,x;"N": NEXT f

```

```

8090 LET y=19: LET x=29: LET y1=
y: LET x1=x
8100 LET dyh=0: LET dxh=1: LET d
yv=1: LET dxv=0
8150 LET c=0
8200 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;sc1: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;sc2: CLOSE
#4
8210 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;sc2: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;sc1: CLOSE
#4
8230 PRINT AT 0,0; INK 6;"SC:";
INK 7;sc1;"-";sc2
8300 IF sc1>=300 OR sc2>=300 THE
N GO TO 520
8999 RETURN
9000 FOR f=USR "a" TO USR "n"+7:
READ q: POKE f,q: NEXT f
9010 DATA 0,1,14,59,127,63,3,1,0
,248,4,248,132,124,128,224,31,3,
6,12,0,0,3,7,240,248,254,253,126
,48,240,240
9020 DATA 1,14,59,127,63,3,1,31,
248,4,248,132,124,128,224,240,3,
6,12,0,0,3,7,0,248,254,253,126,4
8,240,240,0
9030 DATA 0,3,6,125,222,127,6,12
,224,88,172,86,170,255,12,24
9040 DATA 24,60,90,24,60,90,153,
60,90,153,60,90,153,24,24,24
9060 DATA 127,65,93,85,93,65,127
,0,0,3,14,60,124,56,16,0
9100 FOR f=0 TO 31: PRINT INK R
ND*5+1;AT 12,f;"M";AT 13,f;"M":
NEXT f
9110 FOR f=29 TO 0 STEP -1
9120 BEEP .04,-10: PRINT AT 10,f
;"AB ";AT 11,f;"CD "
9130 BEEP .04,-10: PRINT AT 10,f
;"EF";AT 11,f;"GH"
9140 NEXT f: PRINT AT 10,0;" ";
AT 11,0;" "

```

```

9150 LET a$="THE TORTOISE & THE
HARE @ PAUL STANLEY"
9160 FOR f=29 TO 0 STEP -1
9170 PRINT INK 4;AT 11,f;"I";
PAPER 2; INK 6;a$( TO 30-f)
9175 BEEP .1,-30: NEXT f
9176 PRINT INK 4;AT 11,0;"J"; P
APER 2; INK 6;a$( TO 31): BEEP .
1,-30
9180 FOR f=1 TO 40
9185 LET e=LEN a$: IF e>32 THEN
LET e=32
9190 PRINT AT 11,0; PAPER 2; INK
6;a$( TO e); PAPER 0;" " AND e<
32
9200 LET a$=a$(2 TO )
9210 BEEP .1,-30: NEXT f
9220 CLS : PRINT "In this game y
ou control a hare on your screen
, and a tortoise on your oppone
nt's screen!"
9230 PRINT '"The televisions sho
uld be placed so that you can see
both. To advance the tortois
e you press"; INK 4;" L"
9240 PRINT '"HARE CONTROLS:"' IN
K 6;" Z.....LEFT"" X.....RIG
HT"" M.....DOWN"" K.....
UP"
9250 PRINT '"If your hare beats
your opponent then you get 30 poi
nts. You get 10 points for picki
ng a carrot. The winner is the f
irst to 300 points."
9700 PRINT '"PRESS ANY KEY TO RE
AD MORE."
9800 IF INKEY$="" THEN GO TO 98
00
9810 CLS : PRINT '"You cannot go
through trees; you must go round
them. Nor can you move the tort
oise and your hare at the same t
ime."

```

```
9815 PRINT "Remember that the t  
ortoise can only go a fraction  
of the speed of the hare, but yo  
u get points for moving him."  
9820 PRINT "There is another fe  
ature which adds interest, but  
you will have to wait to see what  
it is."  
9830 PRINT "PRESS ANY KEY TO ST  
ART.."  
9840 IF INKEY$="" THEN GO TO 98  
40  
9900 CLS : RETURN  
9990 CLS : PRINT "Type on other  
Spectrum"" LOAD *""n";1 <e  
nter>": SAVE *""n";1;"a" LINE 9  
9991 RETURN
```

# 16 Blackjack

## **Hardware requirements**

Two 48K Spectrums on ZX network

Two televisions

This is a card game which may not be familiar to you, but since the rules are simple it's easy to master. Play starts with the computer shuffling the cards and dealing seven to each of the two players.

The player who starts may place any card of any value. All subsequent selections must be of the same suit as the last, or the same value. If certain cards are played, then new rules apply. If your opponent plays any 2, then you must take two cards from the 'pack'. If a black-jack is played (the jack of clubs or spades), you must take five cards. This is the most severe consequence of any card. Other special cards are the ace and king. An ace means that your opponent must miss a go, and the appearance of a king enables you to select a new suit for subsequent play. When selecting a suit it is obviously wise to choose the suit of which you have most cards, or a black suit if you have the jack of that suit. These special rules are permanently displayed on your screen, so there's no need to commit them to memory.

To place a card, players simply enter the number which corresponds to the appropriate card in the hand displayed at the bottom of the screen. All entries are checked for validity, so if you cannot play a card, entering 0 will pick up another card from the 'pack'. To make it easier for you the computer arranges your cards into suit and numerical order.

To win a game you must place all your cards, and if all the cards in the pack are used up the Spectrum declares a draw.

## **Program Notes:**

1-110 Call the initial routines

120-190 Ensure that one Spectrum is transmitting and one is receiving

1000-1035 Enter a card number and check its validity

1040-1065 Transfer the card to the top of the screen  
 1070-1100 Decide if the card is a 2, king, blackjack or an ace  
     1200 Transmit the card's number and suit  
         Transmit the special variable which decides what moves  
         should be made  
 1205-1999 Act on the card played  
 2000-2020 Receive the card's number and suit and the special variable  
 2030-2510 Act on the card played  
 2900-2999 Print the current suit and number of cards left  
         Check that the pack is not empty  
 4000-4520 Print the final score  
         Wait for a key to be pressed to restart the game  
 5000-5050 Take a card if a blackjack or 2 was played  
 5500-5540 Take a card if you cannot play one  
 6000-6050 Select a new suit  
 6800-6910 Send the 'pack' from station one to station two  
 7000-7040 Bubble sort to place cards in suit and numerical order  
 7050-7999 Print the cards  
 8000-8160 Generate and shuffle the cards  
 8170-8999 Print the screen display  
 9000-9050 Define the graphics characters  
 9100-9170 Instructions  
     9999 Send the program to the other computer

```

1 REM @ Paul Stanley 1984
5 CLS : GO SUB 9999
10 RANDOMIZE : CLS : BORDER 0:
PAPER 0: INK 7: CLS : POKE 2375
0,0
20 GO SUB 9000
50 INPUT "Station No (1 or 2)
?";st: IF st<>1 AND st<>2 THEN
GO TO 50
60 FORMAT "n";st: LET st1=3-st
100 GO SUB 8000
110 GO SUB 7000
130 GO SUB 1000*st
160 GO SUB 1000*st1
190 GO TO 130
1000 BEEP .1,10: LET os=s: INPUT
"CARD ?(0 IF YOU CAN'T GO) ";c
1005 IF c>cards THEN BEEP .3,-2
0: GO TO 1000
  
```

```

1010 IF c=0 THEN GO SUB 5500: G
O TO 1200
1015 IF s=0 THEN GO TO 1040
1020 IF h(c,1)=s THEN GO TO 104
0
1030 IF h(c,2)=v THEN GO TO 104
0
1035 BEEP .3,-20: GO TO 1000
1040 LET d=2*(h(c,1)=2)+2*(h(c,1
)=4)
1045 LET v=h(c,2): LET s=h(c,1)
1047 LET os=s
1050 PRINT AT 1,7; PAPER 7; INK
0;"F 6";AT 2,7; INK d;b$(v);" "
;AT 3,7;a$(s);" ";a$(s);AT 4,7;"
";b$(v); INK 0;AT 5,7;"I H"
1060 LET sp=0
1065 LET cards=cards-1: IF cards
=0 THEN LET sp=5: GO TO 1200
1070 IF h(c,2)=2 THEN LET sp=1
1080 IF h(c,2)=1 THEN LET sp=2
1090 IF h(c,2)=11 THEN IF h(c,1
)=1 OR h(c,1)=3 THEN LET sp=3
1100 IF h(c,2)=13 THEN GO SUB 6
000
1110 LET h(c,1)=h(cards+1,1): LE
T h(c,2)=h(cards+1,2)
1200 OPEN #4;"n";st1: PRINT #4;s
: PRINT #4;os: PRINT #4;v: PRINT
#4;sp: PRINT #4;p: CLOSE #4
1205 IF sp=5 THEN GO TO 4100
1207 IF p=53 THEN GO TO 4400
1210 GO SUB 7000
1212 IF sp=1 THEN LET p=p+2: IF
p>=53 THEN GO TO 4400
1214 IF sp=3 THEN LET p=p+5: IF
p>=53 THEN GO TO 4400
1220 GO SUB 2900
1900 IF sp=2 THEN GO TO 1000
1999 RETURN
2000 PRINT #1;"Please wait for y
our opponent."
2010 FOR f=1 TO 300: NEXT f

```

```

2020 OPEN #4;"n";st1: INPUT #4;s
: INPUT #4;ns: INPUT #4;v: INPUT
#4;sp: INPUT #4;p: CLOSE #4
2030 IF sp=9 THEN PRINT #1;"You
r opponent had to pass.": GO TO
2500
2033 IF sp=5 THEN GO TO 4000
2035 LET d=2*(ns=2)+2*(ns=4)
2040 PRINT AT 1,7; PAPER 7; INK
0;"F 6";AT 2,7; INK d;b$(v);" "
;AT 3,7;a$(ns);" ";a$(ns);AT 4,7
;" ";b$(v); INK 0;AT 5,7;"I H"
2050 IF sp=1 THEN PRINT #1;"You
have got to take 2 cards.": FOR
f=1 TO 2: GO SUB 5000: NEXT f
2060 IF sp=3 THEN PRINT #1;"You
have to take 5 cards!": FOR f=1
TO 5: GO SUB 5000: NEXT f
2070 IF sp=4 THEN PRINT #1;"You
r opponent chose ";a$(s);" suit"
2080 IF sp=2 THEN PRINT #1;"You
have to miss a go."
2210 GO SUB 2900: GO SUB 7000
2500 FOR f=1 TO 150: NEXT f: INP
UT ""
2510 IF sp=2 THEN GO TO 2000
2900 PRINT AT 3,21;"CLUBS " AND
S=1;"HEARTS" AND S=2;"SPADES" AN
D S=3;"D'MNDS" AND S=4
2910 PRINT AT 1,27;53-p;" "
2920 IF p=53 THEN GO TO 4400
2999 RETURN
4000 CLS : PRINT AT 10,0;"Your o
pponent has placed all hiscards
and is the winner."
4010 GO TO 4500
4100 CLS : PRINT AT 10,0;"You ha
ve placed all your cards and ar
e therefore the winner."
4110 GO TO 4500
4400 CLS : PRINT AT 10,0;"There
are no cards left and so it is
a draw."

```



```

4500 FOR f=1 TO 20: BEEP .1,INT
(RND*10): NEXT f
4510 INPUT "Press ENTER to play
again."; LINE p$
4520 CLS : GO TO 50
5000 IF cards=15 OR p=53 THEN R
ETURN
5010 LET cards=cards+1
5020 LET h(cards,1)=p(p,1)
5030 LET h(cards,2)=p(p,2)
5040 LET p=p+1
5050 RETURN
5500 LET sp=9: LET cards=cards+1
5510 IF p=53 OR cards=16 THEN L
ET cards=cards-1: RETURN
5520 LET h(cards,1)=p(p,1): LET
h(cards,2)=p(p,2)
5530 LET p=p+1
5540 RETURN
6000 PRINT #1;"When the suit app
ears press 0"
6010 LET s=2: LET s$="HEARTS": F
OR f=1 TO 6: PRINT AT f,13;s$(f)
: NEXT f: GO SUB 6500
6015 IF r=1 THEN GO TO 6100
6020 LET s=3: LET s$="SPADES": F
OR f=1 TO 6: PRINT INK 6;AT f,1
3;s$(f): NEXT f: GO SUB 6500
6025 IF r=1 THEN GO TO 6100
6030 LET s=4: LET s$="D'MNDS": F
OR f=1 TO 6: PRINT AT f,13;s$(f)
: NEXT f: GO SUB 6500
6035 IF r=1 THEN GO TO 6100
6040 LET s=1: LET s$="CLUBS ": F
OR f=1 TO 6: PRINT INK 6;AT f,1
3;s$(f): NEXT f: GO SUB 6500
6045 IF r=1 THEN GO TO 6100
6050 GO TO 6010
6100 BEEP .2,10
6110 PRINT AT 3,21;s$
6120 FOR f=1 TO 6: PRINT AT f,13
;" ": NEXT f
6130 INPUT ""
6140 LET sp=4: RETURN

```

```

6500 BEEP .1,10
6510 FOR f=1 TO 100
6520 LET r=0: IF INKEY$="0" THEN
  LET r=1: RETURN
6530 NEXT f: RETURN
6800 OPEN #4;"n";st1: PRINT #4;p
(f,1): PRINT #4;p(f,2): CLOSE #4
6810 RETURN
6900 OPEN #4;"n";st1: INPUT #4;p
(f,1): INPUT #4;p(f,2): CLOSE #4
6910 RETURN
7000 FOR f=14 TO 21: PRINT AT f,
0;c$: BEEP .03,f: NEXT f
7005 LET c=0: FOR f=1 TO cards-1
7010 IF h(f+1,1)<h(f,1) THEN LE
T a=h(f+1,1): LET b=h(f+1,2): LE
T h(f+1,1)=h(f,1): LET h(f+1,2)=
h(f,2): LET h(f,1)=a: LET h(f,2)
=b: LET c=1
7020 IF h(f+1,1)=h(f,1) AND h(f+
1,2)<h(f,2) THEN LET a=h(f+1,1)
: LET b=h(f+1,2): LET h(f+1,1)=h
(f,1): LET h(f+1,2)=h(f,2): LET
h(f,1)=a: LET h(f,2)=b: LET c=1
7030 NEXT f
7040 IF c=1 THEN GO TO 7005
7050 FOR f=1 TO cards
7055 LET c=2*(h(f,1)=2)+2*(h(f,1
)=4)
7060 LET x=f*2-2
7070 LET e=(f/2=INT (f/2))-6*(f=
1)
7075 PRINT INK 5;AT 14+(f/2=INT
(f/2)),x;f
7080 PRINT AT 17,x; PAPER 7-(f/2
=INT (f/2)); INK 6+e;"F"; INK 0;
" 6";AT 18,x; INK c;b$(h(f,2));"
";AT 19,x;a$(h(f,1));" ";a$(h(
f,1));AT 20,x;" ";b$(h(f,2)); I
NK 6+e;AT 21,x;"I"; INK 0;" H"
7090 NEXT f
7999 RETURN

```

```

8000 DIM h(15,2): DIM p(52,2): L
ET a$="ABDC": LET b$="A23456789E
JQK"
8002 DIM c$(32)
8005 LET cards=7: PRINT FLASH 1
;AT 10,10;"PLEASE WAIT"
8010 FOR f=1 TO 4: FOR q=1 TO 13
8020 LET c=f*13-13+q
8030 LET p(c,1)=f: LET p(c,2)=q
8040 NEXT q: NEXT f
8100 FOR f=1 TO 50
8110 LET a=INT (RND*52)+1
8120 LET b=INT (RND*52)+1
8130 LET c=p(a,1): LET d=p(a,2)
8140 LET p(a,1)=p(b,1): LET p(a,
2)=p(b,2)
8150 LET p(b,1)=c: LET p(b,2)=d
8160 NEXT f
8165 FOR f=1 TO 52: GO SUB 6700+
100*st: NEXT f
8170 NEXT f: FOR f=1 TO 7: LET h
(f,1)=p(f*2+st-2,1): LET h(f,2)=
p(f*2+st-2,2)
8200 NEXT f
8210 LET P=15: LET v=0: LET S=v
8500 FOR f=7 TO 12: PRINT AT f,0
; PAPER 2;c$: BEEP .05,f: NEXT f
8510 PRINT PAPER 2; INK 6;AT 8,
1;"2...OPPONENT TAKES 2"
8520 PRINT PAPER 2; INK 6;" A..
.OPPONENT MISSES NEXT GO"" K...
CHOOSE NEW SUIT"" BJ..OPPONENT
TAKES 5 CARDS"
8530 PRINT INK 6;AT 1,15;"CARDS
LEFT: "; INK 7;53-P
8540 PRINT INK 6;AT 3,15;"SUIT:
"
8999 RETURN
9000 FOR f=USR "a" TO USR "i"+7:
READ q: POKE f,q: NEXT f
9010 DATA 28,28,107,127,107,8,8,
28,54,119,127,127,127,62,28,8
9020 DATA 8,28,62,127,62,28,8,0,
8,28,62,127,127,127,42,8

```

```

9030 DATA 0,70,201,73,73,73,230,
0
9040 DATA 240,192,128,128,0,0,0,
0,15,3,1,1,0,0,0,0
9050 DATA 0,0,0,0,1,1,3,15,0,0,0
,0,128,128,192,240
9100 PRINT PAPER 4; INK 0;" B L
A C K J A C K @ P STANLEY "
9110 PRINT "This is a 2 player c
ard game. It is simple to unde
rstand and your cards are arran
ged in order to help you."
9120 PRINT "Certain cards which
you play have dire consequen
ces on your opponent; most nota
ble of these is when you play a
black jack- your opponent then
has to pick up 5 cards."
9130 PRINT "You can play a card
if it is the same suit or the sa
me number as the one shown at th
e top of the screen."
9140 PRINT "The winner is the f
irst to get rid of his cards. I
f the pack is exhausted, then
there is no winner."
9150 INPUT "Press ENTER to start
."; LINE p$
9170 CLS : RETURN
9999 PRINT "Type on other Spectr
um: "" LOAD *""n";1 <enter>":
SAVE *""n";1;"a" LINE 10: RETURN

```

# 17 Pontoon

## **Hardware requirements**

Two 48K Spectrums on ZX network

Two televisions

This is a computerised version of the ever popular Pontoon, or for those of you of a Continental bent, Vingt-et-un. In the final analysis the game is largely a matter of luck. This said, it's also great fun to play, particularly since this version gives you a chance to make money (even though you lose it when the computer is switched off)!

After the pack has been shuffled by the computer, two cards are dealt to each player, one face up and one face down. You are then told to make a bet. You start the game with a generous stash of £1000, of which you can bet any amount in five pound units. If you win a hand, you receive the amount you placed and your money back.

After placing a bet, the value of your second card is revealed. The object of the game is to add cards until your total is as close to twenty-one as your nerve allows.

If your cards total more than twenty-one you lose the round. Court cards count as ten points and aces count as one or eleven. If you have a total of twenty-one with two cards (an ace and a court card or ten) then you have Pontoon. This hand is unbeatable, and if your opponent also gets Pontoon, then you both get your winnings. The next best hand is one containing five cards. Even if their total is lower than your opponent's total, it will still be a winning hand. If neither of these hands occur then the winner will be the player with the highest total.

Although the computers will continue to deal hands until somebody's money runs out, it's a good idea to agree a time-limit beforehand, so that if nobody becomes bankrupt the winner is the person with the most money when the time runs out.

This game is fun to play, with amusing graphics for the court cards. Finally, do not be alarmed if the game should be interrupted after a time, it is merely the computer shuffling the cards!

**Program Notes:**

- 5-100 Call the initial routines
- 110-270 Ensure that one computer is transmitting and the other receiving. Check that the pack does not need shuffling
- 1000-1200 Enter whether another card is requested or not  
Check that the score does not exceed 21
- 1500 Send the number of cards left and whether you have exceeded 21 or stuck.
- 1520-1999 Act on the cards played  
Print the total
- 2000-2010 Receive the number of cards remaining
- 2020-2999 Print the result of your opponent's move
- 3000-3140 Decide whose score is the larger, if anybody has Pontoon or five cards  
Print the result  
Adjust the money
- 5000-5060 Print a card
- 6000-6060 Total the score of somebody's hand
- 6500-6600 Check for bankruptcy  
Wait for a key to be pressed
- 7500-7540 Send the amount of money you have  
Receive the amount your opponent has  
Print the money totals
- 8000-8120 Generate and shuffle the pack
- 8140-8150 Station one sends the pack to station two
- 8160-8230 Deal two cards, one face down
- 8300-8999 Input a bet and check its validity
- 9000-9095 Define the graphics characters
- 9100-9500 Instructions
- 9990-9994 Send the program to the other Spectrum

```

5 GO SUB 9990
10 RANDOMIZE : CLS : BORDER 0:
PAPER 0: CLS : INK 7: POKE 2375
0,0
20 POKE 23658,8
30 GO SUB 9000
50 INPUT "Station? (1 or 2) ";
st: IF st<>1 AND st<>2 THEN GO
TO 50
60 FORMAT "n";st: LET st1=3-st
100 LET m1=1000: LET m2=m1
110 GO SUB 8000
130 GO SUB 1000*st

```

```

135 PRINT AT 3,9;"
140 IF k$="B" AND st=2 THEN LE
T m1=m1+2*m: GO TO 175
145 IF k$="B" THEN GO TO 175
160 GO SUB 1000*st1
165 PRINT AT 3,9;"
170 IF k$="B" AND st=1 THEN LE
T m1=m1+2*m
175 GO SUB 3000
180 IF m1=0 OR m2=0 THEN GO TO
6500
185 IF p>40 THEN GO SUB 8000:
GO TO 200
190 CLS : GO SUB 8160
200 GO SUB 1000*st1
205 PRINT AT 3,9;"
210 IF k$="B" AND st=1 THEN LE
T m1=m1+2*m: GO TO 245
215 IF k$="B" THEN GO TO 245
220 GO SUB 1000*st
235 PRINT AT 3,9;"
240 IF k$="B" AND st=2 THEN LE
T m1=m1+2*m
245 GO SUB 3000
246 IF m1=0 OR m2=0 THEN GO TO
6500
250 IF p>40 THEN GO SUB 8000:
GO TO 130
260 CLS : GO SUB 8160
270 GO TO 130
1000 PRINT PAPER 6; INK 1; BRIG
HT 1; AT 0,0;"T..TAKE ANOTHER CAR
D S..STICK?"
1010 LET K$=INKEY$: IF K$<>"T" A
ND K$<>"S" THEN GO TO 1010
1020 BEEP .1,10: IF K$="S" THEN
GO TO 1500
1100 LET cards=cards+1
1110 LET h(cards,1)=VAL a$(p)
1115 LET h(cards,2)=VAL b$(p*2-1
TO p*2): LET p=p+1: LET x=x+5
1120 LET s=h(cards,1): LET c=h(c
ards,2): GO SUB 5000
1130 GO SUB 6000

```

```

1140 IF k$<>"B" AND cards<5 THEN
  BEEP .1,10: GO TO 1000
1200 IF cards=5 AND t<22 THEN L
  ET k$="S"
1500 OPEN #4;"n";st1: PRINT #4;k
  $: PRINT #4;p: CLOSE #4
1520 IF k$="B" THEN PRINT FLAS
  H 1;AT 3,11;" B U S T "
1530 IF k$="S" THEN PRINT FLAS
  H 1;AT 3,10;"TOTAL :";t: IF t=21
  AND cards=2 THEN PRINT FLASH
  1;AT 3,10;" PONTOON! "
1540 FOR f=1 TO 10: BEEP .1,INT
  (RND*10): NEXT f
1999 RETURN
2000 PRINT AT 0,0;" PLEASE WAIT
  FOR YOUR OPPONENT. "
2010 OPEN #4;"n";st1: INPUT #4;k
  $: INPUT #4;p: CLOSE #4
2020 IF k$="B" THEN PRINT AT 3,
  11; FLASH 1;" B U S T "
2030 IF k$="S" THEN PRINT AT 3,
  10; FLASH 1;"STICKING"
2040 FOR f=1 TO 10: BEEP .1,INT
  (RND*10): NEXT f
2999 RETURN
3000 IF k$="B" THEN GO SUB 7500
  : RETURN
3010 IF st=1 THEN OPEN #4;"n";s
  t1: PRINT #4;cards: PRINT #4;t:
  CLOSE #4: OPEN #4;"n";st1: INPUT
  #4;cards1: INPUT #4;t1: CLOSE #
  4
3020 IF st=2 THEN OPEN #4;"n";s
  t1: INPUT #4;cards1: INPUT #4;t1
  : CLOSE #4: OPEN #4;"n";st1: PRI
  NT #4;cards: PRINT #4;t: CLOSE #
  4
3030 LET w=0: IF cards=2 AND t=2
  1 THEN LET w=1: PRINT AT 3,0;"Y
  OUR PONTOON MEANS THAT YOU WIN!"
  : GO TO 3110

```



```

3040 IF cards1=2 AND t1=21 THEN
  LET w=2: PRINT AT 3,0;"YOUR OPP
  ONENT GOT PONTOON & WINS": GO TO
  3110
3050 IF cards=5 THEN LET w=1: P
  RINT AT 3,0;"YOU HAVE 5 CARDS AN
  D WIN": GO TO 3110
3060 IF cards1=5 THEN LET w=2:
  PRINT AT 3,0;"YOUR OPPONENT HAS
  5 CARDS & WINS": GO TO 3110
3080 IF t>t1 THEN LET w=1: PRIN
  T AT 3,0;"YOU HAVE A GREATER SCO
  RE AND WIN"
3090 IF t<t1 THEN LET w=2: PRIN
  T AT 3,0;"YOU HAVE A SMALLER SCO
  RE & LOSE"
3100 IF t=t1 THEN PRINT AT 3,0;
  "YOU HAVE EQUAL SCORES - A DRAW!
  ": LET m1=m1+m
3110 FOR f=1 TO 20: BEEP .1,INT
  (RND*10): NEXT f
3120 IF w=1 THEN LET m1=m1+2*m
3130 GO SUB 7500
3140 RETURN
5000 PRINT INK 0; PAPER 7;AT 14
  ,x;"F G";AT 20,x;"I H"
5010 LET i=2*(s=2)+2*(s=4)
5020 PRINT AT 15,x; PAPER 7; INK
  i;n$(c);" ";AT 16,x;s$(s);"
  ";AT 17,x;" ";AT 18,x;" ";
  s$(s);AT 19,x;" ";n$(c)
5030 IF c=12 THEN PRINT INK 1;
  PAPER 7;AT 16,x+1;"QR";AT 17,x+
  1;"MN";AT 18,x+1;"S7"
5040 IF c=11 OR c=13 THEN PRINT
  INK 1; PAPER 7;AT 16,x+1;"KL";
  AT 17,x+1;"MN";AT 18,x+1;"OP"
5060 RETURN
6000 LET a=0: LET t=a: FOR f=1 T
  O cards
6010 LET t=t+(h(f,2) AND h(f,2)<
  11)+(10 AND h(f,2)>10)
6020 IF h(f,2)=1 THEN LET t=t+1
  0: LET a=a+1

```

```

6030 NEXT f
6040 IF t<22 THEN RETURN
6050 IF a THEN LET a=a-1: LET t
=t-10: GO TO 6040
6060 LET k$="B": RETURN
6500 CLS
6510 FOR f=0 TO 10 STEP .25: BEE
P .03,f: NEXT f
6520 IF m1=0 THEN PRINT AT 10,7
;"You are bankrupt!"
6530 IF m2=0 THEN PRINT AT 8,3;
"Your opponent is bankrupt!"
6540 PRINT AT 20,0;"Press any ke
y to play again..."
6600 PAUSE 0: CLS : GO TO 100
7500 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;m1: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;m2: CLOSE #
4
7510 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;m2: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;m1: CLOSE #
4
7520 PRINT AT 10,10;m1;"      "
7530 PRINT INK 6;AT 12,19;m2;"
      "
7540 RETURN
8000 CLS : DIM h(6,2): DIM a$(52
): DIM b$(104)
8005 PRINT FLASH 1;AT 10,9;"PLE
ASE WAIT"
8010 LET s$="ABDC": LET n$="A234
56789EJQK"
8020 FOR f=1 TO 4: FOR q=1 TO 13
8030 LET a=f*13-13+q
8040 LET a$(a)=STR$ f: LET b$(a*
2-1 TO a*2)=STR$ q
8050 NEXT q: NEXT f
8060 FOR f=1 TO 50
8070 LET a=INT (RND*52)+1
8075 LET b=INT (RND*52)+1
8080 LET c$=a$(a): LET d$=b$(a*2
-1 TO a*2)

```

```

8090 LET a$(a)=a$(b): LET b$(a*2
-1 TO a*2)=b$(b*2-1 TO b*2)
8100 LET a$(b)=c$: LET b$(b*2-1
TO b*2)=d$
8110 NEXT f
8120 LET p=1
8140 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;a$: PRINT #4;b$: CL
OSE #4
8150 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;a$: INPUT #4;b$: CL
OSE #4
8160 FOR f=1 TO 2
8170 LET h(f,1)=VAL a$(p+st-1):
LET h(f,2)=VAL b$((p+st)*2-3 TO
(p+st)*2-2)
8180 LET p=p+2: NEXT f
8190 LET s=h(1,1): LET c=h(1,2)
8195 LET cards=2: LET x=0
8200 GO SUB 5000
8210 PRINT AT 14,5; INK 0; PAPER
7;"F 6";AT 20,5;"I H": PRINT
AT 10,10;" "
8220 FOR f=15 TO 19: PRINT AT f,
5; PAPER 7;" "; INK 3; PAPER 6;"
JJ"; PAPER 7;" ": NEXT f
8230 LET X=5
8300 PRINT AT 10,0;"YOU HAVE £";
M1
8310 PRINT INK 6;AT 12,0;"YOUR
OPPONENT HAS £";M2
8320 INPUT "How much do you want
to bet?"m
8330 IF m>m1 OR m<5 THEN BEEP .
2,-20: GO TO 8300
8340 IF m/5<>INT (m/5) THEN BEE
P .2,-20: GO TO 8320
8350 LET s=h(2,1): LET c=h(2,2):
GO SUB 5000
8355 LET m1=m1-m
8360 GO SUB 7500
8370 GO SUB 6000
8999 RETURN

```

```

9000 LET o=0: FOR f=USR "a" TO U
SR "t"+7: READ q: POKE f,q: NEXT
f
9010 DATA 28,28,107,127,107,8,8,
28,54,119,127,127,127,62,28,8
9020 DATA 8,28,62,127,62,28,8,0,
8,28,62,127,127,127,42,8
9030 DATA 0,70,201,73,73,73,230,
0
9040 DATA 240,192,128,128,0,0,0,
0,15,3,1,1,0,0,0,0
9050 DATA 0,0,0,0,1,1,3,15,0,0,0
,0,128,128,192,240
9060 DATA 165,90,153,102,102,153
,90,165,10,15,15,15,7,8,10,72
9070 DATA 160,224,224,224,192,32
,160,32,104,107,100,115,63,31,7,
7,32,160,64,128,240,248,220,204
9080 DATA 7,15,15,14,28,28,152,1
20,196,224,224,224,112,112,50,60
9090 DATA 1,10,10,39,29,72,122,7
2,0,160,168,244,72,48,164,48
9095 DATA 15,15,31,31,63,63,4,12
,228,224,240,240,248,248,64,96
9100 PRINT PAPER 7; INK 2;"B";
INK 0;" A PONTON @ PAUL STANL
EY D "; INK 2;"C"
9110 PRINT "This is a gambling
game for 2 players. The object
is to get the total of your c
ards to total as near to 21 as po
ssible but not over 21."
9120 PRINT "Aces can count as 1
or 11, and picture cards (wait
till you see these!!) count as 1
0."
9130 PRINT "If you get PONTON
(a picture & an ace) that will b
eat any other hand. Five cards wi
ll beat any- thing else, even if
their total is less than your o
pponent's."

```

```
9140 PRINT "You are dealt 2 cards initially, one face down. You then have to place a bet (only multiples of £5 are permitted.)"
9150 INPUT "Press ENTER to start ."; LINE r$
9500 RETURN
9990 CLS : PRINT "Type on other Spectrum:''' LOAD *\"n\"";1 <enter>"
9992 SAVE *\"n\";1;"a" LINE 10
9994 RETURN
```

# 18 Chase

## **Hardware requirements**

Two 16/48K Spectrums on ZX network

Two televisions

This is a tactical game for two players, which tests the player's capacity for quick thinking and logical strategy. Each controls a small spacecraft which moves continuously, but can be steered by pressing one of the cursor keys.

The television displays both crafts, and the trail left behind your opponent's. In other words, your spaceship will leave a trail on your opponent's screen but not on your own! The object of the game is to avoid crashing into either your opponent's (or your own) trail or the side wall, but at the same time create a trail with your craft in an effort to trap your opponent.

After a trail has been on the screen for a while it begins to disintegrate, thus enabling a ship to pass safely through one of the breaks in it. This often turns out to be your safety-line in times of need, so if you become trapped you should avoid a collision and hope that a gap will appear facilitating an escape.

## **Program Notes:**

- 1-105 Call the initial routines
- 110-199 Ensure that one computer is receiving and one transmitting  
Print a blank space
- 1000 Receive the key pressed by your opponent
- 1010-1200 Check that your opponent has not crashed  
Move opponent's ship  
Print the trail and ship
- 2000-2005 Send a key which you press
- 2010-2999 Move ship  
Check that ship has not crashed  
Print ship

- 4000-4010 Receive keypress (otherwise both computers would be transmitting together)  
 Send a header so that your opponent knows that you have crashed
- 4030-4920 Crash the ship  
 Wait for a key to play again
- 8000-8070 Set up the screen
- 8080-8999 Ensure that both ships start together
- 9000-9050 Define the graphics characters
- 9100-9300 Title page
- 9305-9989 Instructions
- 9990 Send the program to the other computer

```

1 REM @ Paul Stanley 1984
5 GO SUB 9990
10 CLS : BORDER 0: PAPER 0: IN
K 7: CLS : POKE 23750,0
20 POKE 23658,8
50 GO SUB 9000
60 INPUT "Station Number (1 or
2)?";st
70 IF st<>1 AND st<>2 THEN GO
TO 60
80 FORMAT "n";st: LET st1=3-st
100 GO SUB 8000
105 FOR f=15 TO 20 STEP .2: BEE
P .02,f: BEEP .02,20-f: NEXT f
110 GO SUB 1000*st
120 GO SUB 1000*st1
130 PRINT AT INT (RND*20)+1,INT
(RND*30)+1;" "
199 GO TO 110
1000 OPEN #4;"n";st1: INPUT #4;k
$: CLOSE #4
1010 IF k$="Q" THEN GO TO 4400
1015 IF k$="" THEN LET k$=1$
1020 IF 1$="M" THEN LET p$=("F"
AND (k$="M" OR k$="K"))+("I" AN
D k$="X")+("J" AND k$="Z")
1030 IF 1$="K" THEN LET p$=("F"
AND (k$="M" OR k$="K"))+("G" AN
D k$="X")+("H" AND k$="Z")

```

```

1040 IF 1$="Z" THEN LET p$=("E"
AND (k$="Z" OR k$="X"))+("G" AN
D k$="M")+("I" AND k$="K")
1050 IF 1$="X" THEN LET p$=("E"
AND (k$="Z" OR k$="X"))+("H" AN
D k$="M")+("J" AND k$="K")
1052 LET o$=("A" AND k$="K")+("B"
AND k$="M")+("C" AND k$="X")+("D"
AND k$="Z")
1055 LET 1$=k$: LET y1=y: LET x1
=x
1060 LET y=y+(k$="M")-(k$="K")
1070 LET x=x+(k$="X")-(k$="Z")
1080 PRINT AT y1,x1; BRIGHT 1; I
NK 3;p$; INK 6;AT y,x;o$
1200 RETURN
2000 LET k$=INKEY$
2005 OPEN #4,"n",st1: PRINT #4;k
$: CLOSE #4
2007 IF k$="" THEN LET k$=m$
2010 LET p1=p: LET q1=q
2020 LET p=p+(k$="X")-(k$="Z")
2030 LET q=q+(k$="M")-(k$="K")
2040 LET o$=("A" AND k$="K")+("B"
AND k$="M")+("C" AND k$="X")+("D"
AND k$="Z")
2042 LET m$=k$
2045 IF ATTR (q,p)>64 THEN GO T
O 4000
2050 PRINT AT q1,p1;" ";AT q,p;o
$
2999 RETURN
4000 PRINT AT q1,p1;" ";AT q,p;o
$
4005 OPEN #4,"n";st1: INPUT #4;k
$: CLOSE #4
4010 OPEN #4,"n";st1: PRINT #4;"
Q": CLOSE #4
4020 FOR F=0 TO -30 STEP -.5: BE
EP .04,F: NEXT F
4030 PRINT FLASH 1;AT 10,5;"BAD
LUCK, YOU CRASHED!"
4040 GO TO 4900

```



```

4400 FOR F=10 TO 40 STEP .5: BEE
P .04,F: NEXT F
4500 PRINT FLASH 1;AT 10,5;"YOU
R OPPONENT CRASHED!"
4900 PRINT AT 18,2;"PRESS ANY KE
Y TO PLAY AGAIN."
4910 IF INKEY$="" THEN GO TO 49
10
4920 CLS : GO TO 100
8000 LET a$="EFGHIJ"
8010 FOR f=0 TO 21: PRINT BRIGH
T 1; PAPER 4; INK 0;AT f,0;a$(IN
T (RND*6)+1);AT f,31;a$(INT (RND
*6)+1): NEXT f
8020 FOR f=1 TO 30: PRINT BRIGH
T 1; PAPER 4; INK 0;AT 0,f;a$(IN
T (RND*6)+1);AT 21,f;a$(INT (RND
*6)+1): NEXT f
8030 IF st=1 THEN LET y=20: LET
x=30: LET p=1: LET q=1: LET l$=
"K": LET m$="M"
8040 IF st=2 THEN LET q=20: LET
p=30: LET x=1: LET y=1: LET l$=
"M": LET m$="K"
8060 IF st=1 THEN PRINT AT q,p;
"B";AT y,x; INK 6;"A"
8070 IF st=2 THEN PRINT AT q,p;
"A"; INK 6;AT y,x;"B"
8080 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;1: CLOSE #4
8090 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;e: CLOSE #4
8999 RETURN
9000 FOR f=USR "a" TO USR "j"+7:
READ q: POKE f,q: NEXT f
9010 DATA 24,24,60,60,102,102,23
1,231,231,231,102,102,60,60,24,2
4
9020 DATA 192,240,252,15,15,252,
240,192,3,15,63,240,240,63,15,3
9030 DATA 0,0,255,255,255,255,0,
0,60,60,60,60,60,60,60,60
9040 DATA 0,0,63,63,63,63,60,60,
0,0,252,252,252,252,60,60

```

```

9050 DATA 60,60,63,63,63,63,0,0,
60,60,252,252,252,252,0,0
9100 BRIGHT 1: INK 4: FOR f=5 TO
  1 STEP -1: PRINT INK 6;AT 0,f;
"D": BEEP .1,f: PRINT AT 0,f;"E"
: NEXT f
9110 PRINT AT 0,1;"6": FOR f=1 T
O 5: PRINT AT f,1; INK 6;"B": BE
EP .1,f: PRINT AT f,1;"F": NEXT
f
9120 PRINT AT 5,1;"I": FOR f=2 T
O 5: PRINT AT 5,f; INK 6;"C": BE
EP .1,f: PRINT AT 5,f;"E": NEXT
f
9130 FOR q=7 TO 11 STEP 4: FOR f
=0 TO 5: PRINT INK 6;AT f,q;"B"
: BEEP .1,f: PRINT AT f,q;"F": N
EXT f: NEXT q
9140 FOR f=8 TO 10: PRINT AT 2,f
; INK 6;"C": BEEP .1,f: PRINT AT
2,f;"E": NEXT f
9150 FOR f=5 TO 0 STEP -1: PRINT
AT f,13; INK 6;"A": BEEP .1,f:
PRINT AT f,13;"F": NEXT f
9160 PRINT AT 0,13;"6": FOR f=14
TO 18: PRINT AT 0,f; INK 6;"C":
BEEP .1,f: PRINT AT 0,f;"E": NE
XT f
9170 PRINT AT 0,18;"H": FOR f=1
TO 5: PRINT AT f,18; INK 6;"B":
BEEP .1,f: PRINT AT f,18;"F": NE
XT f
9180 FOR f=14 TO 17: PRINT AT 2,
f; INK 6;"C": BEEP .1,f: PRINT A
T 2,f;"E": NEXT f
9190 FOR f=24 TO 20 STEP -1: PRI
NT AT 0,f; INK 6;"D": BEEP .1,f:
PRINT AT 0,f;"E": NEXT f
9200 PRINT AT 0,20;"6": FOR f=1
TO 2: PRINT AT f,20; INK 6;"B":
BEEP .1,f: PRINT AT f,20;"F": NE
XT f

```

```

9210 PRINT AT 2,20;"I": FOR f=21
  TO 24: PRINT AT 2,f; INK 6;"C":
  BEEP .1,f: PRINT AT 2,f;"E": NE
XT f
9220 PRINT AT 2,24;"H": FOR f=3
  TO 5: PRINT AT f,24; INK 6;"B":
  BEEP .1,f: PRINT AT f,24;"F": NE
XT f
9230 PRINT AT 5,24;"J": FOR f=23
  TO 20 STEP -1: PRINT AT 5,f; IN
K 6;"D": BEEP .1,f: PRINT AT 5,f
;"E": NEXT f
9240 FOR f=30 TO 26 STEP -1: PRI
NT AT 0,f; INK 6;"D": BEEP .1,f:
  PRINT AT 0,f;"E": NEXT f
9250 PRINT AT 0,26;"G": FOR f=1
  TO 5: PRINT AT f,26; INK 6;"B":
  BEEP .1,f: PRINT AT f,26;"F": NE
XT f
9260 PRINT AT 5,26;"I": FOR f=27
  TO 30: PRINT AT 5,f; INK 6;"C":
  BEEP .1,f: PRINT AT 5,f;"E": NE
XT f
9270 FOR f=27 TO 30: PRINT AT 2,
f; INK 6;"C": BEEP .1,f: PRINT A
T 2,f;"E": NEXT f
9300 INK 7: BRIGHT 0
9305 PRINT AT 7,6;"@ PAUL STANLE
Y 1984"
9310 PRINT "A simple game for 2
  players in which the object is
  to surround your opponent in yo
ur tail, and at the same time av
oid his."
9320 PRINT "You can see both sh
ips on your television, but you
  cannot see the tail your ship
  leaves. This only shows up on yo
ur opponent's television."
9330 PRINT INK 6;"CONTROLS:"; I
NK 7;" Z..LEFT  X..RIGHT"
9340 PRINT TAB 10;"K..UP      M..
DOWN"

```

```
9350 INPUT "Press ENTER to begin
."; LINE P$
9500 CLS
9989 RETURN
9990 PRINT "Type on other Spect
rum: "" LOAD *""n";1 <enter>
"
9991 SAVE *""n";1;"a" LINE 10
9999 RETURN
```

# 19 Bowling Alley

## **Hardware requirements**

Two 16/48K Spectrums on ZX network

Two televisions

This is an unusual game, to say the least! Each player has control of a bowler and Derek the dinosaur. On the right-hand side of the screen you will see your dinosaur and, on the left, your opponent's bowler. Your bowler appears on the opponent's television at a random position which, at the start, will always be more than six spaces away from the dinosaur.

While the scenario may be bizarre, the object of the game is straightforward: you must simply bowl a ball which hits your opponent's dinosaur! But, like so much in life, the theory may be simple, but the practice is considerably more difficult!

When it's your turn to bowl, you are first asked to input the strength of your next shot. A strength of nineteen is required to reach your opponent's dinosaur. However, you only have a limited strength, to be shared over three bowls, which is not enough to reach Derek three times. After the third bowl, your strength is replenished. You are also asked to move your bowler, either up, down or not at all. While you do this your opponent can also move his dinosaur in a similar manner. When this operation has been completed, the bowl takes place on your opponent's screen. If you have bowled in the right gully, but with insufficient strength to obliterate Derek, you will be told so, otherwise you will be told that you missed.

When both players have bowled three times, they are told whether they should move their bowler up or down. Eventually one player should know where Derek is, and be able to bowl him over.

The game features very nice animation and should prove to be very entertaining. As you've probably guessed, this is our entry for the Most Bizarre Game of the Year Award!

**Program Notes:**

1-100 Call initial routines  
 150-230 Ensure that one Spectrum transmits while the other receives  
 240-999 Transmit and receive the dinosaur positions  
     Print where the bowler should move to  
 1000-1020 Receive the bowler's movement and strength from stream 4  
 1025-1160 Bowl graphically  
 1200-1250 Check for a hit  
     Transmit whether hit or not  
 1270-1380 Move the dinosaur  
 1400-1450 Move the bowler back to the start position  
 2000-2090 Input the strength  
 2100-2200 Move the bowler  
     Transmit the bowler's position and the strength of his bowl  
 2205-2260 Receive whether the bowl hit or not  
     Print the success of the bowl  
 3000-3650 End of game messages  
     Wait for a key to be pressed  
 7000-7530 Erase and print a dinosaur  
     Erase and print a bowler  
 8000-8120 Set up the screen and variables  
 8130-8999 Transmit your bowler's position  
     Receive your opposing bowler's position  
 9000-9090 Define the graphics characters  
 9100-9899 Instructions  
 9990-9992 Send the program to the other computer

```

1 REM @ Paul Stanley
5 GO SUB 9990
10 RANDOMIZE : CLS : BORDER 0:
PAPER 0: INK 7: CLS : POKE 2375
0,0
20 GO SUB 9000
30 INPUT "Station Number (1 or
2) ";st
35 IF st<>1 AND st<>2 THEN GO
TO 35
40 FORMAT "n";st: LET st1=3-st
100 GO SUB 8000
150 LET str=40
160 FOR s=1 TO 3
165 PRINT INK 5;AT 3,12;t$(s*3
-2 TO s*3);" BOWL"

```

```

170 GO SUB 1000*st
200 GO SUB 1000*st1
230 NEXT s
240 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;dy: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;k: CLOSE #4
250 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;k: CLOSE #4: OPEN #
4;"n";st1: PRINT #4;dy: CLOSE #4
270 IF y<k THEN PRINT AT 1,0;
PAPER 2; INK 6; FLASH 1;" YOU N
EED TO MOVE BOWLER DOWN "
280 IF y>k THEN PRINT AT 1,0;
PAPER 2; INK 6; FLASH 1;"YOU NEE
D TO MOVE BOWLER UPWARDS."
290 FOR f=1 TO 15 STEP .5: BEEP
.04,f: NEXT f
999 GO TO 150
1000 GO SUB 1270
1005 PRINT AT 1,0; PAPER 1; FLAS
H 1;" WAIT FOR YOUR OPPONENT'S M
OVE. "
1010 LET by1=by
1020 OPEN #4;"N";st1: INPUT #4;b
y: INPUT #4;s1: CLOSE #4
1025 IF s1=0 THEN PRINT AT 1,0;
PAPER 2; FLASH 1;" YOUR OPPON
ENT CANNOT BOWL! ": FOR f=0 TO
20 STEP 2: BEEP .2,f: NEXT f: R
ETURN
1030 GO SUB 7500
1100 FOR f=0 TO 6
1110 PRINT PAPER 1;AT by-2,f;"
AB";AT by-1,f;" CDE";AT by,f;"
FGH"
1120 BEEP .05,f: NEXT f
1130 PRINT PAPER 1;AT by-2,f;"
IB";AT by-1,f;" JK";AT by,f;" GL
"
1135 LET i=2+(4 AND by/2=INT (by
/2))
1137 IF s1>20 THEN LET s1=20
1140 FOR q=10 TO 10+s1

```

```

1150 PRINT INK 9; PAPER i; AT by
,q; " Q"
1160 BEEP .02,q: NEXT q
1200 LET h=0
1210 IF by=dy THEN LET h=1
1220 IF by=dy AND s1>18 THEN LE
T h=2
1230 OPEN #4;"n";st1: PRINT #4;h
: CLOSE #4
1240 IF h=2 THEN GO TO 3000
1250 GO TO 1400
1270 PRINT FLASH 1; PAPER 6; IN
K 3; AT 1,0; " DEREK: 6..DOWN, 7..
UP, 0..STAY "
1280 BEEP .1,10
1285 LET dy1=dy
1290 LET k$=INKEY$
1300 IF k$="6" AND dy<20 THEN G
O TO 1350
1310 IF k$="7" AND dy>4 THEN GO
TO 1350
1320 IF k$="0" THEN GO TO 1350
1330 GO TO 1290
1350 BEEP .1,15
1360 LET dy=dy+(k$="6")-(k$="7")
1370 GO SUB 7000
1380 RETURN
1400 PRINT AT by,7; PAPER 1; "
"; AT by-1,7; " "; AT by-2,7; "
"
1410 PRINT PAPER i; AT by,q; " "
1415 GO SUB 7500
1417 GO SUB 7000
1420 FOR f=1 TO 100: NEXT f
1430 RETURN
2000 IF str=0 THEN PRINT PAPER
1; FLASH 1; AT 1,0; "YOUR ENERGY
IS 0. YOU CAN'T BOWL": LET sa=0:
FOR f=0 TO -20 STEP -2: BEEP .2
,f: NEXT f: GO TO 2200
2010 PRINT PAPER 1; FLASH 1; AT
1,0; "YOUR ENERGY IS ";str;" ENTE
R STRENGTH"; TAB 0;

```



```

2020 INPUT "Strength? ";sa
2030 IF sa>str OR sa<1 THEN GO
TO 2020
2040 LET str=str-sa
2090 BEEP .1,10
2100 PRINT PAPER 6; INK 3; FLAS
H 1;AT 1,0;" BOWLER: 6=DOWN, 7=
UP, 0=STAY "
2110 LET k$=INKEY$
2120 IF k$="6" AND y<20 THEN GO
TO 2150
2130 IF k$="7" AND y>4 THEN GO
TO 2150
2140 IF k$="0" THEN GO TO 2150
2145 GO TO 2110
2150 BEEP .1,15
2160 LET y=y+(k$="6")-(k$="7")
2170 PRINT AT 1,0;a$;a$( TO 10)
2200 OPEN #4;"n";st1: PRINT #4;y
: PRINT #4;sa: CLOSE #4
2205 IF sa=0 THEN RETURN
2210 OPEN #4;"N";st1: INPUT #4;h
: CLOSE #4
2220 IF h=0 THEN PRINT PAPER 3
;AT 1,0; FLASH 1;"THAT COMPLETEL
Y MISSED DINOSAUR.": BEEP .2,-10
: BEEP .2,-30
2230 IF h=1 THEN PRINT PAPER 3
;AT 1,0; FLASH 1;" THAT WAS IN
THE RIGHT GULLEY ": BEEP .2,0:
BEEP .2,1
2240 IF h=2 THEN GO TO 3500
2250 FOR f=1 TO 100: NEXT f
2260 RETURN
3000 GO SUB 7000
3005 OVER 1
3010 FOR f=1 TO 16: GO SUB,7000:
BEEP .04,f: NEXT f
3020 OVER 0
3030 CLS : PRINT AT 10,0; FLASH
1;" DEREK HAS GONE TO THE GREAT
DINOSAUR GRAVEYARD IN THE S
KY!"
3100 GO TO 3600

```

```

3500 CLS : PRINT AT 10,0; FLASH
1;" A GREAT BOWL, YOU HAVE HIT Y
OUR OPPONENT'S DEREK THE DINODAU
R! "
3600 FOR f=1 TO 30: BEEP .03,f:
BEEP .03,15: BEEP .03,31-f: NEXT
f
3610 PRINT AT 18,2;"Press any ke
y to play again."
3620 IF INKEY$="" THEN GO TO 36
20
3650 GO TO 100
7000 LET p=2+(4 AND dy1/2=INT (d
y1/2))
7010 LET p1=(2 AND p=6)+(6 AND p
=2)
7020 PRINT PAPER p1;AT dy1-1,30
;" "; PAPER p;AT dy1,30;" "
7030 LET p=2+(4 AND dy/2=INT (dy
/2))
7040 LET p1=(2 AND p=6)+(6 AND p
=2)
7050 PRINT INK 0; PAPER p1;AT d
y-1,30;"MN"; PAPER p;AT dy,30;"O
P"
7060 RETURN
7510 PRINT AT by1,0; PAPER 1;"
";AT by1-1,0;" ";AT by1-2,0;"
"
7520 PRINT AT by-2,0; PAPER 1;"
AB";AT by-1,0;"CDE";AT by,0;"FGH
"
7530 RETURN
8000 DIM a$(22): LET t$="1st2nd3
rd"
8020 LET b$="RRRRRRRRRRRRRRRRRRRR
RRRRRRRRRRRRRR"
8030 CLS : PRINT PAPER 3; INK 6
;AT 0,0;b$;AT 2,0;b$;AT 21,0;b$
8035 PRINT AT 2,0; PAPER 1;TAB 1
0;
8040 LET p=2: FOR f=3 TO 20
8050 PRINT PAPER 1;AT f,0;TAB 1
0; PAPER p;a$

```

```

8060 LET p=(6 AND p=2)+(2 AND p=
6)
8070 NEXT f
8080 LET dy=INT (RND*17)+4
8090 LET by=INT (RND*17)+4
8100 IF ABS (dy-by)<6 THEN GO T
O 8080
8110 LET dy1=dy: GO SUB 7000
8120 LET by1=by: GO SUB 7500
8130 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;by: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;y: CLOSE #4
8140 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;y: CLOSE #4: OPEN #
4;"n";st1: PRINT #4;by: CLOSE #4
8999 RETURN
9000 FOR f=USR "a" TO USR "r"+7:
READ q: POKE f,q: NEXT f
9010 DATA 0,0,0,0,0,31,127,255,6
0,126,255,227,203,130,190,192
9020 DATA 1,3,7,14,29,57,49,38,2
40,207,63,255,255,255,255,250
9030 DATA 224,240,248,252,222,14
2,6,7,47,63,22,0,0,0,0,0
9040 DATA 127,61,29,29,59,118,25
1,127,3,129,128,128,0,0,0,128
9050 DATA 0,0,0,0,0,0,7,15,47,11
1,239,239,247,251,253,254
9060 DATA 96,96,96,160,160,192,1
92,224,240,112,184,156,12,7,0,12
8
9070 DATA 0,0,96,176,241,51,55,6
3,0,0,0,0,224,240,121,121
9080 DATA 63,63,63,63,63,31,24,4
9,249,249,241,241,249,255,206,12
8
9090 DATA 0,28,62,127,127,127,62
,28,199,124,68,68,68,68,124,199
9100 FOR f=0 TO 19 STEP .25: PRI
NT AT f,f; INK INT (RND*5)+3;"BO
WLING ALLEY": BEEP .04,f: NEXT f
9110 PRINT AT 17,0;"@ Paul Stanl
ey"

```

```

9120 INPUT INK 3;"Press ENTER f
or instructions: "; LINE u$
9130 CLS : PRINT "For some reaso
n unknown to you, Derek The Dino
saur has taken a wrong turning,
and has landed himself right
into trouble!"
9140 PRINT "He is at the wrong
end of a bowling alley. Your
opponent is determined to send
the ball straight at you."
9150 PRINT "The bowler has limi
ted energy to be shared over 3 bo
wls. A bowl strength of 19 is r
equired to hit a dinosaur. Aft
er 3 bowls a bowler regains hi
s strength and you are told wh
ere your opponent has his di
nosaur."
9160 PRINT "On your screen you
can see your dinosaur and the en
emy bowler."
9170 INPUT INK 3;"Press ENTER t
o start."; LINE u$
9180 CLS
9899 RETURN
9990 PRINT "Type on other Spect
rum: "" LOAD *""n"";1 <enter>"
9991 SAVE *""n"";1;"a" LINE 10
9992 RETURN

```

# 20 Minelay

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

In this game, each player has control of a tank which can be guided about the screen using the cursor controls. These are no ordinary tanks, since they change colour as they trundle about and when green lay two mines at their current position.

This mine-laying operation continues until one of the tanks hits a mine, which won't take too long because the screen soon fills up with deadly bombs. In your efforts to avoid planted mines you should make sure that you also manage to avoid the mines bordering the screen.

Your tank will rumble around the screen with or without your guidance, so you must react quickly to prevent a collision. In this game, both televisions show the same picture, therefore if you only have one set available minelay can still be played.

## Program Notes:

- 1-70 Call the starting routines
- 100-190 Ensure that one computer transmits while one receives
- 200-999 Work out where to place mines
- 1000-1010 Send the value of the key pressed to the other computer
- 1020-1999 Move and print the tank
  - Check for mines
- 2000 Receive the value of the key pressed by your opponent
- 2010-2999 Check that your opponent has not crashed
  - Move and print a tank
- 3000-3040 Print mines behind the tanks

4000-4010 Erase the tank  
           Receive keypress to prevent both computers transmitting together  
           Send a header to inform the other Spectrum that you have crashed  
 4020-4940 Tank collision  
           Print winning messages  
           Wait for a key to play again  
 8000-8999 Set up the screen display and variables  
 9000-9050 Set up the graphics characters  
 9060-9899 Instructions  
 9910-9930 Send the program to the other computer

```

      1 REM * @ PAUL STANLEY 1984 *
      5 GO SUB 9900
     10 RANDOMIZE : CLS : BORDER 7:
    PAPER 7: INK 0: CLS
     20 POKE 23658,8
     30 POKE 23750,7
     40 GO SUB 9000
     50 INPUT "Which Station? (1 or
    2) ";st
     55 IF st<>1 AND st<>2 THEN GO
    TO 55
     60 FORMAT "n";st: LET st1=3-st
     70 GO SUB 8000
    100 FOR i=0 TO 4
    130 GO SUB 1000*st
    160 GO SUB 1000*st1
    190 NEXT i
    200 LET f=d: LET h=y: LET q=x:
GO SUB 3000
    210 LET f=e: LET h=q: LET q=p:
GO SUB 3000
    220 BEEP .04,10
    999 GO TO 100
  1000 LET k$=INKEY$
  1010 OPEN #4;"n";st1: PRINT #4;k
  $: CLOSE #4
  1020 LET d=(d AND k$="")+ (1 AND
  k$="Z")+ (2 AND k$="M")+ (3 AND k$
  ="X")+ (4 AND k$="K")
  1030 LET y1=y: LET x1=x

```

```

1040 LET y=y+(d=2)-(d=4)
1050 LET x=x+(d=3)-(d=1)
1055 IF ATTR (y,x)>64 OR ATTR (y
,x+1)>64 OR ATTR (y+1,x)>64 OR A
TTR (y+1,x+1)>64 THEN GO TO 400
0
1060 PRINT AT y1,x1;" ";AT y1+1
,x1;" "; INK i;AT y,x;a$(d);AT
y+1,x;b$(d)
1999 RETURN
2000 OPEN #4;"n";st1: INPUT #4;k
$: CLOSE #4
2010 IF k$="T" THEN GO TO 4500
2020 LET e=(e AND k$="")+ (1 AND
k$="Z")+ (2 AND k$="M")+ (3 AND k$
="X")+ (4 AND k$="K")
2030 LET q1=q: LET p1=p
2040 LET q=q+(e=2)-(e=4)
2050 LET p=p+(e=3)-(e=1)
2060 PRINT AT q1,p1;" ";AT q1+1
,p1;" "; INK 4-i;AT q,p;a$(e);A
T q+1,p;b$(e)
2999 RETURN
3000 IF f=1 THEN PRINT BRIGHT
1; INK RND*4;AT h,q+2;"M";AT h+1
,q+2;"M"
3010 IF f=2 THEN PRINT BRIGHT
1; INK RND*4;AT h-1,q;"MM"
3020 IF f=3 THEN PRINT BRIGHT
1; INK RND*4;AT h,q-1;"M";AT h+1
,q-1;"M"
3030 IF f=4 THEN PRINT BRIGHT
1; INK RND*4;AT h+2,q;"MM"
3040 RETURN
4000 PRINT AT y1,x1;" ";AT y1+1
,x1;" "; INK i;AT y,x;a$(d);AT
y+1,x;b$(d)
4005 OPEN #4;"n";st1: INPUT #4;k
$: CLOSE #4
4010 OPEN #4;"n";st1: PRINT #4;"
T": CLOSE #4
4020 FOR k=1 TO 10: FOR f=1 TO 4
4030 PRINT INK 3;AT y,x;a$(f);A
T y+1,x;b$(f)

```

```

4040 BEEP .01,10
4050 NEXT f
4060 NEXT k
4070 CLS : PRINT AT 10,1; FLASH
1;" BAD LUCK, YOU HIT A MINE!
"
4080 GO TO 4800
4500 FOR k=1 TO 10: FOR f=1 TO 4
4510 PRINT INK 3;AT q,p;a$(f);A
T q+1,p;b$(f)
4520 BEEP .01,10
4530 NEXT f
4540 NEXT k
4560 CLS : PRINT AT 10,1; FLASH
1;" YOUR OPPONENT HIT A MINE!
"
4800 FOR f=1 TO 10: BEEP .1,f: B
EEP .1,11-f: NEXT f
4900 PRINT AT 20,2; INK 3;"PRESS
ANY KEY TO PLAY AGAIN."
4910 IF INKEY$="" THEN GO TO 49
10
4920 CLS
4940 GO TO 70
8000 DIM a$(4,2): DIM b$(4,2)
8010 LET a$(1)="AB": LET b$(1)="
CD"
8020 LET a$(2)="EF": LET b$(2)="
GH"
8030 LET a$(3)="IJ": LET b$(3)="
KL"
8040 LET a$(4)="ON": LET b$(4)="
QP"
8050 IF st=1 THEN LET y=1: LET
x=1: LET d=3: LET p=29: LET q=19
: LET e=1
8060 IF st=2 THEN LET q=1: LET
p=1: LET e=3: LET x=29: LET y=19
: LET d=1
8070 FOR f=0 TO 31: PRINT BRIGH
T 1; INK RND*4;AT 0,f;"M";AT 21,
f;"M": NEXT f

```



```

8080 FOR f=1 TO 20: PRINT BRIGH
T 1; INK RND*4;AT f,0;"M";AT f,3
1;"M": NEXT f
8090 PRINT AT y,x;a$(d);AT y+1,x
;b$(d)
8100 PRINT AT q,p;a$(e);AT q+1,p
;b$(e)
8999 RETURN
9000 FOR f=USR "a" TO USR "m"+7:
  READ q: POKE f,q: NEXT f
9010 FOR f=0 TO 31: POKE USR "q"
+7-f,PEEK (USR "e"+f): NEXT f
9020 DATA 0,0,0,15,31,27,16,255,
0,0,0,254,5,115,249,253,255,16,2
7,31,15,0,0,0,253,249,115,5,254,
0,0,0
9030 DATA 15,20,25,19,23,23,23,1
9,240,40,152,200,232,232,232,200
,29,29,25,29,15,1,1,1,184,184,15
2,184,240,128,128,128
9040 DATA 0,0,0,127,160,206,159,
191,0,0,0,240,248,216,8,255,191,
159,206,160,127,0,0,0,255,8,216,
248,240,0,0,0
9050 DATA 124,214,130,170,186,25
4,124,0
9060 INK 7
9100 PRINT "ppppppppppapaaapppppa
paaaappaaap"
9110 PRINT "papapaaapaaappaaapapaaa
paaapaaapap"
9120 PRINT "papapaaapaaapapapppppa
paaapppppappp"
9130 PRINT "papapaaapaaapaaappapaaa
paaapaaapaa"
9140 PRINT "paaapppppapaaapppppa
pppapaapaaap"
9150 INK 0
9160 FOR f=0 TO 3: FOR q=0 TO 30
  STEP 2
9170 PRINT INK 3; OVER 1;AT f,q
;"IJ";AT f+1,q;"KL"
9180 BEEP .1,q

```

```

9190 PRINT OVER 1;AT f,q;"IJ";A
T f+1,q;"KL"
9200 NEXT q
9210 NEXT f
9220 PRINT INK 3;AT 6,5;"@ Pau
l Stanley 1984"
9230 PRINT '"Each player has a t
ank which changes colour from
black to green. When they re
ach green they lay a mine beh
ind them, so eventually the scre
en becomes filled with mines."
9240 PRINT '"If you hit a mine y
ou will be destroyed, and will
have lost."
9250 PRINT '"The contols are:"
9260 PRINT INK 2;" Z....LEFT
X....RIGHT K....UP
M....DOWN"
9270 INPUT INK 3;" Press ENTER
to start "; LINE p$
9280 CLS
9899 RETURN
9900 CLS
9910 PRINT '"TYPE ON OTHER SPECT
RUM:-"' ' " LOAD *"'n"' ;1 <ente
r>"
9920 SAVE *"'n"' ;1;"a" LINE 10
9930 RETURN

```

# 21 Nessie

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

This game makes the assumption that there really is a Loch Ness Monster, but it has yet to be caught! One player controls Nessie and tries to outsmart the other player who is in command of a small submersible equipped with the most up-to-date Monster catching equipment!

The water of the Loch is so murky that neither protagonist can see the other, but each has a number of useful resources at hand. Nessie can detect motion, so when nets shoot out of the sides of the submarine, she knows in which direction to dodge.

The sub is armed with a limited radar system which will only work when the detector is cast out with the nets. Each time a net shoots out the radar uses its arrow display to indicate whether Nessie is above or below the cast. When the sub is at the same depth as its prey the radar will flash green, otherwise it will remain red when nets are cast.

If you are the Monster Catcher you should not cast the nets continuously, since this will betray your position to your opponent. If you manage to catch Nessie, the Spectrum will provide the approximate time it took to accomplish your mission. Players should then reverse roles to see how well each copes with a different task. Happy hunting!

## Program Notes:

- 1-50 Call the starting routines
- 100-190 Input Nessie or monster catcher option
- 200 Direct the program to Nessie or submarine loops
- 1000-1040 Read the keyboard
- Move and print Nessie
- 1050 Receive the submarine position and whether or not to fire

1070-1078 Check to see if the net hit Nessie  
           Transmit whether the net hit Nessie or not, and Nessie's  
           position  
 1080-1130 Fire a net  
 2000-2040 Read the keyboard  
           Move and print the submarine  
       2050 Transmit the submarine position and whether you are firing a  
           net  
 2070-2260 Receive the result of casting a net  
           Fire a net  
           Print the radar  
 3000-3050 Transmit Nessie's position  
           Print a cage and the time  
 4000-4040 Receive Nessie's position  
           Print a cage and the time  
 4500-4560 Wait for a key to be pressed  
 8000-8999 Set up the screen and variables  
 9000-9080 Define the graphics characters  
 9100-9200 Instructions  
 9990-9993 Send the program to the other computer

```

      1 REM +-+ @ Paul Stanley +-+
      5 GO SUB 9990
     10 CLS : PAPER 1: INK 7: BORDE
R 2: CLS : POKE 23750,2
     20 POKE 23658,8
     50 GO SUB 9000
    100 PRINT "DO YOU WANT TO BE T
HE LOCH NESS MONSTER, OR TO CATC
H IT? (YOU AND YOUR OPPONENT M
UST NOT BE THE SAME THING)."
    110 PRINT "PRESS:" 1...LOCH
NESS MONSTER" 2...MONSTER CATC
HER"
    120 IF INKEY$="1" THEN LET st=
1: GO TO 150
    130 IF INKEY$="2" THEN LET st=
2: GO TO 150
    140 GO TO 120
    150 BEEP .1,10
    170 FORMAT "n";st: LET st1=3-st
    180 LET t=0
    190 GO SUB 8000
    200 GO TO 1000*st
  
```

```

1000 LET y1=my: LET x1=mx: LET t
=t+1
1010 LET my=my+(INKEY$="M" AND m
y<20)-(INKEY$="K" AND my>4)
1020 LET mx=mx+(INKEY$="C" AND m
x<28)-(INKEY$="X" AND mx>0)
1030 PRINT AT y1,x1;" ";AT y1
+1,x1;" ";AT my,mx;m$;AT my+1,
mx;n$
1040 LET y1=y: LET x1=x
1050 OPEN #4;"n";st1: INPUT #4;y
: INPUT #4;x: INPUT #4;k$: CLOSE
#4
1070 IF k$<>"Z" AND k$<>"O" THEN
GO TO 1000
1072 LET h=2: IF y=my OR y+1=my
THEN LET h=2+(2 AND x>mx AND k$
="Z")+(2 AND x<mx AND k$="O")
1076 IF h=4 AND ABS (x-mx)<10 TH
EN LET h=6
1078 OPEN #4;"n";st1: PRINT #4;h
: PRINT #4;my: CLOSE #4
1080 IF k$="Z" THEN FOR q=x-1 T
O x-6 STEP -1: IF q>=0 THEN PRI
NT INK 4;AT y+1,q;"N": BEEP .02
,q: NEXT q
1090 IF k$="O" THEN FOR q=x+3 T
O x+8: IF q<32 THEN PRINT INK
4;AT y+1,q;"N": BEEP .02,q: NEXT
q
1095 IF h=6 THEN GO TO 3000
1110 IF k$="Z" THEN FOR h=q+1 T
O x-1: PRINT AT y+1,h;" ": BEEP
.02,h: NEXT h
1120 IF k$="O" THEN FOR h=q-1 T
O x+3 STEP -1: PRINT AT y+1,h;"
": BEEP .02,h: NEXT h
1130 GO TO 1000
2000 LET y1=y: LET x1=x: LET t=t
+1
2010 LET k$=INKEY$
2020 LET y=y+(k$="M" AND y<20)-(
k$="K" AND y>4)

```

```

2030 LET x=x+(k$="C" AND x<29)-(
k$="X" AND x>0)
2040 PRINT AT y1,x1;" ";AT y1+
1,x1;" ";AT y,x; INK 6;s$;AT y
+1,x;t$
2050 OPEN #4;"n";st1: PRINT #4;y
: PRINT #4;x: PRINT #4;k$: CLOSE
#4
2060 IF k$<>"O" AND k$<>"Z" THEN
GO TO 2000
2070 OPEN #4;"n";st1: INPUT #4;h
: INPUT #4;my: CLOSE #4
2080 PRINT AT 1,16; FLASH (h=4);
PAPER h;" " AND h=4;"0000"
AND my<y AND h=2;"PPPP" AND my
>y AND h=2
2200 IF k$="Z" THEN FOR q=x-1 T
O x-6 STEP -1: IF q>=0 THEN PRI
NT INK 4;AT y+1,q;"N": BEEP .02
,q: NEXT q
2210 IF k$="O" THEN FOR q=x+3 T
O x+8: IF q<32 THEN PRINT INK
4;AT y+1,q;"N": BEEP .02,q: NEXT
q
2220 IF h=6 THEN GO TO 4000
2230 IF k$="Z" THEN FOR h=q+1 T
O x-1: PRINT AT y+1,h;" ": BEEP
.02,h: NEXT h
2250 IF k$="O" THEN FOR h=q-1 T
O x+3 STEP -1: PRINT AT y+1,h;"
": BEEP .02,h: NEXT h
2260 GO TO 2000
3000 OPEN #4;"n";st1: PRINT #4;m
y: PRINT #4;mx: CLOSE #4
3010 PRINT INK 4;AT my,mx;"N/N/N
";AT my+1,mx;"N/N/N"
3020 FOR f=10 TO -20 STEP -1: BE
EP .04,-10-f: BEEP .04,f: NEXT f
3030 CLS
3040 PRINT AT 4,0;"Nessie has be
en caught after"INT (t/2);" sec
onds."
3050 GO TO 4500

```

```

4000 OPEN #4;"n";st1: INPUT #4;m
y: INPUT #4;mx: CLOSE #4
4010 PRINT INK 4;AT my,mx;"NNNN
";AT my+1,mx;"NNNN"
4020 FOR f=10 TO 40: BEEP .04,50
-f: BEEP .04,f: NEXT f
4030 CLS
4040 PRINT AT 4,3;"You caught Ne
ssie after";AT 5,3;INT (t/2);" s
econds."
4500 PRINT AT 10,0;"How about sw
itching roles to see how you do t
hen?"
4510 PRINT AT 18,2;"Press Any Ke
y To Play Again."
4520 IF INKEY$="" THEN GO TO 45
20
4530 CLS
4560 GO TO 100
8000 LET m$="ABCD": LET n$="EFG
"
8010 LET s$="HIJ": LET t$="KLM"
8020 DIM a$(32)
8030 CLS : FOR f=1 TO 4: PRINT
PAPER 5;a$: NEXT f
8040 LET my=4: LET mx=0: LET y=2
0: LET x=29
8060 IF st=1 THEN PRINT AT my,m
x;m$;AT my+1,mx;n$
8070 IF st=2 THEN PRINT AT 1,10
;"RADAR:"; PAPER 2;"
8999 RETURN
9000 FOR f=USR "a" TO USR "p"+7:
READ q: POKE f,q: NEXT f
9010 DATA 16,48,48,48,120,120,12
0,60,0,0,0,12,30,63,63,127
9020 DATA 0,0,1,7,15,31,190,254,
0,60,246,247,255,14,0,0
9030 DATA 62,31,31,31,15,15,7,3,
127,127,251,241,240,224,192,128
9040 DATA 252,252,248,248,240,0,
0,0,0,3,15,31,25,176,176,249

```

```

9050 DATA 255,195,153,255,231,19
5,195,231,0,192,240,248,152,13,1
3,159
9060 DATA 255,208,191,30,95,127,
67,64,255,66,255,24,255,195,255,
255
9070 DATA 255,29,253,120,250,254
,194,2,255,73,73,255,73,73,255,0
9080 DATA 16,56,124,254,56,56,56
,56,56,56,56,56,254,124,56,16
9100 FOR f=1 TO 50
9110 LET y=INT (RND*21)
9111 LET x=INT (RND*21)
9115 PRINT INK INT (RND*4)+4;AT
y,x;"NESSIE-ABCD";AT y+1,x+7;"E
F6"
9120 BEEP .02,f/2: NEXT f
9125 PAUSE 30
9130 CLS : PRINT INK 5;" N E S
S I E @ PAUL STANLEY"
9140 PRINT "In this game one pl
ayer controlsThe Loch Ness Monst
er, while theother controls a sm
all submarinewhich is out to cat
ch Nessie."
9150 PRINT "Nessie can only see
the nets which the submarine
casts, whilethe sub has the aid
of a radar guidance system. If
the net is cast at the right d
epth in the right direction the
radar will flash."
9160 PRINT INK 6'"Controls:"'"
X..LEFT C..RIGHT"" K..U
P M..DOWN"
9170 PRINT INK 6'"And additiona
lly, for the sub:"'" Z..NET LEF
T O..NET RIGHT"
9180 INPUT INK 0;"Press ENTER..
"; LINE p$
9190 CLS
9200 RETURN
9990 CLS : PRINT "Type on other
Spectrum:"

```



```
9991 PRINT ' "    LOAD *"n";1  
<enter>"  
9992 SAVE *"n";1;"a" LINE 10  
9993 RETURN
```

# 22 Morse Trainer

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

This is the least competitive program of the collection. Although fun to use, it also provides users with a painless introduction to Morse Code, the internationally recognised system for representing letters and symbols by long notes (dashes) and short notes (dots).

The use of this network tutor is very straightforward. One person types in a message in letters, then the Spectrum translates them into the dots and dashes which are displayed on the other player's screen. Note that only capital letters and spaces will be accepted by the Spectrum.

The receiving computer waits for a message and then asks whether the code should be displayed quickly or slowly. You should start by using the slow speed and as you get more accustomed to Morse switch to fast. The computer will sound the Morse Code equivalent of each letter (playing either a long or short note), and to make life easy it also prints, in bold letters, DOT or DASH. After each letter has been displayed there is a short pause and, after a word, a much longer pause.

When the message has been completed, the computer asks whether you want to hear it again. It is advisable to write the letters down as you decode them.

## Program Notes:

- 1-100 Call the starting routines
- 1000-1080 Type in a message
  - Transmit the message
- 1090-1120 Await a header to indicate that your partner is ready
- 1500-1600 Check for invalid characters in the message
- 2000-2030 Wait for the message to be transmitted
- 2040-2090 Input the desired speed
- 2100-2200 Translate the message
  - Print dots, dashes and pause
- 2210-2260 Input whether the message should be repeated
- 2270-2900 Send a header to the other Spectrum

3000-3999 Print a translation table  
 8000-8050 Input whether to transmit or receive  
 8060-8090 Send your option and receive your opponent's option  
     Compare the two choices  
 8500-8550 Tell the players to input their options again  
 9000-9120 Define the graphics characters  
     9130 The dot/dash codes for the 26 letters  
 9200-9300 Instructions  
 9800-9840 Set up the codes  
 9990-9992 Send the program to the other Spectrum

```

      1 REM -A- MORSE PRACTICE -A-
      2 REM -A- @ Paul Stanley -A-
      5 GO SUB 9990
    10 CLS : BORDER 7: PAPER 7: IN
K 0: CLS
    15 POKE 23750,7
    20 POKE 23658,8
    30 GO SUB 9000
    40 INPUT "STATION NUMBER? (1 O
R 2) ";ST
    45 IF ST<>1 AND ST<>2 THEN GO
TO 40
    50 FORMAT "N";ST: LET ST1=3-ST
    100 GO TO 8000
  1000 GO SUB 3000
  1010 PRINT "Enter your sentence
:"
  1070 INPUT S$
  1075 GO TO 1500
  1080 OPEN #4;"N";st1: PRINT #4;s
$: CLOSE #4
  1090 CLS : PRINT FLASH 1;AT 10,
10;"Please Wait"
  1100 OPEN #4;"n";st1
  1105 INPUT #4;A$
  1110 CLOSE #4
  1120 GO TO 8000
  1500 FOR f=1 TO LEN S$
  1510 IF S$(f)=" " THEN GO TO 15
30
  1520 IF S$(f)<"A" OR S$(f)>"Z" T
HEN GO TO 1540
  1530 NEXT f: GO TO 1080

```

```

1540 BEEP .3,-20
1550 CLS
1560 PRINT "Your sentence can o
nly have      capital letters A-Z
and spaces."
1570 PRINT INK 2"Other punctua
tion and charactersare not permi
tted."
1580 PRINT INK 1""Press any ke
y to try again."
1590 IF INKEY$="" THEN GO TO 15
90
1600 BEEP .1,10: GO TO 1000
2000 CLS : PRINT FLASH 1;AT 10,
0;" PLEASE WAIT FOR THE MESSAGE
TO BE RECEIVED FROM YOUR PARTNE
R. "
2010 OPEN #4,"N",ST1
2020 INPUT #4;S$
2030 CLOSE #4
2040 BEEP .1,5: CLS
2050 PRINT "MESSAGE RECEIVED: P
RESS:"' INK 2"1...FAST""2...S
LOW"
2060 LET K$=INKEY$: IF K$<>"1" A
ND K$<>"2" THEN GO TO 2060
2062 BEEP .2,10
2065 LET B=VAL k$
2070 GO SUB 3000
2080 FOR f=1 TO 10: BEEP .02,f:
NEXT f
2090 FOR q=1 TO 40: NEXT q
2100 FOR f=1 TO LEN s$
2120 LET a=CODE s$(f)-64
2130 IF a=-32 THEN FOR d=1 TO 1
00*b: NEXT d: GO TO 2200
2140 FOR l=1 TO l(a)
2150 IF z$(a,l)="1" THEN PRINT
PAPER 6;AT 16,11;" ";AT 1
7,11;" LMNOP ";AT 18,11;" QRSTU
";AT 19,11;" ": BEEP .1*b,
20

```

```

2160 IF z$(a,1)="2" THEN PRINT
AT 16,11; PAPER 2; INK 6;"
";AT 17,11;" BCDEF ";AT 18,11;"
GHIJK ";AT 19,11;" ": BEE
P .4*b,20
2180 FOR d=1 TO 20*b: IF d=14*b
THEN PRINT AT 16,11;" ";A
T 17,11;" ";AT 18,11;"
";AT 19,11;" "
2190 NEXT d: NEXT 1
2200 FOR d=1 TO 50*b: NEXT d: NE
XT f
2210 CLS : PRINT "Press: 1..TO
HEAR MESSAGE AGAIN" INK 2"
2..TO CONTINUE"
2220 LET K$=INKEY$
2230 IF K$<>"1" AND K$<>"2" THEN
GO TO 2220
2260 IF K$="1" THEN GO TO 2040
2270 OPEN #4;"N";ST1
2280 PRINT #4;"1"
2290 CLOSE #4
2900 GO TO 8000
3000 CLS : PRINT "A A-";TAB 15;
"N -A"
3010 PRINT "B -AAA";TAB 15;"O
---"
3020 PRINT "C -A-A";TAB 15;"P
A--A"
3030 PRINT "D -AA";TAB 15;"Q -
-A-"
3040 PRINT "E A";TAB 15;"R A-A
"
3050 PRINT "F AA-A";TAB 15;"S
AAA"
3060 PRINT "G --A";TAB 15;"T -
"
3070 PRINT "H AAAA";TAB 15;"U
AA-"
3080 PRINT "I AA";TAB 15;"V AA
A-"
3090 PRINT "J A---";TAB 15;"W
A--"

```

```

3100 PRINT "K  -A-";TAB 15;"X  -
AA-"
3110 PRINT "L  A-AA";TAB 15;"Y
-A--"
3120 PRINT "M  --";TAB 15;"Z  --
AA"
3130 FOR F=1 TO 13 STEP 2
3140 PRINT OVER 1;AT F,0; INK 3
;"
"
3150 NEXT F
3999 RETURN
8000 BEEP .1,10: CLS : PRINT PA
PER 2; INK 6;" WHICH DO YOU WANT
TO DO? "
8010 PRINT INK 3'"1...SEND A M
ESSAGE TO PARTNER"
8020 PRINT INK 1'"2...RECEIVE
MESSAGE FROM PARTNER"
8030 LET K$=INKEY$
8040 IF K$<>"1" AND K$<>"2" THEN
GO TO 8030
8050 BEEP .2,0
8060 IF ST=1 THEN OPEN #4;"n";s
t1: PRINT #4;k$: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;p$: CLOSE #
4
8070 IF ST=2 THEN OPEN #4;"n";s
t1: INPUT #4;p$: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;k$: CLOSE #
4
8080 IF k$=p$ THEN BEEP .3,-20:
GO TO 8500
8090 GO TO 1000*VAL k$
8500 CLS : PRINT '"You and your
partner cannot do the same thin
g."
8510 PRINT INK 1'"One must send
a message, and theother receive
it."
8520 PRINT INK 2'"Press any ke
y.."
8540 IF INKEY$="" THEN GO TO 85
40

```

```

8550 BEEP .1,10: GO TO 8000
9000 FOR f=USR "a" TO USR "u"+7:
  READ q: POKE f,q: NEXT f
9010 DATA 0,0,0,24,24,0,0,0
9030 DATA 252,255,199,195,193,19
3,193,193,2,7,31,189,184,176,176
,191
9040 DATA 3,7,206,236,236,108,11
0,231,227,243,51,3,3,3,3,227
9050 DATA 142,6,6,6,6,6,6,254,19
3,193,193,193,195,199,255,252
9060 DATA 191,176,176,176,176,48
,48,56,227,96,96,96,96,108,111,2
31
9070 DATA 243,59,27,27,27,59,243
,227,254,6,6,6,6,6,6,142
9080 DATA 255,255,193,192,192,19
2,192,192,0,129,195,231,118,54,5
4,54
9090 DATA 252,254,135,3,1,1,1,1,
63,63,32,128,128,128,128,128
9100 DATA 255,255,193,192,192,19
2,192,192,192,192,192,192,193,19
5,255,254
9110 DATA 54,54,118,230,199,131,
1,0,1,1,1,1,3,135,254,252
9120 DATA 128,128,128,128,128,0,
0,1,192,192,192,192,192,192,192,
224
9130 DATA "12","2111","2121","21
1","1","1121","221","1111","11",
"1222","212","1211","22","21","2
22","1221","2212","121","111","2
","112","1112","122","2112","212
2","2211"
9200 PRINT INK 3; PAPER 6;" MOR
SE TRAINER"; INK 1;" BY PAUL ST
ANLEY "
9210 PRINT "This is a simple pr
ogram which allows 2 people to
test their morse code."

```

```
9220 PRINT INK 1;"One person en
ters a message. This is then
translated into morse code by
the computer, and the other per
son has to decode it again."
```

```
9230 PRINT "It is played back w
ith tones and visible dots and da
shes. There are 2 speeds at whi
ch the morse is transmitted."
```

```
9240 PRINT INK 1;"The computer
displays the codes on the screen
to help you and you are given
the option of repeating the
message."
```

```
9250 INPUT "Press ENTER to give
it a try"; LINE p$
```

```
9300 CLS
```

```
9800 DIM z$(26,4): DIM l(26)
```

```
9810 FOR f=1 TO 26: READ a$
```

```
9820 LET l(f)=LEN a$: LET z$(f)=
a$
```

```
9830 NEXT f
```

```
9840 RETURN
```

```
9990 CLS : PRINT "Type on other
Spectrum:''' LOAD *"n";1 <
enter>"
```

```
9991 SAVE *"n";1;"a" LINE 10
```

```
9992 RETURN
```



# 23 Bank Robber

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

This is a fun game with what will hopefully prove to be diverting animation. Each player has control of a small-time crook, who moves about the screen collecting ten pence pieces.

Each player's screen displays their own villain and thirty coins. If your opponent manages to pick up a coin, it will disappear from both screens. You can thus locate your opponent by keeping an eye out for the latest coin-snatch.

There are five different screen layouts, culminating in the Bank display. The bank robbers will progress to a new screen when all thirty coins have been collected by both players. After each screen, you will be told how much money you both have, and after the final screen, the richest robber will be the winner.

This game is fun to play as it stands, but if you prefer to speed things up a bit by playing with less coins, you can simply change line 3000. Let's say, for example, that you only want to collect twenty-five coins from each screen, you should change line 3000 to read:

```
3000 IF m1+m2<25 THEN RETURN
```

## Program Notes:

- 1-110 Call starting routines
- 200-290 Ensures that one computer transmits and one receives
- 1000-1040 Read the keyboard
  - Move your robber
  - Check for walls and coins
- 1045 Transmit robber's position and money collected
- 1050-1090 Print and erase your robber

2000 Receive the co-ordinate of your opponent's robber and the money he has collected  
 2010-2030 Print the second position of your man  
 3000-3040 Check that all the coins have been collected  
     Increase the scores  
     Move on to the next screen  
 4000-4300 Print the final score  
     Wait for a key to be pressed  
 7000-7080 Print thirty coins  
 7510-7600 Set up the variables  
 8000-8550 Draw the five different screens  
 9000-9050 Define the graphics characters  
 9100-9980 Instructions  
     9990 Send the program to the other computer

```

1 REM BANK ROBBER @ P STANLEY
5 GO SUB 9990
10 CLS : BORDER 1: PAPER 0: IN
K 7: CLS : POKE 23750,1
20 POKE 23658,8
30 INPUT "Which Station Are Yo
u?(1 or 2)";st
35 IF st<>1 AND st<>2 THEN GO
TO 30
40 FORMAT "n";st
50 LET st1=3-st
60 GO SUB 9000
90 LET scr=1: LET tm1=0: LET t
m2=0
110 GO SUB 7000
200 GO SUB 1000*st
210 GO SUB 3000
230 GO SUB 1000*st1
240 GO SUB 3000
290 GO TO 200
1000 LET y1=y: LET x1=x
1010 LET y=y+(INKEY$="M" AND y<2
0)-(INKEY$="K" AND y>0)
1020 LET x=x+(INKEY$="X" AND x<3
0)-(INKEY$="Z" AND x>0)
1030 IF ATTR (y,x)>64 OR ATTR (y
+1,x)>64 OR ATTR (y,x+1)>64 OR A
TTR (y+1,x+1)>64 THEN LET y=y1:
LET x=x1

```

```

1040 IF ATTR (y,x)=6 OR ATTR (y+
1,x)=6 OR ATTR (y,x+1)=6 OR ATTR
(y+1,x+1)=6 THEN LET m1=m1+1:
BEEP .02,15
1045 OPEN #4;"n";st1: PRINT #4;y
: PRINT #4;x: PRINT #4;m1: CLOSE
#4
1050 PRINT AT y1,x1;" ";AT y1+1
,x1;" "
1060 PRINT AT y,x;"AB";AT y+1,x;
"CD"
1090 RETURN
2000 OPEN #4;"n";st1: INPUT #4;q
: INPUT #4;p: INPUT #4;m2: CLOSE
#4
2010 PRINT AT y,x;"AB";AT y+1,x;
"EF"
2020 PRINT AT q,p;" ";AT q+1,p;
" "
2030 RETURN
3000 IF m1+m2<30 THEN RETURN
3010 LET scr=scr+1
3012 CLS : LET tm1=tm1+m1: LET t
m2=tm2+m2
3015 PRINT AT 9,0;"You have £";I
NT (tm1/10);".";10*(tm1/10-INT (
tm1/10));0
3017 PRINT AT 12,0;"Your opponen
t has £";INT (tm2/10);".";10*(tm
2/10-INT (tm2/10));0
3025 FOR f=1 TO 10 STEP .3: BEEP
.02,f: BEEP .02,10-f: NEXT f
3027 IF scr=21 THEN GO TO 4000
3030 GO SUB 7000
3040 RETURN
4000 IF tm1>tm2 THEN PRINT AT 1
5,6; FLASH 1;" YOU ARE THE WINNE
R "
4010 IF tm1<tm2 THEN PRINT AT 1
5,6; FLASH 1;" YOU ARE THE LOSER
"
4020 IF tm1=tm2 THEN PRINT AT 1
5,8; FLASH 1;" IT IS A DRAW "

```

```

4030 PRINT INK 6;AT 20,2;"PRESS
  ANY KEY TO PLAY AGAIN."
4200 IF INKEY$="" THEN GO TO 42
00
4300 CLS : GO TO 90
7000 GO SUB 7900+100*scr
7010 RANDOMIZE scr*100
7020 FOR f=1 TO 30
7030 LET a=INT (RND*20)+1
7040 LET b=INT (RND*30)+1
7050 IF ATTR (a,b)>64 THEN GO T
O 7030
7055 IF ATTR (a,b)=6 THEN GO TO
  7030
7060 IF a+b<2 OR a+b>49 THEN GO
  TO 7030
7065 IF ATTR (a+1,b)=6 OR ATTR (
a-1,b)=6 OR ATTR (a,b-1)=6 OR AT
TR (a,b+1)=6 THEN GO TO 7030
7070 PRINT INK 6;AT a,b;"6"
7080 NEXT f
7510 LET y=0: LET x=y
7520 IF st=2 THEN LET y=20: LET
  x=30
7530 LET m1=0: LET m2=0
7600 RETURN
8000 CLS
8010 FOR f=2 TO 19: PRINT BRIGH
T 1; INK 3;AT f,15;"H": NEXT f
8020 FOR f=2 TO 29: PRINT BRIGH
T 1; INK 3;AT 11,f;"H": NEXT f
8030 RETURN
8100 CLS
8110 FOR f=4 TO 27: PRINT INK 4
; BRIGHT 1;AT 4,f;"H";AT 17,f;"H
": NEXT f
8120 FOR f=4 TO 17: PRINT INK 4
; BRIGHT 1;AT f,4;"H";AT f,27;"H
": NEXT f
8130 PRINT AT 4,15;"  "
8140 PRINT AT 17,15;"  "
8150 RETURN
8200 CLS

```

```

8210 FOR f=1 TO 11: PRINT BRIGHT 1; INK 5; AT 1+f, 13-f; "H": NEXT f
8220 FOR f=1 TO 11: PRINT BRIGHT 1; INK 5; AT 8+f, 30-f; "H": NEXT f
8230 FOR f=1 TO 12: PRINT BRIGHT 1; INK 5; AT 4+f, 9+f; "H": NEXT f
8240 RETURN
8300 CLS
8310 RANDOMIZE 410
8320 FOR q=2 TO 29 STEP 3
8330 FOR f=0 TO 21
8340 PRINT BRIGHT 1; INK 2; AT f, q; "H": NEXT f
8345 LET a=INT (RND*20): PRINT AT a, q; " "; AT a+1, q; " "
8350 NEXT q
8380 NEXT f: NEXT q
8390 RETURN
8400 CLS : RESTORE 9050
8405 BRIGHT 1: INK 3
8410 FOR q=1 TO 7: READ a
8420 FOR f=2 TO 8: PRINT AT f, a; "H": NEXT f
8430 NEXT q
8440 PRINT AT 2, 7; "HH"; AT 8, 7; "HH"; AT 5, 7; "HH "
8450 PRINT AT 2, 12; "HH"; AT 5, 12; "HH"
8460 FOR f=1 TO 3: PRINT AT 2+f, 16+f; "H"; AT f+1, 27-f; "H"; AT 5+f, 23+f; "H": NEXT f
8470 PRINT AT 5, 23; "H"; AT 6, 23; " "
8480 INK 4
8490 FOR f=5 TO 26: PRINT AT 11, f; "H"; AT 19, f; "H": NEXT f
8500 FOR f=12 TO 19: PRINT AT f, 5; "H"; AT f, 26; "H": NEXT f
8510 FOR f=12 TO 16: PRINT AT f, 13; "H"; AT f, 18; "H": NEXT f

```

```

8520 FOR f=14 TO 18: PRINT AT f,
10;"H";AT f,21;"H": NEXT f
8530 INK 7: BRIGHT 0
8540 PRINT AT 11,15;"  ";AT 19,6
;"  ";AT 19,24;"  "
8545 LET scr=20
8550 RETURN
9000 FOR f=USR "a" TO USR "h"+7:
  READ q: POKE f,q: NEXT f
9010 DATA 3,12,15,29,31,8,12,2,1
92,48,240,184,248,16,48,64
9020 DATA 3,1,7,9,19,2,2,6,192,1
28,224,144,200,64,64,96
9030 DATA 19,9,7,1,7,12,56,16,20
0,144,224,128,224,48,28,8
9040 DATA 60,126,219,213,213,219
,126,60,127,85,127,85,127,85,127
,0
9050 DATA 6,9,11,14,16,20,22
9100 PRINT PAPER 1;" AB"; INK 6
;"  B A N K R O B B E R  "; I
NK 7;"AB "
9110 PRINT PAPER 1;" CD"; INK 5
;"  @ PAUL STANLEY 1984  "; I
NK 7;"EF "
9120 PRINT "This is a simple co
mpetition      between 2 small-tim
e crooks to  rob more cash than
the opponent."
9130 PRINT "There are 5 differe
nt screens    and on each there a
re 30 coins   to be collected. Yo
u cannot see your opponent, but
if he takes  some cash it will d
isappear fromyour screen."
9140 PRINT "The winner is the p
layer who has the most money afte
r 5 screens."
9150 PRINT INK 5;"CONTROLS:"
9160 PRINT INK 6;"      Z...LEFT
      X...RIGHT      K.....UP
      M....DOWN"
9170 PRINT #1;"Press ENTER to st
art."

```

```

9180 PRINT PAPER 1;AT 1,1;"EF";
AT 1,29;"CD"
9185 FOR h=1 TO 15: NEXT h
9190 IF INKEY$<>" " THEN GO TO 9
900
9200 PRINT PAPER 1;AT 1,1;"CD";
AT 1,29;"EF"
9205 FOR h=1 TO 15: NEXT h
9210 IF INKEY$<>" " THEN GO TO 9
900
9220 GO TO 9180
9900 INPUT ""
9910 CLS
9920 BEEP .2,10
9980 RETURN
9990 CLS : PRINT "Type on other
Spectrum:''" LOAD *""n"";1 <e
nter>"
9991 SAVE *""n"";1;"a" LINE 10
9992 RETURN

```

# 24 Money Spiders

## Hardware requirements

Two 16/48K Spectrums on ZX network

Two televisions

This is a novel gambling game for those of you who find it more diverting to exploit your luck than test your judgement. In other words, Money Spiders is primarily a game of chance.

There are ten black money spiders standing on a wall and under the right circumstances they can all be persuaded to fall! You're asked to bet on one spider by parting with some of your £100 kitty. If you and your opponent bet on the same spider, then you both win. If you bet on different spiders, they both disappear and you must try again.

There may be no spiders left if you have placed your money on different spiders to your opponent, under these circumstance a fresh batch of ten spider appear on the wall.

As the spiders disappear, the chances of betting on the same one as your opponent increase, therefore the odds decrease and you will win less money. Another way of increasing your winning is if you happen to bet on one of the three Money Spiders, which will earn you £10.

This is a very addictive game which features some unusual graphics and flesh-crawlingly realistic animation.

## Program Notes

- 1-90 Call the starting routines
- 100-225 Input the spider number and the bet
  - Compare the spider number with your opponent's
- 230-999 Call for the spider animation routines
- 1000-1030 Print the spider in four different positions
- 2000-2025 Remove the spiders
  - Print £10 if required
- 2030-2060 Increase or decrease the player's money



2070-2080 Transmit how much money you have left  
 Receive how much money your opponent has  
 2090-2130 Check for bankruptcy  
 Print the new odds  
 3000-3150 Print the final result  
 Wait for a key to start again  
 4000-4070 Compare the final quantities of money to decide who wins  
 7000-7120 Transmit your spider number  
 Receive your opponent's spider number  
 7500-7540 Clear the top few lines of the screen  
 8000-8200 Set up the screen display  
 8500-8999 Input the number of rounds on station one  
 Station one transmits the number of rounds to station two  
 9000-9100 Define the graphics characters  
 9200-9989 Instructions  
 9990 Send the program to the other Spectrum

```

1 REM *** MONEY SPIDER ***
2 REM ** @ Paul Stanley **
5 GO SUB 9990
10 CLS : BORDER 7: PAPER 7: IN
K 0: CLS
15 POKE 23750,7
17 POKE 23609,25
18 GO SUB 9000
20 INPUT "Station Number (1 or
2): ";st
21 IF st<>1 AND st<>2 THEN GO
TO 20
22 FORMAT "n";st: LET st1=3-st
50 GO SUB 8500
90 GO SUB 8000
100 PRINT AT 0,0;"YOU HAVE £";m
1
110 PRINT AT 2,0;"YOUR OPPONENT
HAS £";m2
120 PRINT AT 4,0;"WHICH SPIDER
DO YOU WANT TO BET ON? ";
130 INPUT "Spider Number: ";sp
135 IF sp<1 OR sp>10 THEN BEEP
.3,-20: GO TO 130
140 IF m(sp)=0 THEN BEEP .3,-2
0: GO TO 130
150 PRINT INK 2;"SPIDER ";sp

```

```

160 PRINT "How much do you want to bet?"
170 INPUT "Bet: £";bet
180 IF bet=0 OR bet>m1 OR bet<>INT bet THEN GO TO 170
190 GO SUB 7500
200 GO SUB 6900+100*st
210 PRINT AT 0,0;"You chose spider ";sp
220 PRINT INK 1;"Your opponent chose spider ";sp1
225 IF sp=sp1 THEN PRINT FLASH 1 INK 2;"YOU HAVE BET ON THE SAME SPIDER!"
230 LET x(1)=sp*3-2
240 LET x(2)=sp1*3-2
910 LET i=0: FOR f=1 TO 6
920 FOR q=0 TO 3: FOR l=1 TO 2: GO SUB 1000+10*q
930 NEXT l: NEXT q: NEXT f
940 FOR i=0 TO 5
950 FOR q=0 TO 3: FOR l=1 TO 2: GO SUB 1000+10*q
960 NEXT l: NEXT q: NEXT i
970 GO SUB 2000
980 GO SUB 7500
990 GO TO 100
1000 PRINT INK i;AT 13,x(1);"ABC";AT 14,x(1); INK i+1;"DEF": RETURN
1010 PRINT INK i;AT 13,x(1);"GHC"; INK i+1;AT 14,x(1);"IJK": RETURN
1020 PRINT INK i;AT 13,x(1);"GLC";AT 14,x(1); INK i+1;"MNO": RETURN
1030 PRINT INK i;AT 13,x(1);"GHC"; INK i+1;AT 14,x(1);"IJK": RETURN
2000 PRINT AT 13,x(1);" ";AT 13,x(2);" "
2005 PRINT AT 14,x(1);" ";AT 14,x(2);" "

```

```

2010 IF m(sp)=10 THEN PRINT IN
K 4;AT 14,x(1);"$S ": LET m1=m1+
10
2020 IF m(sp1)=10 THEN PRINT I
NK 4;AT 14,x(2);"$S "
2025 FOR f=1 TO 10: BEEP .03,f:
BEEP .03,10-f: NEXT f
2030 LET m1=m1-bet
2035 LET m(sp)=0: LET m(sp1)=0
2040 IF sp=sp1 THEN LET m1=m1+b
et+no*bet
2050 IF sp<>sp1 THEN LET no=no-
1
2060 LET no=no-1
2070 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;m1: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;m2: CLOSE #
4
2080 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;m2: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;m1: CLOSE #
4
2090 IF sp=sp1 THEN GO TO 8000
2100 IF m1=0 THEN GO TO 3000
2110 IF m2=0 THEN GO TO 3100
2115 IF no=0 THEN GO TO 8000
2120 PRINT AT 20,18; INK 1;" ODD
S: ";no;"-1 "
2130 RETURN
3000 CLS
3010 PRINT AT 4,0;"YOU HAVE RUN
OUT OF MONEY!";AT 6,0;"YOUR OPPO
NENT IS THE WINNER."
3020 GO TO 3120
3100 CLS
3110 PRINT AT 4,0;"YOUR OPPONENT
HAS RUN OUT OF MONEY!";AT 7,
0;"YOU ARE THE WINNER."
3120 FOR f=1 TO 20: BEEP .1,f: N
EXT f
3130 PRINT AT 15,0; INK 3;"Press
any key to play again."
3140 IF INKEY$="" THEN GO TO 31
40

```

```

3150 CLS : GO TO 50
4000 CLS
4010 PRINT AT 2,0;"You have £";m
1
4020 PRINT AT 4,0;"Your opponent
has £";m2
4030 IF m1>m2 THEN PRINT AT 7,0
; INK 2;"Therefore you are the w
inner!"
4040 IF m1<m2 THEN PRINT AT 7,0
; INK 2;"Therefore your opponent
is the winner!"
4050 IF m1=m2 THEN PRINT AT 7,0
; INK 2;"Therefore it is a draw!"
"
4070 GO TO 3120
7000 OPEN #4;"n";st1: PRINT #4;s
p: CLOSE #4
7010 OPEN #4;"n";st1: INPUT #4;s
p1: CLOSE #4
7020 RETURN
7100 OPEN #4;"n";st1: INPUT #4;s
p1: CLOSE #4
7110 OPEN #4;"n";st1: PRINT #4;s
p: CLOSE #4
7120 RETURN
7500 FOR f=0 TO 9
7510 PRINT AT f,0;"
"
7520 BEEP .02,f
7530 NEXT f
7540 RETURN
8000 LET w$="PQPQPQPQPQPQPQPQP
QPQPQPQPQPQPQPQPQP"
8002 LET r=r-1: IF r<0 THEN GO
TO 4000
8005 CLS
8007 LET no=10
8010 FOR f=15 TO 21
8020 LET a=(f/2=INT (f/2))
8030 PRINT AT f,0; INK 2; PAPER
6;w$(1+a TO 32+a)
8040 NEXT f

```

```

8042 PRINT INK 1;AT 20,2;" ROUN
DS:";r+1;" ";AT 20,18;" ODDS: ";
no;"-1 "
8050 FOR f=1 TO 28 STEP 3
8060 PRINT AT 13,f;"ABC"
8070 PRINT INK 1;AT 14,f;"DEF"
8080 PRINT INK INT (RND*4);AT 1
1,f+1;"R"
8090 IF f<28 THEN PRINT INK IN
T (RND*4);AT 10,f+1;(f+2)/3
8100 IF f=28 THEN PRINT INK 3;
AT 10,f+1;"S"
8110 NEXT f
8120 DIM m(10): DIM x(2)
8125 FOR f=1 TO 10: LET m(f)=1:
NEXT f
8130 FOR f=1 TO 3
8140 LET a=INT (RND*10)+1
8150 IF m(a)=10 THEN GO TO 8140
8160 LET m(a)=10
8170 NEXT f
8200 RETURN
8500 LET m1=100: LET m2=m1
8502 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;r: CLOSE #4: RETURN

8505 PRINT INK 1'"A ROUND IS CO
MPLETE WHEN BOTH PLAYERS HAVE
PLACED THEIR MONEY ON THE SAME S
PIDER."
8510 PRINT '"DECIDE HOW MANY RO
UNDS YOU WISH TO PLAY THEN TYPE
IN THIS FIGUREAND PRESS"; INK 2;
" ENTER"
8520 INPUT "Number Of Rounds: ";
r
8525 IF r<1 THEN GO TO 8520
8530 OPEN #4;"n";st1: PRINT #4;r
: CLOSE #4
8999 RETURN
9000 FOR f=USR "a" TO USR "s"+7:
READ q: POKE f,q: NEXT f
9010 DATA 0,3,6,14,15,15,7,131,1
26,255,79,79,255,255,255,255

```

```

9020 DATA 0,192,224,240,240,240,
224,192,193,126,1,2,2,4,4,24
9030 DATA 170,170,73,68,68,66,36
,36,129,113,14,128,128,64,64,32
9040 DATA 0,3,7,15,15,15,7,3,126
,255,147,147,255,255,255,255
9050 DATA 1,198,56,0,1,1,2,2,170
,165,146,146,18,9,8,16
9060 DATA 128,96,152,70,65,64,19
2,160,126,255,228,228,255,255,25
5,255
9070 DATA 1,2,68,56,0,0,0,0,17,1
69,84,82,74,73,132,132
9080 DATA 128,64,176,72,40,20,14
4,136
9090 DATA 0,127,127,127,127,127,
127,0,0,254,254,254,254,254,254,
0
9100 DATA 56,56,56,56,254,124,56
,16,0,70,201,73,73,73,230,0
9200 LET r=11: GO SUB 8000
9220 FOR f=1 TO 18: PRINT : FOR
q=1 TO 5: BEEP .01,q: NEXT q: PO
KE 23692,200: NEXT f
9240 PRINT AT 0,0: PAPER 5: INK
1;"    M O N E Y    S P I D E R S
"
9250 PRINT "WRITTEN & DESIGNED B
Y P.STANLEY"
9260 PRINT AT 16,0;"These are th
e money spiders!"
9270 PRINT INK 1;"This game is
a gambling game    based mainly
on luck, but you    do need to ju
dge how much money to bet."
9280 INPUT "Press ENTER ..."; LI
NE p$
9290 CLS : PRINT "The object is
to guess which    spider your op
ponent is going to bet on, and th
en place a bet on that one."

```

```

9300 PRINT INK 1;"If you do bet
on the same one then you get
the odds shown. Three spiders
are true ""money spiders"",an
d if you bet on one of these yo
u will get £10."
9310 PRINT '"If you bet on diffe
rent spiders then these disappea
r, and the odds decrease. When
all the spiders have gone,
or you bet on the same one the ro
und is over."
9320 PRINT INK 1;"The game cont
inues until all the rounds have b
een played or one player has no
more money."
9330 INPUT "Press ENTER to start
."; LINE p$
9335 BEEP .1,20
9340 CLS
9989 RETURN
9990 CLS : PRINT '"Type on other
Spectrum:"
9991 PRINT '" LOAD *""n"";1
<enter>"
9992 SAVE *""n"";1;"a" LINE 10
9993 RETURN

```

# 25 Art Dealer

## Hardware requirements

Two 48K Spectrums on ZX network

Two televisions

This rather lengthy program gives you the chance to wheel and deal in millions of pounds and agent unique paintings by the world famous artist, Simon Pectrum. You are given £100,000 at the start of the game and this may be topped up as business progresses.

For each move, the computer will randomly select one of four possibilities. The first is that your paintings will be displayed in Europe. This earns you £250,000. The second option offers the chance of buying a painting at an auction. Here, you bid against your opponent for a painting and the highest bidder will get the painting and a price tag. This lets you know how much the painting is *actually* worth and can be anything between £1,000 and £1,000,000. It might even turn out that you've been sold a forgery, but whatever the real value of the purchase, it is only made known to the buyer.

Another possible option allows players to sell a painting to the Bank for a tenth of £1,000,000. Obviously you should seize the opportunity to sell low value paintings, preferably the forgeries. Finally, you may be asked if you want to buy a painting from your opponent's gallery for £100,000. You don't know the values of these paintings (unless it was you who sold them), and so a certain degree of luck is needed.

This is an entertaining game, which is well worth the typing marathon required to key it in. After all, it is probably the only opportunity you will have to deal with such large sums of money!



**Program Notes:**

- 5-45 Call starting routines
- 50-100 Enter, transmit and receive players' names
- 200-250 Select a random move
  - Ensure that one player transmits while the other receives
- 1000-1020 Receive move type from the other Spectrum
- 1100-1110 Opponent sell pictures in Europe
- 1200-1240 Receive whether opponent is selling a picture to the bank or not
  - Erase pictures if necessary
- 1300-1385 Receive whether opponent is buying a picture from you and the picture number.
  - Transmit the value of the painting and the random number
- 1400-1440 Bank auction is only possible if both players have less than eight paintings
- 2000-2020 Transmit the move type to the other computer
- 3000-3520 Increase your money for European display
- 4000-4520 Input a painting number, check it and erase the painting
  - Transmit whether a painting is sold or not
- 5000-5580 Input a painting number for buying your opponent's pictures
  - Transmit whether you are buying a painting or not and the number
  - Receive the value of the painting and the random number
- 6000-6065 Bank auction is only possible if both players have less than eight paintings
  - Input and transmit a bid
- 6070-6110 Receive a bid
- 6200-2300 Decrease the number of paintings in the bank
  - Fetch the painting value and find a place in the gallery
  - Transmit the value and random number of the painting
- 6400-6430 Receive the value and random number of the painting
- 6500-6680 Total up the players' money
  - Transmit and receive their money
  - Compare and decide upon the winner
  - Wait for a key to play again
- 7000-7100 Create a painting
- 7800-7890 Print the value of the painting
- 7900-8910 Clear the bottom of the screen
- 8000-8999 Set up the screen and variables
- 9000-9015 Define the graphics characters
- 9100-9110 Painting values
- 9200-9989 Instructions
- 9990-9993 Send the program to the other Spectrum

```

1 REM Art Dealer- @ P.Stanley
5 GO SUB 9990
10 CLS : BORDER 2: PAPER 7: IN
K 0: CLS
15 POKE 23750,2
20 POKE 23658,8
25 GO SUB 9000
30 INPUT "Station No? (1 or 2)
";st
35 IF st<>1 AND st<>2 THEN GO
TO 30
40 FORMAT "n";st: LET st1=3-st
45 BEEP .2,14
50 PRINT "ENTER Your First Na
me:": INPUT a$
60 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;a$: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;b$: CLOSE #
4
70 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;b$: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;a$: CLOSE #
4
100 GO SUB 8000
200 LET move=INT (RND*4)+1
210 GO SUB 1000*st1
220 LET move=INT (RND*4)+1
230 GO SUB 1000*st
250 GO TO 200
1000 OPEN #4;"n";st1: INPUT #4;m
ove: CLOSE #4
1010 GO SUB 1000+100*move
1020 RETURN
1100 PRINT AT 17,0;b$;"'S PAINTI
NGS ARE'"BEING SHOWN IN EUROPE.
"
1110 GO TO 3500
1200 PRINT AT 17,0;"YOUR OPPONEN
T MAY SELL A PICTURE TO THE BANK
FOR £100 000"
1210 OPEN #4;"n";st1: INPUT #4;x
: INPUT #4;p: CLOSE #4
1215 IF p=0 THEN PRINT "Your o
pponent is not going to sell a
picture.": GO TO 3500

```

```

1220 FOR f=11 TO 14: PRINT AT f,
x; PAPER 1;"      ": NEXT f
1225 LET p2=p2-1: LET q(p)=1
1230 GO SUB 7900
1240 RETURN
1300 PRINT AT 17,0;"Your opponen
t may buy one of      your picture
s without consent."
1310 OPEN #4;"n";st1: INPUT #4;s
p: INPUT #4;p: CLOSE #4
1320 IF sp=1 THEN PRINT '"Your
opponent cannot afford the £100
000 asking price.": GO TO 3500
1330 IF sp=2 THEN PRINT '"Your
opponent cannot have any- more
paintings.": GO TO 3500
1340 IF sp=3 THEN PRINT '"You h
ave no paintings to offer.": GO
TO 3500
1345 IF p=0 THEN GO TO 1355
1350 LET ran=r(p): LET v=h(p)
1353 OPEN #4;"n";st1: PRINT #4;v
: PRINT #4;ran: CLOSE #4
1355 IF p=0 THEN PRINT 'b$;" do
esn't want one.": GO TO 3500
1357 OPEN #4;"n";st1: INPUT #4;o
: CLOSE #4
1360 LET p1=p1-1: LET p2=p2+1
1370 FOR f=2 TO 7: PRINT PAPER
1;AT f,p*4-4;"      ": NEXT f
1381 LET q(o)=v: LET h(p)=1
1382 LET y=11: LET x=o*4-4
1383 GO SUB 7000
1384 GO SUB 7900
1385 RETURN
1400 PRINT AT 17,0; INK 1;"BANK
AUCTION"
1410 IF p1=8 THEN PRINT '"The a
uction must be cancelled becau
se you have 8 paintings.": GO TO
3500
1420 IF p2=8 THEN PRINT '"The a
uction must be cancelled becau
se your opponent has eight paint
ings.": GO TO 3500

```

```

1430 LET b=0
1440 GO TO 6070
2000 OPEN #4;"n";st1: PRINT #4;m
ove: CLOSE #4
2010 GO SUB 2000+1000*move
2020 RETURN
2999 REM ** Gain £250 000 **
3000 PRINT AT 17,0;"YOUR PAINTIN
GS ARE TO BE SHOWN IN EUROPE. T
HIS WILL EARN YOU £250 000."
3010 LET m1=m1+250000
3500 FOR f=1 TO 400: NEXT f
3510 GO SUB 7900
3520 RETURN
3999 REM ** Sell to bank **
4000 PRINT AT 17,0;"You can sell
a painting to the bank for £10
0 000."
4010 IF p1=0 THEN PRINT '"You h
ave no paintings and so canno
t sell one.': LET p=0: LET x=p:
GO TO 4500
4030 PRINT '"ENTER the number of
the painting you want to sell (o
r 0)."'
4040 INPUT "Painting Number? ";p
4050 IF p=0 THEN GO TO 4500
4055 IF p<1 OR p>8 THEN GO TO 4
040
4060 IF h(p)=1 THEN GO TO 4040
4070 LET x=p*4-4: FOR f=2 TO 7:
PRINT AT f,x; PAPER 1;" ": NE
XT f
4080 LET h(p)=1
4085 LET p1=p1-1
4090 LET m1=m1+1000000
4500 OPEN #4;"n";st1: PRINT #4;x
: PRINT #4;p: CLOSE #4
4505 IF p=0 THEN GO TO 3500
4510 GO SUB 7900
4520 RETURN
4999 REM *Buy From Other Player*

```

```

5000 PRINT AT 17,0;"You can buy
one of ";b$;"'s'"paintings for
£100000"
5010 LET sp=0: IF m1<100000 THEN
    LET sp=1: PRINT "Unfortunately
you cannot afford £100000.": G
O TO 5040
5020 IF p1=8 THEN LET sp=2: PRI
NT "However, you cannot hold mo
re than 8 paintings.": GO TO 5
040
5030 IF p2=0 THEN LET sp=3: PRI
NT "Unfortunately, your opponen
t has no pictures to sell.": GO T
O 5040
5040 IF sp<>0 THEN LET p=0: GO
TO 5500
5050 PRINT "Which painting do y
ou want to buy? (0 if you do n
ot want one)"
5055 INPUT "Painting Number: ";p
5060 IF p=0 THEN GO TO 5500
5065 IF p<1 OR p>8 THEN GO TO 5
055
5070 IF q(p)=1 THEN GO TO 5055
5080 LET q(p)=1: LET c=1
5090 LET x=4*p-4: FOR f=11 TO 14
: PRINT PAPER 1;AT f,x;" ":
NEXT f
5500 OPEN #4;"n";st1: PRINT #4;s
p: PRINT #4;p: CLOSE #4
5505 IF p=0 THEN GO TO 3500
5510 OPEN #4;"n";st1: INPUT #4;v
: INPUT #4;ran: CLOSE #4
5515 LET y=2
5520 FOR f=1 TO 8
5530 IF h(f)<>1 THEN NEXT f
5535 LET p=f: OPEN #4;"n";st1: P
RINT #4;p: CLOSE #4
5537 LET x=p*4-4: GO SUB 7000
5540 LET h(p)=v
5545 LET r(p)=ran
5550 GO SUB 7800

```

```

5560 LET m1=m1-100000: LET m2=m2
+100000
5565 LET p1=p1+1: LET p2=p2-1
5570 GO SUB 7900
5580 RETURN
5999 REM ** Bank Auction **
6000 PRINT AT 17,0; INK 1;"BANK
AUCTION"
6010 IF p1=8 THEN PRINT "The a
uction must be cancelled becau
se you have 8 paintings.": GO TO
3500
6020 IF p2=8 THEN PRINT "The a
uction must be cancelled becau
se your opponent has eight paint
ings.": GO TO 3500
6025 LET hi=0
6030 PRINT AT 19,0;"You have £";
m1;TAB 0;"What is your bid? (0 i
f no bid)"
6040 INPUT "BID: ";b
6050 IF b<>0 THEN IF b>m1 OR b<
=hi THEN GO TO 6040
6060 OPEN #4;"n";st1: PRINT #4;b
: CLOSE #4
6065 IF b=0 THEN GO TO 6400
6070 PRINT AT 19,0;"Please wait
for your opponent's bid. ";c$
6080 OPEN #4;"n";st1: INPUT #4;h
i: CLOSE #4
6085 IF hi=0 THEN PRINT AT 19,0
;"Your opponent does not want th
e painting. ";c$: FOR f=1 TO 200:
NEXT f: GO TO 6200
6090 PRINT AT 19,0;"Your opponen
t offered £";hi;c$
6100 FOR f=1 TO 200: NEXT f
6110 GO TO 6030
6200 LET m1=m1-b
6205 LET p1=p1+1
6210 FOR f=1 TO 8
6220 IF h(f)<>1 THEN NEXT f
6230 LET p=f: LET h(p)=v(bank):
LET bank=bank+1

```

```

6240 LET r(p)=INT (RND*65000): L
ET ran=r(p)
6250 LET x=p*4-4: LET y=2
6260 OPEN #4;"n";st1: PRINT #4;r
an: PRINT #4;p: PRINT #4;h(p): P
RINT #4;bank: CLOSE #4
6265 IF BANK=21 THEN GO TO 6500
6270 GO SUB 7000
6280 LET v=h(p): GO SUB 7800
6290 GO SUB 7900
6300 RETURN
6400 OPEN #4;"n";st1: INPUT #4;r
an: INPUT #4;p: INPUT #4;q(p): I
NPUT #4;bank: CLOSE #4
6405 IF BANK=21 THEN GO TO 6500
6410 LET x=p*4-4: LET y=11: GO S
UB 7000
6420 LET m2=m2-hi: LET p2=p2+1
6430 GO SUB 7900: RETURN
6500 CLS : PRINT ""There are no
more paintings in the bank. Th
erefore the game is complete."
6510 PRINT INK 1'"All that rema
ins is to decide upon the winn
er!"
6520 FOR f=1 TO 8
6530 IF h(f)<>1 THEN LET m1=m1+
h(f)
6540 NEXT f
6550 IF st=1 THEN OPEN #4;"n";s
t1: PRINT #4;m1: CLOSE #4: OPEN
#4;"n";st1: INPUT #4;m2: CLOSE #
4
6560 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;m2: CLOSE #4: OPEN
#4;"n";st1: PRINT #4;m1: CLOSE #
4
6570 FOR f=1 TO 30: BEEP .1,f-20
: BEEP .1,f: NEXT f: CLS
6580 PRINT INK 1'"YOU HAVE £";M
1
6590 PRINT INK 2'"YOUR OPPONENT
HAS £";M2

```

```

6600 IF m1>m2 THEN PRINT ""YOU
ARE THE WINNER!"
6610 IF m1<m2 THEN PRINT ""YOU
R OPPONENT IS THE WINNER!"
6620 IF m1=m2 THEN PRINT ""THA
T'S INCREDIBLE - IT'S A DRAW!"
6630 PRINT AT 13,0; INK 3;"Press
any key to play again"
6640 IF INKEY$="" THEN GO TO 66
40
6650 BEEP .1,20: CLS
6680 GO TO 50
7000 RANDOMIZE ran
7010 FOR f=y TO y+3
7020 FOR q=x TO x+2
7025 FOR h=1 TO 3
7030 PRINT OVER 1; INK INT (RND
*7); PAPER 7;AT f,q;CHR$ (INT (R
ND*94)+32)
7040 NEXT h: NEXT q: NEXT f
7060 RANDOMIZE
7100 RETURN
7800 LET 1$=STR$ v
7810 IF v=0 THEN PRINT AT 7,x;
PAPER 2; INK 7;"CDEF": RETURN
7820 PRINT AT 7,x; PAPER 0; INK
6;1$(1);"A";
7830 LET 1=LEN 1$
7840 IF 1=4 THEN PRINT PAPER 0
; INK 6;"B"
7850 IF 1=5 THEN PRINT PAPER 0
; INK 6;"A"
7860 IF 1=6 THEN PRINT PAPER 0
; INK 6;"AB"
7870 IF 1=7 THEN PRINT PAPER 0
; INK 6;"AA"
7890 RETURN
7900 FOR f=16 TO 21: PRINT AT f,
0;c$: NEXT f
7910 RETURN
8000 CLS : LET bank=1
8005 DIM q(8): DIM h(8): DIM r(8
): DIM v(20): DIM p(20)

```



```

8007 FOR f=1 TO 8: LET q(f)=1: L
ET h(f)=1: NEXT f
8008 IF st=1 THEN LET l=INT (RN
D*65000): OPEN #4;"n";st1: PRINT
#4;l: CLOSE #4
8009 IF st=2 THEN OPEN #4;"n";s
t1: INPUT #4;l: CLOSE #4
8010 RANDOMIZE l: RESTORE 9100:
FOR f=1 TO 20: READ b
8020 LET a=INT (RND*20)+1
8030 IF p(a)=1 THEN GO TO 8020
8040 LET p(a)=1: LET v(a)=b
8050 NEXT f
8060 DIM c$(32)
8100 PRINT AT 0,10-(LEN a$)/2; P
APER 5;a$;"'S GALLERY"
8110 FOR f=1 TO 7: PRINT PAPER
1;c$: NEXT f
8120 PRINT AT 9,10-(LEN b$)/2; P
APER 2; INK 6;b$;"'S GALLERY"
8130 FOR f=1 TO 6: PRINT PAPER
1;c$: NEXT f
8140 FOR q=1 TO 10 STEP 9: FOR f
=1 TO 8: PRINT INK 3;AT q,f*4-4
; PAPER 6;f: NEXT f: NEXT q
8200 LET p1=0: LET p2=p1
8210 LET m1=100000: LET m2=m1
8999 RETURN
9000 FOR f=USR "a" TO USR "f"+7:
READ q: POKE f,q: NEXT f
9005 DATA 0,238,170,170,170,170,
238,0,0,224,160,160,160,160,224,
0
9010 DATA 0,29,17,29,17,17,17,0,
0,221,85,93,89,85,213,0
9015 DATA 0,221,17,29,81,81,221,
0,0,212,84,220,136,72,72,0
9100 DATA 0,0,0,0,1000,2000,4000
,10000,20000,30000,40000,50000,7
0000
9110 DATA 80000,100000,200000,40
0000,700000,1000000,1000000
9200 PRINT PAPER 1; INK 6;"
ART DEALER "
```

```

9210 PRINT INK 3;AT 2,6;"@ Paul
Stanley 1984"
9220 PRINT '"This is a complex g
ame where the2 players get to ha
ndle millionsof pounds!"
9230 PRINT '"The computer select
s a move fromthe following:-"
9235 PRINT '"1...European exhibi
tion"' "2...Sale of painting to t
he bank"
9240 PRINT "3...Purchase from ot
her player"' "4...Purchase from b
ank"
9245 INPUT "Press ENTER ..."; LI
NE 1$
9250 CLS : PRINT INK 3;"EUROPEA
N EXHIBITIONS"
9260 PRINT '"Whenever one of the
se occurs, you will be paid £2
50 000 for allowing your paint
ings to be displayed in Europe
.'"
9270 PRINT '"Even if you have no
paintings ondisplay in your gal
lery, you will still be paid
the money.'"
9280 INPUT "Press ENTER ..."; LI
NE 1$
9290 CLS : PRINT INK 3;"SALE TO
THE BANK"
9300 PRINT '"All paintings are b
y that famousartist Simon Pectru
m. They are valued at between £
1000 and one million pounds.'"
9305 PRINT "There are also 3 for
geries."
9310 PRINT '"If you obtain a pai
nting it willhave a price-tag be
neath it. Your opponent cannot
know this value. A sale to th
e bank is always £100 000, so
you could give them a forgery
and they would still pay you
£100 000.'"

```

```

9320 PRINT "Therefore, if you have a picture valued at less than
£100 000 you should sell it."
9330 INPUT "Press ENTER .."; LINE 1$
9340 CLS : PRINT INK 3;"PURCHASE FROM THE OTHER DEALER"
9350 PRINT "If your opponent has a painting in his gallery, and you have
£100 000, you can buy a picture off your opponent without his consent."
9360 PRINT "Unless you have once owned the painting you choose to buy, you do not know how valuable it will be."
9370 INPUT "Press ENTER .."; LINE 1$
9380 CLS : PRINT INK 3;"BANK AUCTION"
9390 PRINT "When this occurs, the 2 dealers can bid against each other to buy the painting offered by the bank."
9400 PRINT "When the bank has sold twenty paintings the game is over, and the winner is the dealer with the most money."
9410 INPUT "Press ENTER to start ."; LINE 1$
9420 CLS
9989 RETURN
9990 CLS : PRINT "Type on the other Spectrum:"
9991 PRINT " LOAD *\"n\"";1 <enter>"
9992 SAVE *\"n\"";1;"a" LINE 10
9993 RETURN

```

**More Pan/PCN Computer Library titles for the  
Sinclair ZX Spectrum**

Robert Erskine & Humphrey Walwyn,  
Paul Stanley & Michael Bews

**Sixty Programs for the Sinclair ZX  
Spectrum** £5.95

0 330 28260 3

A massive software library for the price of a single cassette. Explosive games, dynamic graphics and invaluable utilities, this specially commissioned collection takes BASIC to the limits and beyond.

Four of the country's best-selling software writers have pooled their talents to bury programming clichés and exploit your micro's potential to the full.

Whether you are a games player or a more serious user, here's the book to make your micro work for you.

Jean Frost

**Instant Arcade Games for the Sinclair ZX  
Spectrum £3.95**

0 330 28260 3

For the newcomer to computing, Jean Frost's Instant Arcade Games will be nothing short of a revelation.

With little or no knowledge of BASIC, you can still take a suite of 'skeleton' programs and create your own arsenal of dynamic and totally unique arcade games.

For Spectrum users who already write their own software, Instant Arcade Games offers an invaluable library of imaginative subroutines and user-defined graphics to enhance the efficiency and visual impact of their games programs.

Tony Takoushi

**The Best Software Guide: Spectrum  
Games £3.95**

0 330 284576

Tony Takoushi's devotion to the art of arcade action makes The Best Software Guide a book no games player can afford to be without.

Indexes, including a comprehensive ratings system of available arcade games for the Sinclair ZX Spectrum are combined with reviews and screen-shots of the action, giving you access to the best of the small-screen software. A unique gameplay reference system gives you guidance on the tactics needed to rack up the highest scores.

Ian Scales

**The Spectrum Peripherals Guide £4.95**    0 330 28459 2

The success of the Spectrum has spawned a bewildering array of competing products to enhance and expand the capabilities of the world's most popular micro. The Spectrum Peripherals Guide provides the data you need to cut your way through this jungle. Ian Scales explains the function and characteristics of each type of accessory available, giving you the information needed to assess your requirements, evaluate the advertising claims and make an informed purchase decision.

Robert Erskine & Humphrey Walwyn with Paul Stanley and Michael Bews  
**Sixty Programs for the Sinclair ZX Spectrum** £5.95

**Sixty Programs for the BBC Micro** £5.95

**Sixty Programs for the Dragon 32** £5.95

**Sixty Programs for the Oric 1** £5.95

**Sixty Programs for the Atari** £5.95

**Sixty Programs for the Commodore 64** £5.95

**Sixty Programs for the Vic 20** £5.95

**Sixty Programs for the Electron** £5.95

Ian Adamson

**The Companion to the Oric 1** £5.95

Geoff Wheelwright

**The Companion to the BBC Micro** £4.95

Keith Bowden

**The Companion to the Commodore 64** £5.95

Jeremiah Jones and Geoff Wheelwright

**The Companion to the Electron** £5.95

Jean Frost

**Instant Arcade Games for the Sinclair ZX Spectrum** £3.95

**Instant Arcade Games for the BBC Micro** £3.95

**Instant Arcade Games for the Dragon 32** £3.95

**Instant Arcade Games for the Electron** £3.95

J. J. Clessa

**Micropuzzles** £2.95

Ian Scales

**Spectrum Peripherals Guide** £4.95

Tony Takoushi

**Best Software Guide: Vic 20/Commodore 64** £3.95

**Best Software Guide: Spectrum Games** £3.95

Jeff Aughton

**Invaluable Utilities for your BBC Micro** £5.95

All these books are available at your local bookshop or newsagent, or can be ordered direct from the publisher. Indicate the number of copies required and fill in the form below.

Name \_\_\_\_\_  
(Block letters please)

Address \_\_\_\_\_

---

Send to Pan Books (CS Department), PO Box 40, Basingstoke, Hants. Please enclose remittance to the value of the cover price plus: 35p for the first book plus 15p per copy for each additional book ordered to a maximum charge of £1.25 to cover postage and package. Applicable only in the UK.

While every effort is made to keep prices low, it is sometimes necessary to increase prices at short notice. Pan Books reserve the right to show on covers and charge new retail prices which may differ from those advertised in the text or elsewhere.



## **MULTI-PLAYER ACTION ON YOUR SPECTRUM!**

The arrival of the long-awaited ZX Microdrive provides Spectrum users with an interfacing capability that dramatically enhances the power of the world's most popular micro. The twenty-five programs in this book harness that power in a collection of games that take full advantage of the expanded Sinclair system.

Multi-player games on a single screen, multi-player games on a multitude of screens! Multi-player innovations that demonstrate the techniques of network programming, multi-player classics that add a new game-play dimension to old favourites.

Each program is accompanied by an outline of the gameplay along with a description of the system required for play. All the listings are fully documented and offer Spectrum programmers a valuable insight into the techniques and structures of multi-user programs.

**0 330 28674 9**

**U.K. £5.95**  
**Not for sale in Canada**



# 25 PROGRAMS for the SINGLE ZYMIC RODRIVE



Pan  
PCN